HEWLETT-PACKARD

HP-11C

SOLUTIONS HANDBOOK



NOTICE

HEWLETT PACKARD

HP-11C

Solutions Handbook

April 1981

00011-90009

Printed in U.S.A.

Introduction

Congratulations on owning an HP-11C. We know you will be pleased with its quality, versatility and ease of use. Its programmable capability and powerful built-in functions combined with Continuous Memory make it a uniquely useful calculator.

This applications book is designed to help you get the most from your calculator. It provides programs to give you answers to "real world" problems, as well as games and other programs of general interest.

The programs include interesting techniques which you may find useful in writing your own programs. We are confident that you will find this book useful, and we welcome your comments and suggestions.

Contents

A Word About Program Usage

Each program is accompanied by a brief description of the problem, the applicable equations, a listing of program keystrokes, a set of instructions for using the program and one or more example problems showing the actual keystrokes required for the solution.

Program listings are provided in the following format: (This example, which is not a complete listing, is from *Ohm's Law*.)

KEYSTROKES	DISPLAY	KEYSTROKES	DISPLAY
[f] CLEAR [PRGM]	000-	[R/S]	008- 31
[f] [LBL] [A]	001-42,21,11	[f] [LBL] [C]	009-42,21,13
[STO] 0	002- 44 0	[STO] 2	010- 44 2
[g] [CLx]	003- 43 35	[g] [CLx]	011- 43 35
[R/S]	004- 31	[R/S]	012- 31
[f] [LBL] [B]	005-42,21,12	[f] [LBL] [D]	013-42,21,14
[STO] 1	006- 44 1	[STO] 3	014- 44 3
[g] [CLx]	007- 43 35	[g] [CLx]	015- 43 35

The leftmost column, headed KEYSTROKES, shows the keys which must be pressed to enter the program into program memory. All the key designations are identical to the way they appear on the keyboard. The second column, headed DISPLAY, shows the appearance of the display on the calculator as you key in the program. The first three numerals on the left is the line number. This is followed by a dash, then the numeric keycode corresponding to the keystrokes in the KEYSTROKES column. Storage register contents are shown at the end of the program listing.

The user instructions form is your guide to using the program to solve your own problem. The first column, labeled STEP, gives the instruction step number. Steps are executed in sequential order except where otherwise noted. The INSTRUCTIONS column gives instructions and comments concerning the operations to be performed. The INPUT DATA/UNITS column specifies the input data to be supplied, and, if applicable, the units of the data. Data input keys consist of 0 through 9 and decimal point (the numeric keys), [EEX] (enter exponent) and [CHS] (change sign). The KEYSTROKES column specifies the keys to be pressed after keying in the corresponding input data. The OUTPUT DATA/UNITS column specifies intermediate and final outputs, and, where applicable, their units.

The form below is for the same program.

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYSTROKES	OUTPUT DATA/UNITS
1	Clear User mode, then			
	key in the program.			
2	Set User mode.		[f] [USER]	
3	Input two known values			
	(input 0 if unknown).	*V*, volts	[A]	0
		I, amps	[B]	0
		R, ohms	[C]	0
		P, watts	[D]	0
4	Calculate the two unknowns.		[E]	*V*, volts
			[R/S]	*I*, amps
			[R/S]	*R*, ohms
			[R/S]	*P*, watts

Step 1 requires you to key in the program. If the USER annunciator is on, press [f] [USER]. Then put the HP-11C into program mode (press [g] [P/R]) and key in the program steps as shown on the complete listing. Then put the calculator into run mode ([g] [P/R]) and proceed with the user instructions.

Step 2 asks you to set User mode, activating the first five top-row keys as subroutine execution keys, turning [$\sqrt{x}$] into [GSB] [A] etc.

Step 3 asks you to input two of four variables via the user-assigned keys [A] through [E], activated in step 2.

Step 4 calculates the unknown values and begins to display the results with *V* (voltage). The user then presses [R/S] to view the other variables' values.

Mathematics

Coordinate Transformations

This program performs translation, rotation and rescaling of coordinates from one coordinate system to another. A point in the original system (x_0, y_0) becomes the origin of the new system. The x and y axes are rotated through an angle α with respect to the original coordinate system. The new coordinate system may also have a different scale, if desired. A point P having coordinates (x_p, y_p) in the original system will have coordinates (x_p', y_p') in the new system.



Equations:

$$\theta = \text{Tan}^{-1}\left(\frac{y_p - y_0}{x_p - x_0}\right) - \alpha$$

$$D = S\sqrt{(y_p - y_0)^2 + (x_p - x_0)^2}$$

$$x_p' = D\cos\theta$$

$$y_p' = D\sin\theta$$

where:

x_0, y_0 = coordinates of the new origin in the old system.

x_p, y_p = coordinates of a point P in the original system.

x_p', y_p' = coordinates of a point P in the new system.

S = scaling factor.

Input data required are the coordinates of the new origin in the original coordinate system, the rotation angle and the scaling factor (if other than 1).

Points in the original system may be converted to the transformed system using key [A]. Points in the new system may be converted to the original system using key [B].

Remarks:

- The scale factor need not be input unless it is other than one.
- Be sure the rotation angle is input as a decimal and that the calculator is set to the proper mode (i.e., DEG, RAD, GRAD).
- For pure translation, input zero for the rotation angle.
- For pure rotation, input zeros for (x_0, y_0).

KEYSTROKES	DISPLAY	KEYSTROKES	DISPLAY
[f] CLEAR [PRGM]	000–	[R/S]	005– 31
[f] [LBL] [A]	001–42,21,11	[STO] 2	006– 44 2
[STO] 1	002– 44 1	1	007– 1
[R↓]	003– 33	[STO] 3	008– 44 3
[STO] 0	004– 44 0	[R/S]	009– 31

KEYSTROKES	DISPLAY	KEYSTROKES	DISPLAY
[STO] 3	010– 44 3	[R/S]	028– 31
[R/S]	011– 31	[f] [LBL] [C]	029–42,21,13
[f] [LBL] [B]	012–42,21,12	[x≷y]	030– 34
[RCL] 1	013– 45 1	[g] [→P]	031– 43 26
[−]	014– 30	[RCL] 3	032– 45 3
[x≷y]	015– 34	[÷]	033– 10
[RCL] 0	016– 45 0	[x≷y]	034– 34
[−]	017– 30	[RCL] 2	035– 45 2
[g] [→P]	018– 43 26	[+]	036– 40
[RCL] 3	019– 45 3	[x≷y]	037– 34
[×]	020– 20	[f] [→R]	038– 42 26
[x≷y]	021– 34	[RCL] 0	039– 45 0
[RCL] 2	022– 45 2	[+]	040– 40
[−]	023– 30	[R/S]	041– 31
[x≷y]	024– 34	[x≷y]	042– 34
[f] [→R]	025– 42 26	[RCL] 1	043– 45 1
[R/S]	026– 31	[+]	044– 40
[x≷y]	027– 34		

REGISTERS			R_I: Unused
R_0: x_o	R_1: y_o	R_2: α	R_3: S
R_4–$R_{.9}$ Unused			

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYSTROKES	OUTPUT DATA/UNITS
1	Clear User mode, then key in the			
	program.			
2	Set User mode.		[f] [USER]	
3	Set appropriate angular mode.			

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYSTROKES	OUTPUT DATA/UNITS
4	Input the coordinates of the new			
	origin with respect to the original			
	system.	x_o	[ENTER]	
		y_o	[A]	x_o
5	Input the angle of rotation.	α	[R/S]	1
6	Input the scale factor if			
	other than 1.	S	[R/S]	S
7	Transform a point from the old			
	system to the new,	x_p	[ENTER]	
		y_p	[B]	x_p'
			[R/S]	y_p'
	or from the new system to the old.	x_p'	[ENTER]	
		y_p'	[C]	x_p
			[R/S]	y_p
8	Repeat step 7 for all other points.			
9	For a new transformation go to			
	step 3.			

Example: The point (7, -4) in one coordinate system is to become the origin of a new coordinate system. The new system will also be rotated 27° counterclockwise with respect to the original system. Find the new coordinates of the points (−9, 7), (−5, −4) and (6, 8). What were the former coordinates of the new points (2.7, −3.6)?

Keystrokes	**Display**	
Set User mode.		
[g] [DEG]		
[f] [FIX] 4		
7 [ENTER]		
4 [CHS] [A]	7.0000	x_o

Keystrokes	Display	
27 [R/S]	1.0000	
9 [CHS] [ENTER] 7 [B]	-9.2622	x_p'
[R/S]	17.0649	y_p'
5 [CHS] [ENTER]		
4 [CHS] [B]	-10.6921	x_p'
[R/S]	5.4479	y_p'
6 [ENTER] 8 [B]	4.5569	x_p'
[R/S]	11.1461	y_p'
2.7 [ENTER]		
3.6 [CHS] [C]	11.0401	x_p
[R/S]	-5.9818	y_p

Complex Operations

This program allows for chained calculations involving complex numbers in rectangular form. The four operations of complex arithmetic ($+, -, \times, \div$) are provided, as well as several of the most used functions of a complex variable z ($|z|$, $1/z$, z^n, $z^{1/n}$, e^z, $\ln z$, $\sin z$, $\cos z$, $\tan z$, a^z and $\log_a z$). Functions and operations may be mixed in the course of a calculation to allow evaluation of expressions like $z_3/(z_1+z_2)$, $e^{z_1 z_2}$, $|z_1+z_2|+|z_2-z_3|$, etc., where z_1, z_2, z_3 are complex numbers of the form $x+iy$.

The logic system for these programs may be thought of as a kind of Reverse Polish Notation (RPN) with a stack whose capacity is two complex numbers. Let the bottom register of the complex stack be ξ and the top register τ. These are analogous to the X- and T-registers in the calculator's own four-register stack.* A complex number z_1 is input to the ξ-register by the keystrokes y_1 [ENTER] x_1. Upon input of a second complex number z_2 ([ENTER] y_2 [ENTER] x_2), z_1 is moved to τ and z_2 is placed in ξ. The previous contents of τ are lost.

Functions operate on the ξ-register, and the result (except for $|z|$ which returns a real number) is left in ξ. Arithmetic operations involve both the ξ- and τ-registers; the result of the operation is left in ξ.

---

* Each register of the complex stack must actually hold two real numbers; the real and the imaginary part of its complex contents. Thus it takes two of the calculator registers to represent one register in the complex stack. We will speak of the complex stack registers as though they were each just one register.

Equations:

Let $z = x + iy = re^{i\theta}$

and let the result in each case have the form $u + iv$.

$$z_1 + z_2 = (x_1 + x_2) + i(y_1 + y_2) \qquad z_1/z_2 = \frac{r_1}{r_2} e^{i(\theta_1 - \theta_2)}$$

$$z_1 - z_2 = (x_1 - x_2) + i(y_1 - y_2) \qquad |z| = \sqrt{x^2 + y^2}$$

$$z_1 z_2 = r_1 r_2 e^{i(\theta_1 + \theta_2)} \qquad 1/z = \frac{x}{r^2} - i\frac{y}{r^2}$$

$$z^n = r^n e^{in\theta}$$

$$z^{1/n} = r^{1/n} e^{i\left(\frac{\theta}{n} + \frac{360k}{n}\right)}, k = 0, 1, ..., n\text{-}1$$

(All n roots will be output, $k = 0, 1, ..., n-1$)

where n is an integer, and θ is in degrees.

$e^z = e^x (\cos y + i \sin y)$, where y is in radians

$\ln z = \ln r + i\theta$, where $z \neq 0$ and θ is in radians.

$a^z = e^{z \ln a}$, where $a > 0$ and real

$\log_a z = \dfrac{\ln z}{\ln a}$, where $a > 0$ and real, $z \neq 0$

$\sin z = \sin x \cosh y + i \cos x \sinh y$, where x and y are in radians

$\cos z = \cos x \cosh y - i \sin x \sinh y$, where x and y are in radians

$\tan z = \dfrac{\sin 2x + i \sinh 2y}{\cos 2x + \cosh 2y}$, where x and y are in radians

KEYSTROKES	DISPLAY	KEYSTROKES	DISPLAY
f CLEAR PRGM	000-	R↓	029- 33
f LBL A	001-42,21,11	x≷y	030- 34
x≷y	002- 34	R↓	031- 33
R↓	003- 33	×	032- 20
+	004- 40	R↓	033- 33
R↓	005- 33	+	034- 40
+	006- 40	g R↑	035- 43 33
g R↑	007- 43 33	f →R	036- 42 26
g RTN	008- 43 32	g RTN	037- 43 32
f LBL B	009-42,21,12	f LBL D	038-42,21,14
CHS	010- 16	g →P	039- 43 26
x≷y	011- 34	1/x	040- 15
CHS	012- 16	R↓	041- 33
x≷y	013- 34	R↓	042- 33
GTO A	014- 22 11	g →P	043- 43 26
f LBL E	015-42,21,15	x≷y	044- 34
g →P	016- 43 26	R↓	045- 33
1/x	017- 15	×	046- 20
x≷y	018- 34	R↓	047- 33
CHS	019- 16	−	048- 30
x≷y	020- 34	g R↑	049- 43 33
f →R	021- 42 26	f →R	050- 42 26
g RTN	022- 43 32	g RTN	051- 43 32
f LBL C	023-42,21,13	f LBL 0	052-42,21, 0
g →P	024- 43 26	g DEG	053- 43 7
R↓	025- 33	R↓	054- 33
R↓	026- 33	g →P	055- 43 26
g →P	027- 43 26	g R↑	056- 43 33
R↓	028- 33	y^x	057- 14

KEYSTROKES	DISPLAY	KEYSTROKES	DISPLAY
x≷y	058– 34	g →P	087– 43 26
g LSTx	059– 43 36	RCL I	088– 45 25
×	060– 20	1/x	089– 15
x≷y	061– 34	y^x	090– 14
f →R	062– 42 26	x≷y	091– 34
g RTN	063– 43 32	RCL I	092– 45 25
f LBL 1	064–42,21, 1	÷	093– 10
g RAD	065– 43 8	x≷y	094– 34
e^x	066– 12	f LBL 9	095–42,21, 9
f →R	067– 42 26	f →R	096– 42 26
g DEG	068– 43 7	R/S	097– 31
g RTN	069– 43 32	f DSE	098– 42 5
f LBL 2	070–42,21, 2	GTO 9	099– 22 9
g RAD	071– 43 8	g RTN	100– 43 32
g →P	072– 43 26	f LBL 9	101–42,21, 9
g LN	073– 43 12	g →P	102– 43 26
g DEG	074– 43 7	x≷y	103– 34
g RTN	075– 43 32	RCL 0	104– 45 0
f LBL 3	076–42,21, 3	+	105– 40
g DEG	077– 43 7	x≷y	106– 34
g INT	078– 43 44	GTO 9	107– 22 9
STO I	079– 44 25	f LBL 4	108–42,21, 4
3	080– 3	g RAD	109– 43 8
6	081– 6	STO 0	110– 44 0
0	082– 0	COS	111– 24
x≷y	083– 34	x≷y	112– 34
÷	084– 10	STO 1	113– 44 1
STO 0	085– 44 0	f HYP SIN	114–42,22,23
R↓	086– 33	×	115– 20

KEYSTROKES	DISPLAY	KEYSTROKES	DISPLAY
[RCL] 0	116– 45 0	[$x \gtrless y$]	145– 34
[SIN]	117– 23	2	146– 2
[RCL] 1	118– 45 1	[×]	147– 20
[f] [HYP] [COS]	119–42,22,24	[STO] 1	148– 44 1
[×]	120– 20	[f] [HYP] [COS]	149–42,22,24
[g] [DEG]	121– 43 7	[+]	150– 40
[g] [RTN]	122– 43 32	[RCL] 1	151– 45 1
[f] [LBL] 5	123–42,21, 5	[f] [HYP] [SIN]	152–42,22,23
[g] [RAD]	124– 43 8	[$x \gtrless y$]	153– 34
[STO] 0	125– 44 0	[÷]	154– 10
[SIN]	126– 23	[g] [LSTx]	155– 43 36
[$x \gtrless y$]	127– 34	[RCL] 0	156– 45 0
[STO] 1	128– 44 1	[SIN]	157– 23
[f] [HYP] [SIN]	129–42,22,23	[$x \gtrless y$]	158– 34
[×]	130– 20	[÷]	159– 10
[CHS]	131– 16	[g] [DEG]	160– 43 7
[RCL] 0	132– 45 0	[g] [RTN]	161– 43 32
[COS]	133– 24	[f] [LBL] 7	162–42,21, 7
[RCL] 1	134– 45 1	[R↓]	163– 33
[f] [HYP] [COS]	135–42,22,24	[STO] 0	164– 44 0
[×]	136– 20	[$x \gtrless y$]	165– 34
[g] [DEG]	137– 43 7	[g] [R↑]	166– 43 33
[g] [RTN]	138– 43 32	[g] [LN]	167– 43 12
[f] [LBL] 6	139–42,21, 6	[×]	168– 20
[g] [RAD]	140– 43 8	[g] [LSTx]	169– 43 36
2	141– 2	[RCL] 0	170– 45 0
[×]	142– 20	[×]	171– 20
[STO] 0	143– 44 0	[GTO] 1	172– 22 1
[COS]	144– 24	[f] [LBL] 8	173–42,21, 8

KEYSTROKES	DISPLAY	KEYSTROKES	DISPLAY
[g] [LN]	174– 43 12	[÷]	181– 10
[STO] 0	175– 44 0	[x≷y]	182– 34
[R↓]	176– 33	[RCL] 0	183– 45 0
[g] [RAD]	177– 43 8	[÷]	184– 10
[g] [→P]	178– 43 26	[x≷y]	185– 34
[g] [LN]	179– 43 12	[g] [DEG]	186– 43 7
[RCL] 0	180– 45 0		

REGISTERS			R_I: Used
R_0: Used	R_1: Used		

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYSTROKES	OUTPUT DATA/UNITS
1	Clear User mode, then key in the			
	program.			
	(Note: One need only key in lines			
	to line 040 to use the arithmetic			
	operations, the inverse and			
	magnitude functions.)			
2	Set User mode.		[f] [USER]	
	ARITHMETIC			
3	Key in z_1 $(x_1 + iy_1)$.	y_1	[ENTER]	y_1
		x_1	[ENTER]	x_1
4	Key in z_2 $(x_2 + iy_2)$.	y_2	[ENTER]	y_2
		x_2		x_2
5	Choose one of four operations:			
	• Addition (+)		[A]	u
	• Subtraction (−) $(z_1 - z_2)$		[B]	u
	• Multiplication (×)		[C]	u
	• Division (÷) $(z_1 \div z_2)$		[D]	u

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYSTROKES	OUTPUT DATA/UNITS
6	Find the imaginary component.		[x≷y]	v
7	Go to step 3 for more arithmetic.			
	(Note: to use $u + iv$ in further oper-			
	ations, u must be replaced in the X-			
	register and v in the Y-register.)			
	FUNCTIONS			
8	Key in $x + iy$ (or use $u + iv$ from a			
	previous operation).	y	[ENTER]	y
		x		x
9	Choose one of the following			
	operations:			
	• Magnitude ($\|z\|$).		[g] [R→P]	R
	• Reciprocal ($1/z$).		[E]	u
			[x≷y]	v
	• Raise z to an integer power (z^n).		[ENTER]	x
		n	[GSB] 0	u
			[x≷y]	v
	• Raise e to the z power (e^z).		[GSB] 1	u
			[x≷y]	v
	• Take the natural logarithm of			
	z ($\ln z$).		[GSB] 2	u
			[x≷y]	v
	• Find the nth root of z ($z^{1/n}$).			
	(Note: n roots will be found.)		[ENTER]	x
		n	[GSB] 3	u
			[x≷y]	v

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYSTROKES	OUTPUT DATA/UNITS
	Repeat the following three steps n			
	times.		[$x \gtrless y$]	
			[R/S]	u
			[$x \gtrless y$]	v
	● Find the sine of z ($\sin z$).		[GSB] 4	u
			[$x \gtrless y$]	v
	● Find the cosine of z ($\cos z$).		[GSB] 5	u
			[$x \gtrless y$]	v
	● Find the tangent of z ($\tan z$).		[GSB] 6	u
			[$x \gtrless y$]	v
	● Raise an integer (a) to the z			
	power (a^z).		[ENTER]	
		a	[GSB] 7	u
			[$x \gtrless y$]	v
	● Find the logarithm in base a of z			
	($\log_a z$).		[ENTER]	x
		a	[GSB] 8	u
			[$x \gtrless y$]	v
	(Note: to use $u + iv$ in further			
	functions u must be in the X-register			
	and v in the Y-register.)			
10	Go to step 3 for arithmetic or 8 for			
	another function.			

Example 1: Evaluate the expression

$$\frac{z_1}{z_2 + z_3},$$

where: $z_1 = 23 + 13i$, $z_2 = -2 + i$, $z_3 = 4 - 3i$.

Suggestion: Since the program can deal with only two numbers at a time, perform the calculation as

$$z_1 \times [1/(z_2 + z_3)].$$

Keystrokes	Display	
Set User mode.		
[f] [FIX] 4		
1 [ENTER]		
2 [CHS] [ENTER]		z_2
3 [CHS] [ENTER] 4 [A]	2.0000	u; $v = -2.0000$ in Y-register ($u + iv = z_2 + z_3$).
[E]	0.2500	u; $v = 0.2500$ in Y-register ($u + iv = 1/(z_2 + z_3)$).
13 [ENTER] 23 [C]	2.500	u
[x≷y]	9.0000	v

Example 2: Find the 3 cube roots of 8.

Keystrokes	Display	
0 [ENTER] 8 [ENTER]		
3 [GSB] 3	2.0000	u
[x≷y]	0.0000	v
[x≷y]	2.0000	Return u and v to the X- and Y-registers.
[R/S]	-1.0000	u
[x≷y]	1.7321	v
[x≷y]	-1.0000	Return u and v to the X- and Y-registers.
[R/S]	-1.0000	u
[x≷y]	-1.7321	v

Example 3: Evaluate $e^{z^{-2}}$, where $z = (1 + i)$.

Keystrokes	**Display**	
1 [ENTER] [ENTER]		
2 [GSB] 0	**0.0000**	$u; v = 2.0000$ in the Y-register ($u + iv = z^2$).
[E]	**0.0000**	$u; v = -0.5000$ in the Y-register ($u + iv = z^{-2}$).
[GSB] 1	**0.8776**	u
[$x \gtrless y$]	**-0.4794**	v

Example 4: Evaluate sin $(2 + 3i)$.

Keystrokes	**Display**	
3 [ENTER] 2 [GSB] 4	**9.1545**	u
[$x \gtrless y$]	**-4.1689**	v

First- and Second-Order Differential Equations

This program solves first- and second-order differential equations by the fourth-order Runge-Kutta method. A first-order equation is of the form $y' = f(x, y)$, with initial values x_0, y_0; a second-order equation is of the form $y'' = f(x, y, y')$, with initial values x_0, y_0, y'_0.

In either case, the function f should be keyed into program memory under [LBL] 0, and should assume that x_0 and y_0 are in the X- and Y-registers, respectively, and that y'_0 will be in the Z-register for second-order equations. There are 99 lines left after the 1st-order equation program and 66 lines left after the 2nd-order equation program to define the function.

The solution is a numerical one which calculates y_i for $x_i = x_0 + ih$ $(i = 1, 2, 3, ...)$, where h is an increment specified by the user. The value for h may be changed at any time during the program's execution. This allows solution of the equation for values of x arbitrarily close to a pole ($y \to \pm\infty$), or a zero ($y \to 0$).

Equations:

First-Order

$$y_{i+1} = y_i + \frac{1}{6}(c_1 + 2c_2 + 2c_3 + c_4)$$

where

$$c_1 = hf(x_i, y_i)$$

$$c_2 = hf\left(x_i + \frac{h}{2}, y_i + \frac{c_1}{2}\right)$$

$$c_3 = hf\left(x_i + \frac{h}{2}, y_i + \frac{c_2}{2}\right)$$

$$c_4 = hf(x_i + h, y_i + c_3)$$

Second-Order

$$y_{i+1} = y_i + h\left[y'_i + \frac{1}{6}(k_1 + k_2 + k_3)\right]$$

$$y'_{i+1} = y'_i + \frac{1}{6}(k_1 + 2k_2 + 2k_3 + k_4)$$

where

$$k_1 = hf(x_i, y_i, y'_i)$$

$$k_2 = hf\left(x_i + \frac{h}{2}, y_i + \frac{h}{2}y'_i + \frac{h}{8}k_1, y'_i + \frac{k_1}{2}\right)$$

$$k_3 = hf\left(x_i + \frac{h}{2}, y_i + \frac{h}{2}y'_i + \frac{h}{8}k_1, y'_i + \frac{k_2}{2}\right)$$

$$k_4 = hf\left(x_i + h, y_i + hy'_i + \frac{h}{2}k_3, y'_i + k_3\right)$$

Remarks:

- When inputting values for a second-order solution, x_0 and y_0 must be input before y'_0. All values must be input even if zero.
- If the program is to be run on different functions, be sure that the previous function is no longer in program memory when the next is keyed in. Refer to the Instructions for deleting the previous function.

First-Order

KEYSTROKES	DISPLAY	KEYSTROKES	DISPLAY
[f] CLEAR [PRGM]	000–	[STO] 4	020– 44 4
[f] [LBL] [A]	001–42,21,11	[RCL] 2	021– 45 2
2	002– 2	[+]	022– 40
[÷]	003– 10	[RCL] 1	023– 45 1
[STO] 0	004– 44 0	[RCL] 0	024– 45 0
[R↓]	005– 33	[+]	025– 40
[STO] 2	006– 44 2	[GSB] 0	026– 32 0
[R↓]	007– 33	[STO] [+] 4	027–44,40, 4
[STO] 1	008– 44 1	[ENTER]	028– 36
[f] [LBL] 1	009–42,21, 1	[+]	029– 40
[RCL] 2	010– 45 2	[RCL] 2	030– 45 2
[RCL] 1	011– 45 1	[+]	031– 40
[GSB] 0	012– 32 0	[RCL] 1	032– 45 1
[STO] 5	013– 44 5	[RCL] 0	033– 45 0
[RCL] 2	014– 45 2	[ENTER]	034– 36
[+]	015– 40	[+]	035– 40
[RCL] 1	016– 45 1	[STO] 3	036– 44 3
[RCL] 0	017– 45 0	[+]	037– 40
[+]	018– 40	[GSB] 0	038– 32 0
[GSB] 0	019– 32 0	[RCL] 5	039– 45 5

KEYSTROKES	DISPLAY	KEYSTROKES	DISPLAY
[+]	040- 40	[RCL] 2	052- 45 2
[RCL] 4	041- 45 4	[R/S]	053- 31
[ENTER]	042- 36	[GTO] 1	054- 22 1
[+]	043- 40	[f] [LBL] [B]	055-42,21,12
[+]	044- 40	2	056- 2
3	045- 3	[÷]	057- 10
[÷]	046- 10	[STO] 0	058- 44 0
[STO] [+] 2	047-44,40, 2	[GTO] 1	059- 22 1
[RCL] 3	048- 45 3	[f] [LBL] 0	060-42,21, 0
[STO] [+] 1	049-44,40, 1	[RCL] 0	061- 45 0
[RCL] 1	050- 45 1	[×]	062- 20
[R/S]	051- 31		

REGISTERS			R_I: Unused
R_0: $h/2$	R_1: x_i	R_2: y_i	R_3: h
R_4: Used	R_5: Used	R_6–$R_{.9}$: Unused	

Second-Order

KEYSTROKES	DISPLAY	KEYSTROKES	DISPLAY
[f] CLEAR [PRGM]	000-	[R↓]	008- 33
[f] [LBL] [A]	001-42,21,11	[STO] 2	009- 44 2
[STO] 0	002- 44 0	[R↓]	010- 33
[g] [CLx]	003- 43 35	[STO] 1	011- 44 1
2	004- 2	[f] [LBL] 1	012-42,21, 1
[STO] [÷] 0	005-44,10, 0	[RCL] 3	013- 45 3
[R↓]	006- 33	[RCL] 2	014- 45 2
[STO] 3	007- 44 3	[RCL] 1	015- 45 1

KEYSTROKES	DISPLAY	KEYSTROKES	DISPLAY
[GSB] 0	016– 32 0	[RCL] 3	045– 45 3
[STO] 5	017– 44 5	[+]	046– 40
[STO] [I]	018– 44 25	[STO] 3	047– 44 3
[RCL] 0	019– 45 0	[g] [LSTx]	048– 43 36
[GSB] 8	020– 32 8	[RCL] 4	049– 45 4
[STO] 4	021– 44 4	[RCL] 5	050– 45 5
[STO] [I]	022– 44 25	[+]	051– 40
[RCL] 0	023– 45 0	3	052– 3
[RCL] [I]	024– 45 25	[÷]	053– 10
[RCL] 3	025– 45 3	[+]	054– 40
[+]	026– 40	[RCL] 0	055– 45 0
[RCL] 5	027– 45 5	[ENTER]	056– 36
[GSB] 7	028– 32 7	[+]	057– 40
[STO] [+] 4	029–44,40, 4	[STO] 5	058– 44 5
[ENTER]	030– 36	[×]	059– 20
[+]	031– 40	[RCL] 2	060– 45 2
[STO] [I]	032– 44 25	[+]	061– 40
[RCL] 0	033– 45 0	[STO] 2	062– 44 2
[ENTER]	034– 36	[RCL] 5	063– 45 5
[+]	035– 40	[STO] [+] 1	064–44,40, 1
[GSB] 8	036– 32 8	[RCL] 1	065– 45 1
[RCL] 4	037– 45 4	[R/S]	066– 31
[ENTER]	038– 36	[RCL] 2	067– 45 2
[+]	039– 40	[R/S]	068– 31
[+]	040– 40	[GTO] 1	069– 22 1
[RCL] 5	041– 45 5	[f] [LBL] 8	070–42,21, 8
[+]	042– 40	[RCL] [I]	071– 45 25
3	043– 3	[RCL] 3	072– 45 3
[÷]	044– 10	[+]	073– 40

KEYSTROKES	DISPLAY	KEYSTROKES	DISPLAY
[RCL] [I]	074– 45 25	[g] [R↑]	085– 43 33
[f] [LBL] 7	075–42,21, 7	[+]	086– 40
2	076– 2	[f] [LBL] 0	087–42,21, 0
[÷]	077– 10	[RCL] 0	088– 45 0
[RCL] 3	078– 45 3	[×]	089– 20
[+]	079– 40	[g] [RTN]	090– 43 32
[g] [R↑]	080– 43 33	[f] [LBL] [B]	091–42,21,12
[×]	081– 20	2	092– 2
[RCL] 2	082– 45 2	[÷]	093– 10
[+]	083– 40	[STO] 0	094– 44 0
[RCL] 1	084– 45 1	[GTO] 1	095– 22 1

REGISTERS			R_I: Used
R_0: $h/2$	R_1: x_i	R_2: y_i	R_3: y'_i
R_4: Used	R_5: Used	R_6–$R_{.4}$: Unused	

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYSTROKES	OUTPUT DATA/UNITS
	FIRST-ORDER			
1	Clear User mode, then key in the			
	program.			
2	Key in the function:			
	Set Run mode.		[g] [P/R] [GTO] 0	
	Set Program mode.		[g] [P/R]	060–42,21, 0
	Key in the function (does not			
	need a RTN).			
	Set Run mode.		[g] [P/R]	
3	Set User mode.		[f] [USER]	

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYSTROKES	OUTPUT DATA/UNITS
4	Input initial values:	x_0	[ENTER]	
		y_0	[ENTER]	
		h	[A]	x_1
5	Calculate successive values of x			
	and y:		[R/S]	y_1
			[R/S]	x_2
			[R/S]	y_2
			[R/S]	x_3
			⋮	etc.
6	(Optional) To change the value of h			
	during calculation:	new h	[B]	x_i
			[R/S]	y_i
			⋮	etc.
7	For a new function:			
	Delete the existing function		[g] [RTN] [g] [P/R]	
			[g] [BST] [g] [BST]	
			[g] [BST]	
	and go to step 2.		[←] ... [←]	060–42,21, 0

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYSTROKES	OUTPUT DATA/UNITS
	SECOND-ORDER			
1	Clear User mode, then key in the			
	program.			
2	Key in the function:			
	Set Run mode.		[g] [P/R] [GTO] 0	
	Set Program mode.		[g] [P/R]	087–42,21, 0

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYSTROKES	OUTPUT DATA/UNITS
	Key in the function (does not			
	need a RTN).			
	Set Run mode.		[g] [P/R]	
3	Set User mode.		[f] [USER]	
4	Input initial values:	x_0	[ENTER]	
		y_0	[ENTER]	
		y'_0	[ENTER]	
		h	[A]	x_1
5	Calculate successive values of x			
	and y:		[R/S]	y_1
			[R/S]	x_2
			[R/S]	y_2
			[R/S]	x_3
			⋮	etc.
6	(Optional) To change the value of h			
	during calculation:	new h	[B]	x_i
			[R/S]	y_i
			⋮	etc.
7	For a new function:			
	Delete the existing function.		[GTO] [B] [g] [P/R]	
			[g] [BST] [g] [BST]	
			[g] [BST] [g] [BST]	
			[←] ... [←]	087–42,21, 0
	and go to step 2.			

Example 1: Solve numerically the first-order differential equation

$$y' = \frac{\sin x + \tan^{-1}(y/x)}{y - \ln(\sqrt{x^2 + y^2})}$$

where $x_0 = y_0 = 1$. Let $h = 0.5$. The angular mode must be set to radians.

Keystrokes	**Display**	
Clear User mode, then key in the first-order program.		
[GTO] 0		
[g] [P/R]	060–42,21, 0	
[g] [RAD] [STO] 6		
[x≷y] [STO] 7 [x≷y]		
[g] [→P] [g] [LN] [STO] 8		
[R↓] [RCL] 6		
[SIN] [+] [RCL] 7		
[RCL] 8 [−] [÷]	076– 10	
[g] [P/R]		
Set User mode.		
[f] [FIX] 2		
1 [ENTER] [ENTER]		
.5 [A]	1.50	x_1
[R/S]	2.06	y_1
[R/S]	2.00	x_2
[R/S]	2.78	y_2
[R/S]	2.50	x_3
[R/S]	3.28	y_3
etc.		

Example 2: Solve the second-order equation

$$(1 - x^2)\, y'' + xy' = x$$

where $x_0 = y_0 = y'_0 = 0$ and $h = 0.1$.

Rewrite the equation as

$$y'' = \frac{x(y' - 1)}{x^2 - 1}\,, \qquad |x| \neq 1$$

Keystrokes	**Display**	
Clear User mode, then key in the second-order program.		
[GTO] 0		
[g] [P/R]	087–42,21, 0	
[STO] 8 [R↓]		
[R↓] 1 [−]		
[RCL] 8 [×]		
[RCL] 8 [g] [x^2]		
1 [−] [÷]	099– 10	
[g] [P/R]		
Set User mode.		
[f] [FIX] 4		
0 [ENTER]		
[ENTER] [ENTER]		
.1 [A]	0.1000	x_1
[R/S]	0.0002	y_1
[R/S]	0.2000	x_2
[R/S]	0.0013	y_2
[R/S]	0.3000	x_3
[R/S]	0.0046	y_3
etc.		

16-Point Gaussian Quadrature

This program approximates $\int_a^b f(x)dx$ by solving the series

$$\int_a^b f(x)dx \approx \frac{(b-a)}{2}\sum_{k=1}^{8} w_k \left\{ f\left(\frac{b-a}{2}x_k + \frac{a+b}{2}\right) + f\left(\frac{b-a}{2}(-x_k) + \frac{a+b}{2}\right)\right\}$$

where x_k and w_k are constants and are stored in registers R_1 to $R_{.6}$ as folows:

R_1 = 2.715245941–02 = w_1
R_2 = -9.894009350–01 = x_1
R_3 = 6.225352394–02 = w_2
R_4 = -9.445750231–01 = x_2
R_5 = 9.515851168–02 = w_3
R_6 = -8.656312024–01 = x_3

$R_7 = 1.246289713\text{-}01 = w_4$
$R_8 = -7.554044084\text{-}01 = x_4$
$R_9 = 1.495959888\text{-}01 = w_5$
$R_{.0} = -6.178762444\text{-}01 = x_5$
$R_{.1} = 1.691565194\text{-}01 = w_6$
$R_{.2} = -4.580167777\text{-}01 = x_6$
$R_{.3} = 1.826034150\text{-}01 = w_7$
$R_{.4} = -2.816035508\text{-}01 = x_7$
$R_{.5} = 1.894506150\text{-}01 = w_8$
$R_{.6} = -9.501250984\text{-}02 = x_8$

After the Gaussian quadrature program is keyed into program memory, there remain 23 program lines to define the function. The function $f(x)$ should be entered under [LBL] [B] and should assume that x is in the X-register.

Reference: John Kennedy (918), PPC Journal, V6N5, p. 19, Aug. 79.

Note: When dealing with trigonometric functions, be sure the appropriate angular mode is set either before or during the calculation.

KEYSTROKES	DISPLAY	KEYSTROKES	DISPLAY
[f] CLEAR [PRGM]	000–	[STO] [I]	015– 44 25
[f] [LBL] [A]	001–42,21,11	[f] [LBL] 0	016–42,21, 0
[$x \gtrless y$]	002– 34	[RCL] [(i)]	017– 45 24
[STO] [·] 7	003– 44 .7	[GSB] 1	018– 32 1
[−]	004– 30	[STO] [·] 9	019– 44 .9
2	005– 2	[RCL] [(i)]	020– 45 24
[÷]	006– 10	[CHS]	021– 16
[STO] [·] 8	007– 44 .8	[GSB] 1	022– 32 1
[RCL] [·] 7	008– 45 .7	[RCL] [·] 9	023– 45 .9
[+]	009– 40	[+]	024– 40
[STO] [·] 7	010– 44 .7	[f] [DSE]	025– 42 5
0	011– 0	[RCL] [(i)]	026– 45 24
[STO] 0	012– 44 0	[×]	027– 20
1	013– 1	[STO] [+] 0	028–44,40, 0
6	014– 6	[f] [DSE]	029– 42 5

KEYSTROKES	DISPLAY	KEYSTROKES	DISPLAY
[GTO] 0	030– 22 0	[RCL] [·] 8	036– 45 .8
[RCL] [·] 8	031– 45 .8	[×]	037– 20
[STO] [×] 0	032–44,20, 0	[RCL] [·] 7	038– 45 .7
[RCL] 0	033– 45 0	[+]	039– 40
[g] [RTN]	034– 43 32	[f] [LBL] [B]	040–42,21,12
[f] [LBL] 1	035–42,21, 1		

REGISTERS			R_I: k
R_0: $\int_a^b f(x)dx$	R_1: w_1	R_2: x_1	R_3: w_2
R_4: x_2	R_5: w_3	R_6: x_3	R_7: w_4
R_8: x_4	R_9: w_5	$R_{.0}$: x_5	$R_{.1}$: w_6
$R_{.2}$: x_6	$R_{.3}$: w_7	$R_{.4}$: x_7	$R_{.5}$: w_8
$R_{.6}$: x_8	$R_{.7}$: $(b-a)/2$	$R_{.8}$: $(b+a)/2$	
$R_{.9}$: $f((b-a)x_k+(a+b))/2$			

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYSTROKES	OUTPUT DATA/UNITS
1	Clear User mode, then key in the			
	program.			
2	Enter the 16 constants into the			
	registers indicated on pp. 30–31.			
3	Go to "LBL B."		[GTO] [B]	
4	Set Program mode.		[g] [P/R]	040–42,21,12
5	Key in the function to be integrated.	$f(x)$		
6	Set Run mode.		[g] [P/R]	
7	Set User mode.		[f] [USER]	
8	Input the limits of integration and			
	compute the integral.	a	[ENTER]	a
		b	[A]	$\int_a^b f(x)dx$

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYSTROKES	OUTPUT DATA/UNITS
9	For new limits go to step 8.			
10	For a new $f(x)$, go to program end,		[g] [RTN] [g] [P/R]	
	set Program mode and delete steps		[g] [BST]	
	back to "LBL B".		[←] ... [←]	040–42,21,12
	Go to step 5.			

Example: Find $\int_0^\pi \sin x \, dx$ in radians.

Keystrokes	Display	
[f] [SCI] 9		
2.715245941 [EEX]		
2 [CHS] [STO] 1	2.7152459–02	
⋮		
9.501250984 [CHS]		
[EEX] 2 [CHS]		
[STO] [·] 6	–9.5012509–02	
[f] [FIX] 4	–0.0950	
[GTO] [B]		
[g] [P/R]	040–42,21,12	
[g] [RAD]	041– 43 8	
[SIN]	042– 23	
[g] [P/R]	–0.0950	
Set User mode.		
0 [ENTER] [f] [π] [A]	2.0000	Note: Calculator is still in RAD mode.

Example 2: Find $\int_0^4 (5x^2 + 3) \, dx$.

Keystrokes	Display
[g] [RTN]	
[g] [P/R]	000–
[g] [BST]	042– 23
[←]	041– 43 8
[←]	040–42,21,12
[g] [x^2]	041– 43 11

Keystrokes	Display	
5	042-	5
[×]	043-	20
3	044-	3
[+]	045-	40
[g] [P/R]		
0 [ENTER] 4 [A]	118.6667	

Base Conversions

This program will convert a positive integer in base b, x_b, to its equivalent representation in base B, x_B. The bases b and B may take on integer values from 2 to 99, inclusive. Inputs to the program are x_b, b, and B where the single output is the value of x_B. Input of either base, b or B, may be omitted if its value is 10 since a default value of 10 is assigned to both b and B upon input of x_b via key [A]. If several conversions are to be done between the same two bases, i.e., there are several values of x_b for the same b and B, then the bases need not be re-input each time. Once the new value of x_b is keyed in, then pressing [E] will cause the calculation of x_B, based on the most recent values for b and B.

A number such as $4B6_{16}$ cannot be represented directly on the display; therefore we use the convention of allocating two digit locations for each character in R_a when $a > 10$.

For example, $4B6_{16}$ is represented as 041106 by our convention (in the hexadecimal system A = 10, B = 11, C = 12, D = 13, E = 14, F = 15). When displayed, this number may appear as 41106 or with an exponent 4.1106 04 which is interpreted as $4.B6 \times 16^2$. The displayed exponent 4 is for base 10 and only serves to locate the radix point (in the same manner as for decimal numbers). When base $a > 10$ (as in the above example), divide the displayed exponent by 2 to get the true exponent of the number. When the displayed exponent is an odd integer, shift the decimal point of the displayed number one place to the left or right) and adjust its exponent accordingly to make the true exponent an integer.

For example, the displayed number 1.112 03 is interpreted as $B.C \times 16^1$ or $0.BC \times 16^2$.

Remarks:

- When the magnitude of the number is very large or very small, this program will take a long time to execute.
- This program will not give any error indication for invalid inputs for x_b. For example, 981_8 will be treated the same as 1201_8.
- General purpose base conversion programs are inexact. A typical error is exemplified by converting 16777220_{10} to base 8. The program gives 100,000,003.8 instead of the exact answer, 100,000,004.0, an error of 2 counts in the tenth place.

KEYSTROKES	DISPLAY	KEYSTROKES	DISPLAY
[f] CLEAR [PRGM]	000–	[g] [RTN]	019– 43 32
[f] [LBL] [A]	001–42,21,11	[f] [LBL] [D]	020–42,21,14
[STO] 4	002– 44 4	[RCL] 2	021– 45 2
1	003– 1	[STO] 0	022– 44 0
0	004– 0	[RCL] 3	023– 45 3
[STO] 2	005– 44 2	[STO] [I]	024– 44 25
[STO] 3	006– 44 3	1	025– 1
[EEX]	007– 26	0	026– 0
1	008– 1	[f] [$x \neq y$]	027– 42 30
2	009– 2	[GTO] 1	028– 22 1
[STO] 6	010– 44 6	[RCL] 2	029– 45 2
[R↓]	011– 33	[GSB] 3	030– 32 3
[R↓]	012– 33	[STO] 3	031– 44 3
[g] [RTN]	013– 43 32	[GSB] 6	032– 32 6
[f] [LBL] [B]	014–42,21,12	[GTO] 5	033– 22 5
[STO] 3	015– 44 3	[f] [LBL] 1	034–42,21, 1
[g] [RTN]	016– 43 32	[RCL] 2	035– 45 2
[f] [LBL] [C]	017–42,21,13	[f] [$x \neq y$]	036– 42 30
[STO] 2	018– 44 2	[GTO] 2	037– 22 2

KEYSTROKES	DISPLAY	KEYSTROKES	DISPLAY
[RCL] 3	038– 45 3	[g] [RTN]	067– 43 32
[GSB] 3	039– 32 3	[f] [LBL] 6	068–42,21, 6
[STO] 2	040– 44 2	0	069– 0
[GSB] 6	041– 32 6	[STO] 7	070– 44 7
[GTO] 5	042– 22 5	[STO] 1	071– 44 1
[f] [LBL] 2	043–42,21, 2	[RCL] 4	072– 45 4
[RCL] 3	044– 45 3	[f] [LBL] 9	073–42,21, 9
[GSB] 3	045– 32 3	1	074– 1
[STO] 2	046– 44 2	[f] [$x>y$]	075– 42 20
[GSB] 6	047– 32 6	[GTO] 8	076– 22 8
[RCL] 1	048– 45 1	[STO] [+] 7	077–44,40, 7
[STO] 4	049– 44 4	[g] [CLx]	078– 43 35
[RCL] 0	050– 45 0	[RCL] 2	079– 45 2
[STO] 2	051– 44 2	[÷]	080– 10
[GSB] 3	052– 32 3	[STO] 4	081– 44 4
[STO] 3	053– 44 3	[GTO] 9	082– 22 9
[GSB] 6	054– 32 6	[f] [LBL] 8	083–42,21, 8
[f] [LBL] 5	055–42,21, 5	[RCL] 2	084– 45 2
[R/S]	056– 31	[RCL] 4	085– 45 4
[f] [LBL] 3	057–42,21, 3	[×]	086– 20
1	058– 1	[STO] 4	087– 44 4
0	059– 0	[g] [INT]	088– 43 44
[STO] 5	060– 44 5	[RCL] 1	089– 45 1
[$x \gtrless y$]	061– 34	[RCL] 3	090– 45 3
[f] [$x>y$]	062– 42 20	[×]	091– 20
[EEX]	063– 26	[+]	092– 40
1	064– 1	[STO] 1	093– 44 1
[STO] [×] 5	065–44,20, 5	[RCL] 4	094– 45 4
[RCL] 5	066– 45 5	[f] [FRAC]	095– 42 44

KEYSTROKES	DISPLAY	KEYSTROKES	DISPLAY
[STO] 4	096– 44 4	[y^x]	109– 14
1	097– 1	[RCL] 1	110– 45 1
[STO] [−] 7	098–44,30, 7	[×]	111– 20
[RCL] 1	099– 45 1	[STO] 1	112– 44 1
[RCL] 6	100– 45 6	[g] [RTN]	113– 43 32
[f] [$x \leq y$]	101– 42 10	[f] [LBL] [E]	114–42,21,15
[GTO] 4	102– 22 4	[STO] 4	115– 44 4
[RCL] 4	103– 45 4	[RCL] 0	116– 45 0
[g] [$x \neq 0$]	104– 43 30	[STO] 2	117– 44 2
[GTO] 8	105– 22 8	[RCL] [I]	118– 45 25
[f] [LBL] 4	106–42,21, 4	[STO] 3	119– 44 3
[RCL] 3	107– 45 3	[GTO] [D]	120– 22 14
[RCL] 7	108– 45 7		

REGISTERS			R_I: b
R_0: B	R_1: x_B	R_2: B, Used	R_3: b, Used
R_4: x_b	R_5: 10, 100	R_6: 10^{12}	R_7: Used
R_8–$R_{.0}$: Unused			

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYSTROKES	OUTPUT DATA/UNITS
1	Clear User mode, then key in the			
	program.			
2	Set User mode.		[f] [USER]	
3	Key in number in first base.	x_b	[A]	
4	(Optional) Key in first base (if			
	omitted, default value of *b* is 10).	b	[B]	
5	(Optional) Key in second base (if			
	omitted, default value of *B* is 10).	B	[C]	

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYSTROKES	OUTPUT DATA/UNITS
6	Calculate number in second base.		[D]	x_B
7	To convert another number between			
	the same two bases, key in the new			
	x_b and find the new x_B.	x_B	[E]	x_B
8	To change either base, go to			
	step 4.			

Example 1: The following octal numbers ($b = 8$) are addresses of a segment of a program in an HP2100 minicomputer: 177700, 177735, 177777. What are the values of these addresses in base 10 ($B = 10$)?

Keystrokes	Display
Set User mode.	
[f] [FIX] 2	
177700 [A] 8 [B] [D]	**65,472.00**
177735 [E]	**65,501.00**
177777 [E]	**65,535.00**

Example 2: Convert the following octal numbers ($b = 8$) into hexadecimal ($B = 16$): 7.200067×8^{10}, $1.513561778 \times 8^{17}$.

Keystrokes	Display	
7.200067 [EEX]		
10 [A] 8 [B]		
16 [C] [D]	**1.130000 16**	
1.513561778 [EEX]		
17 [E]	**1.302141 25**	
[f] CLEAR [PREFIX]		
(hold)	**1302141404**	$1.302141404 \times 10^{25}$ e.g., $13.02141404 \times 10^{24}$ or $D.2EE4 \times 16^{12}$.

Circle Determined by Three Points

This program calculates the center (x_0, y_0) and radius (r) of the circle defined by three non-collinear points.

Equations:

$$(x - x_0)^2 + (y - y_0)^2 = r^2$$

$$\begin{bmatrix} (x_1 - x_3) & (y_1 - y_3) \\ (x_1 - x_2) & (y_1 - y_2) \end{bmatrix} \begin{bmatrix} x_0 \\ y_0 \end{bmatrix} = \begin{bmatrix} A \\ B \end{bmatrix}$$

where

$$A = \tfrac{1}{2} [(x_1 - x_3)(x_1 + x_3) + (y_1 - y_3)(y_1 + y_3)]$$

$$B = \tfrac{1}{2} [(x_1 - x_2)(x_1 + x_2) + (y_1 - y_2)(y_1 + y_2)]$$

Remarks:

- If the determinant of the system is zero, then the three points are collinear, and the program is not applicable.

KEYSTROKES	DISPLAY	KEYSTROKES	DISPLAY
[f] CLEAR [PRGM]	000-	[GSB] 0	009- 32 0
[f] [LBL] [A]	001-42,21,11	[R/S]	010- 31
[STO] 1	002- 44 1	[f] [LBL] [C]	011-42,21,13
[STO] 8	003- 44 8	5	012- 5
[x≷y]	004- 34	[GSB] 0	013- 32 0
[STO] 0	005- 44 0	[RCL] 7	014- 45 7
[R/S]	006- 31	[RCL] 2	015- 45 2
[f] [LBL] [B]	007-42,21,12	[×]	016- 20
2	008- 2	[RCL] 4	017- 45 4

KEYSTROKES	DISPLAY	KEYSTROKES	DISPLAY
[RCL] 5	018– 45 5	[GTO] 2*	046– 22 2
[×]	019– 20	[f] [LBL] 1	047–42,21, 1
[−]	020– 30	[×]	048– 20
[STO] 8	021– 44 8	[−]	049– 30
[R/S]	022– 31	2	050– 2
[f] [LBL] [D]	023–42,21,14	[÷]	051– 10
[RCL] 6	024– 45 6	[RCL] 8	052– 45 8
[RCL] 2	025– 45 2	[÷]	053– 10
[×]	026– 20	[g] [RTN]	054– 43 32
[RCL] 5	027– 45 5	[f] [LBL] 0	055–42,21, 0
[RCL] 3	028– 45 3	[STO] [I]	056– 44 25
[GSB] 1	029– 32 1	[R↓]	057– 33
[R/S]	030– 31	[ENTER]	058– 36
[RCL] 7	031– 45 7	[ENTER]	059– 36
[RCL] 3	032– 45 3	[RCL] 1	060– 45 1
[×]	033– 20	[−]	061– 30
[RCL] 6	034– 45 6	[CHS]	062– 16
[RCL] 4	035– 45 4	[STO] [(i)]	063– 44 24
[GSB] 1	036– 32 1	[x≷y]	064– 34
[R/S]	037– 31	[RCL] 1	065– 45 1
[x≷y]	038– 34	[+]	066– 40
[RCL] 0	039– 45 0	[×]	067– 20
[−]	040– 30	[f] [ISG]	068– 42 6
[x≷y]	041– 34	[f] [PSE]	069– 42 31
[RCL] 1	042– 45 1	[STO] [(i)]	070– 44 24
[−]	043– 30	[g] [CLx]	071– 43 35
[g] [→P]	044– 43 26	[RCL] 0	072– 45 0
[R/S]	045– 31	[−]	073– 30

*Note that this instruction is used to generate an error message.

KEYSTROKES	DISPLAY	KEYSTROKES	DISPLAY
[CHS]	074– 16	[RCL] 0	079– 45 0
[f] [ISG]	075– 42 6	[+]	080– 40
[f] [PSE]	076– 42 31	[×]	081– 20
[STO] [(i)]	077– 44 24	[f] [DSE]	082– 42 5
[$x \geq y$]	078– 34	[STO] [+] [(i)]	083–44,40,24

REGISTERS			R_I: Index
R_0: x_1	R_1: y_1	R_2: $(y_1 - y_2)$	R_3: Used
R_4: $(x_1 - x_2)$	R_5: $(y_1 - y_3)$	R_6: Used	R_7: $(x_1 - x_3)$
R_8: *Det.*	R_9–$R_{.6}$: Unused		

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYSTROKES	OUTPUT DATA/UNITS
1	Clear User mode, then key in the			
	program.			
2	Set User mode.		[f] [USER]	
3	Input (x_1, y_1)	x_1	[ENTER]	
		y_1	[A]	x_1
4	Input (x_2, y_2)	x_2	[ENTER]	
		y_2	[B]	$2B$
5	Input (x_3, y_3)	x_3	[ENTER]	
		y_3	[C]	*Det.*
6	Calculate			
	center		[D]	x_0
			[R/S]	y_0
	radius		[R/S]	r
7	For each new case, go to step 3.			

Example: Find the equation of the circle that includes the three points (1, 1), (3.5, -7.6), (12, 0.8).

Keystrokes	**Display**	
Set User mode.		
[f] [FIX] 2		
1 [ENTER] [A]	**1.00**	
3.5 [ENTER]		
7.6 [CHS] [B]	**-11.25**	
12 [ENTER] 0.8 [C]	**-94.10**	*Det.*
[D]	**6.45**	x_0
[R/S]	**-2.08**	y_0
[R/S]	**6.26**	r

Vector Operations

This program performs the basic vector operations of addition, dot (scalar) product, and cross product for three-dimensional vectors. It also calculates the angles between two vectors. The program is capable of doing chain calculations whenever the result is a vector (refer to the examples).

Equations: Define a vector V in a three-dimensional rectangular coordinate system.

$$V = x\hat{i} + y\hat{j} + z\hat{k}$$

then:

Vector addition:

$$V_1 + V_2 = (x_1 + x_2)\hat{i} + (y_1 + y_2)\hat{j} + (z_1 + z_2)\hat{k}$$

Dot (scalar) product:

$$V_1 \cdot V_2 = x_1x_2 + y_1y_2 + z_1z_2$$

Cross product:

$$V_1 \times V_2 = (y_1z_2 - z_1y_2)\hat{i} + (z_1x_2 - x_1z_2)\hat{j} + (x_1y_2 - y_1x_2)\hat{k}$$

Angle between vectors:

$$\gamma = \cos^{-1} \frac{V_1 \cdot V_2}{|V_1||V_2|}$$

Remarks:

- For two-dimensional vectors, simply consider that the $\hat{k}$ component does not exist; i.e., input 0 for z's.

KEYSTROKES	DISPLAY	KEYSTROKES	DISPLAY
[f] CLEAR [PRGM]	000–	[RCL] 5	021– 45 5
[f] [LBL] [A]	001–42,21,11	[+]	022– 40
[STO] 3	002– 44 3	[R/S]	023– 31
[R↓]	003– 33	[RCL] 3	024– 45 3
[STO] 2	004– 44 2	[RCL] 6	025– 45 6
[R↓]	005– 33	[+]	026– 40
[STO] 1	006– 44 1	[R/S]	027– 31
[R/S]	007– 31	[f] [LBL] [C]	028–42,21,13
[f] [LBL] 1	008–42,21, 1	[RCL] 1	029– 45 1
[STO] 6	009– 44 6	[RCL] 4	030– 45 4
[R↓]	010– 33	[×]	031– 20
[STO] 5	011– 44 5	[RCL] 2	032– 45 2
[R↓]	012– 33	[RCL] 5	033– 45 5
[STO] 4	013– 44 4	[×]	034– 20
[R/S]	014– 31	[+]	035– 40
[f] [LBL] [B]	015–42,21,12	[RCL] 3	036– 45 3
[RCL] 1	016– 45 1	[RCL] 6	037– 45 6
[RCL] 4	017– 45 4	[×]	038– 20
[+]	018– 40	[+]	039– 40
[R/S]	019– 31	[g] [RTN]	040– 43 32
[RCL] 2	020– 45 2	[f] [LBL] [D]	041–42,21,14

KEYSTROKES	DISPLAY	KEYSTROKES	DISPLAY
[RCL] 2	042– 45 2	[f] [LBL] [E]	066–42,21,15
[RCL] 6	043– 45 6	[GSB] [C]	067– 32 13
[×]	044– 20	[RCL] 1	068– 45 1
[RCL] 3	045– 45 3	[g] [x^2]	069– 43 11
[RCL] 5	046– 45 5	[RCL] 2	070– 45 2
[×]	047– 20	[g] [x^2]	071– 43 11
[−]	048– 30	[+]	072– 40
[R/S]	049– 31	[RCL] 3	073– 45 3
[RCL] 3	050– 45 3	[g] [x^2]	074– 43 11
[RCL] 4	051– 45 4	[+]	075– 40
[×]	052– 20	[RCL] 4	076– 45 4
[RCL] 1	053– 45 1	[g] [x^2]	077– 43 11
[RCL] 6	054– 45 6	[RCL] 5	078– 45 5
[×]	055– 20	[g] [x^2]	079– 43 11
[−]	056– 30	[+]	080– 40
[R/S]	057– 31	[RCL] 6	081– 45 6
[RCL] 1	058– 45 1	[g] [x^2]	082– 43 11
[RCL] 5	059– 45 5	[+]	083– 40
[×]	060– 20	[×]	084– 20
[RCL] 2	061– 45 2	[$\sqrt{x}$]	085– 11
[RCL] 4	062– 45 4	[÷]	086– 10
[×]	063– 20	[g] [$\cos^{-1}$]	087– 43 24
[−]	064– 30	[R/S]	088– 31
[R/S]	065– 31		

REGISTERS			R_I: Unused
R_0: Unused	R_1: x_1	R_2: y_1	R_3: z_1
R_4: x_2	R_5: y_2	R_6: z_2	R_7–$R_{.5}$: Unused

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYSTROKES	OUTPUT DATA/UNITS
1	Clear User mode, then key in the			
	program.			
2	Set User mode.		[f] [USER]	
3	Input the first vector, V_1.	x_1	[ENTER]	
		y_1	[ENTER]	
		z_1	[A]	x_1
4	Input the second vector, V_2.	x_2	[ENTER]	
		y_2	[ENTER]	
		z_2	[R/S]	x_2
	Vector Addition.			
5	Calculate $V_1 + V_2$.		[B]	x
			[R/S]	y
			[R/S]	z
	Dot Product.			
6	Calculate $V_1 \cdot V_2$.		[C]	$V_1 \cdot V_2$
	Cross Product.			
7	Calculate $V_1 \times V_2$.		[D]	x
			[R/S]	y
			[R/S]	z
	Angle Between Two Vectors.			
8	Calculate γ*		[E]	γ
9	To review any results go to any			
	of steps 5 thru 7.			
10	For a new case go to step 3.			

* The units of γ (RAD, DEG, GRAD) depend on the present calculator setting.

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYSTROKES	OUTPUT DATA/UNITS
11	The values of the vector com-			
	ponents are undisturbed in their			
	registers during program execution,			
	therefore, one or the other of the			
	two vectors may be replaced and a			
	new function performed.			
	• To replace V_1 go to step 3.			
	• To replace V_2.	x_2	[ENTER]	
		y_2	[ENTER]	
		z_2	[GSB] 1	x_2

Example 1: $V_1 = (2, 5, 2)$, $V_2 = (3, 3, -4)$

Perform the four functions (addition, dot product, cross product and angle between the vectors) on V_1 and V_2.

Keystrokes	Display	
Set User mode.		
[f] [FIX] 2		
[g] [DEG]		
2 [ENTER] 5 [ENTER]		
2 [A]	2.00	
3 [ENTER] [ENTER]		
4 [CHS] [R/S]	3.00	
[B]	5.00	x (Addition).
[R/S]	8.00	y
[R/S]	-2.00	z
[C]	13.00	$V_1 \cdot V_2$ (Dot product).
[D]	-26.00	x (Cross product).
[R/S]	14.00	y
[R/S]	-9.00	z
[E]	67.16	γ (Angle between the vectors in degrees).

Example 2: Calculate $(V_1 + V_2) \cdot V_3$ where $V_1 = (1.10, 3.00, 4.40)$, $V_2 = (1.24, 2.17, 3.03)$ and $V_3 = (0.072, 0.231, 0.409)$.

Keystrokes	Display	
1.1 [ENTER] 3 [ENTER]		
4.4 [A]	**1.10**	
1.24 [ENTER]		
2.17 [ENTER]		
3.03 [R/S]	**1.24**	
[B]	**2.34**	
[R/S]	**5.17**	$V_1 + V_2$
[R/S]	**7.43**	
[A] 0.072 [ENTER]		
0.231 [ENTER]		
0.409 [R/S]	**0.07**	
[C]	**4.40**	$(V_1 + V_2) \cdot V_3$

Statistics

Normal Distribution

This program evaluates the standard normal density function $f(x)$ and the normal integral $Q(x)$ for a given x. The standard normal distribution has mean 0 and standard deviation 1.

Equations:

$$f(x) = \frac{1}{\sqrt{2\pi}} e^{-x^2/2}$$

$$Q(x) = \frac{1}{\sqrt{2\pi}} \int_x^{\infty} e^{-t^2/2}\, dt$$



For $x \geqslant 0$, a polynomial approximation is used to compute $Q(x)$:

$$Q(x) = f(x)(b_1 t + b_2 t^2 + b_3 t^3 + b_4 t^4 + b_5 t^5) + \epsilon(x)$$

where

$$|\epsilon(x)| < 7.5 \times 10^{-8} \qquad t = \frac{1}{1 + r|x|}$$

and r and b_k are constants stored in registers R_3 through R_8 as follows:

$R_3 = 0.2316419 = r$ $\qquad$ $R_6 = 1.781477937 = b_3$

$R_4 = 1.330274429 = b_5$ $\qquad$ $R_7 = -0.356563782 = b_2$

$R_5 = -1.821255978 = b_4$ $\qquad$ $R_8 = 0.31938153 = b_1$

Note that $f(-x) = f(x)$, $Q(-x) = 1 - Q(x)$.

Reference: *Handbook of Mathematical Functions,* Abramowitz and Stegun, National Bureau of Standards, 1964.

KEYSTROKES	DISPLAY	KEYSTROKES	DISPLAY
[f] CLEAR [PRGM]	000–	[R/S]	015– 31
[f] [LBL] [A]	001–42,21,11	[f] [LBL] [B]	016–42,21,12
[STO] 1	002– 44 1	[RCL] 1	017– 45 1
[ENTER]	003– 36	[g] [$x<0$]	018– 43 10
[×]	004– 20	[GTO] 1	019– 22 1
2	005– 2	[f] [LBL] 2	020–42,21, 2
[÷]	006– 10	1	021– 1
[CHS]	007– 16	[RCL] 1	022– 45 1
[e^x]	008– 12	[RCL] 3	023– 45 3
[f] [π]	009– 42 16	[×]	024– 20
2	010– 2	[+]	025– 40
[×]	011– 20	[$1/x$]	026– 15
[$\sqrt{x}$]	012– 11	[ENTER]	027– 36
[÷]	013– 10	[ENTER]	028– 36
[STO] 2	014– 44 2	[ENTER]	029– 36

KEYSTROKES	DISPLAY	KEYSTROKES	DISPLAY
[RCL] 4	030– 45 4	[×]	043– 20
[×]	031– 20	[RCL] 2	044– 45 2
[RCL] 5	032– 45 5	[×]	045– 20
[+]	033– 40	[g] [RTN]	046– 43 32
[×]	034– 20	[f] [LBL] 1	047–42,21, 1
[RCL] 6	035– 45 6	[RCL] 1	048– 45 1
[+]	036– 40	[CHS]	049– 16
[×]	037– 20	[STO] 1	050– 44 1
[RCL] 7	038– 45 7	[GSB] 2	051– 32 2
[+]	039– 40	1	052– 1
[×]	040– 20	[$x \geq y$]	053– 34
[RCL] 8	041– 45 8	[−]	054– 30
[+]	042– 40		

REGISTERS			R_I: Unused
R_0: Unused	R_1: $\|x\|$	R_2: $f(x)$	R_3: r
R_4: b_5	R_5: b_4	R_6: b_3	R_7: b_2
R_8: b_1	R_9–$R_{.9}$: Unused		

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYSTROKES	OUTPUT DATA/UNITS
1	Clear User mode, then key in the			
	program.			
2	Set User mode.		[f] [USER]	
3	Enter the 6 constants into the			
	registers indicated on page 49.			
4	Calculate $f(x)$ (Note: this is not an	x	[A]	$f(x)$
	optional step).			
5	Calculate $Q(x)$.		[B]	$Q(x)$
6	For new case go to step 4.			

Example: Find $f(x)$ and $Q(x)$ for $x = 1.18$ and $x = -2.28$.

Keystrokes	Display		
Set User mode.			
[f] [SCI] 2			
0.2316419 [STO] 3	**2.31**	**-01**	r
1.330274429 [STO] 4	**1.33**	**00**	b_5
1.821255978 [CHS] [STO] 5	**-1.82**	**00**	b_4
1.781477937 [STO] 6	**1.78**	**00**	b_3
0.356563782 [CHS] [STO] 7	**-3.57**	**-01**	b_2
0.31938153 [STO] 8	**3.19**	**-01**	b_1
1.18 [A]	**1.99**	**-01**	$f(x)$
[B]	**1.19**	**-01**	$Q(x)$
2.28 [CHS] [A]	**2.97**	**-02**	$f(x)$
[B]	**9.89**	**-01**	$Q(x)$

Inverse Normal Distribution

This program determines the value of x such that

$$Q = \frac{1}{\sqrt{2\pi}} \int_x^\infty e^{-t^2/2}\, dt$$

where Q is given and $0 < Q < 1$.

The following rational approximation is used:

$$y = t - \frac{c_0 + c_1 t + c_2 t^2}{1 + d_1 t + d_2 t^2 + d_3 t^3} + \epsilon(Q)$$

where: $|\epsilon(Q)| < 4.5 \times 10^{-4}$

$$t = \begin{cases} \sqrt{\ln \dfrac{1}{Q^2}} & \text{if } 0 < Q \leqslant 0.5 \\ \sqrt{\ln \dfrac{1}{(1-Q)^2}} & \text{if } 0.5 < Q < 1 \end{cases}$$

and c_i and d_i are constants stored in registers R_0 through R_5 as follows:

$R_0 = 2.515517 = c_0$ $\quad$ $R_3 = 1.432788 = d_1$

$R_1 = 0.802853 = c_1$ $\quad$ $R_4 = 0.189269 = d_2$

$R_2 = 0.010328 = c_2$ $\quad$ $R_5 = 0.001308 = d_3$

Therefore

$$x = \begin{cases} y & \text{if } 0 < Q \leq 0.5 \\ -y & \text{if } 0.5 < Q < 1 \end{cases}$$

Reference: *Handbook of Mathematical Functions,* Abramowitz and Stegun, National Bureau of Standards, 1964.

Note: If the display shows **Error 4** after pressing [A], Q is probably out of bounds.

KEYSTROKES	DISPLAY	KEYSTROKES	DISPLAY
[f] CLEAR [PRGM]	000–	[×]	015– 20
[f] [LBL] [A]	001–42,21,11	[1/x]	016– 15
[g] [CF] 1	002–43, 5, 1	[g] [LN]	017– 43 12
[g] [x<0]	003– 43 10	[√x]	018– 11
[GTO] 0	004– 22 0	[STO] [I]	019– 44 25
1	005– 1	[RCL] 2	020– 45 2
[f] [x≤y]	006– 42 10	[×]	021– 20
[GTO] 0	007– 22 0	[RCL] 1	022– 45 1
[R↓]	008– 33	[+]	023– 40
[·]	009– 48	[RCL] [I]	024– 45 25
5	010– 5	[×]	025– 20
[x≷y]	011– 34	[RCL] 0	026– 45 0
[f] [x>y]	012– 42 20	[+]	027– 40
[GSB] 1	013– 32 1	[RCL] [I]	028– 45 25
[ENTER]	014– 36	[RCL] 5	029– 45 5

KEYSTROKES	DISPLAY	KEYSTROKES	DISPLAY
[×]	030– 20	[RCL] [I]	042– 45 25
[RCL] 4	031– 45 4	[x≥y]	043– 34
[+]	032– 40	[−]	044– 30
[RCL] [I]	033– 45 25	[g] [F?] 1	045–43, 6, 1
[×]	034– 20	[CHS]	046– 16
[RCL] 3	035– 45 3	[g] [CF] 1	047–43, 5, 1
[+]	036– 40	[R/S]	048– 31
[RCL] [I]	037– 45 25	[f] [LBL] 1	049–42,21, 1
[×]	038– 20	[g] [SF] 1	050–43, 4, 1
1	039– 1	1	051– 1
[+]	040– 40	[−]	052– 30
[÷]	041– 10	[CHS]	053– 16

REGISTERS			R_I: *t*
R_0: c_0	R_1: c_1	R_2: c_2	R_3: d_1
R_4: d_2	R_5: d_3	R_6–$R_{.9}$ Unused	

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYSTROKES	OUTPUT DATA/UNITS
1	Clear User mode, then key in the			
	program.			
2	Set User mode.		[f] [USER]	
3	Enter the 6 constants into the			
	registers indicated on page 50.			
4	Input *Q* and calculate *x*.	*Q*	[A]	*x*
5	For a new case go to step 4.			

Example: Given $Q = 0.12$ and $Q = 0.95$, find the corresponding x's.

Keystrokes	**Display**	
Set User mode.		
[f] [FIX] 4		
2.515517 [STO] 0	**2.5155**	c_0
0.802853 [STO] 1	**0.8029**	c_1
0.010328 [STO] 2	**0.0103**	c_2
1.432788 [STO] 3	**1.4328**	d_1
0.189269 [STO] 4	**0.1893**	d_2
0.001308 [STO] 5	**0.0013**	d_3
0.12 [A]	**1.1751**	x
0.95 [A]	**-1.6452**	x

Chi-Square Distribution

This program evaluates the chi-square density

$$f(x) = \frac{x^{\frac{\nu}{2}-1} e^{-\frac{x}{2}}}{2^{\frac{\nu}{2}} \Gamma\left(\frac{\nu}{2}\right)}$$

where $x \geqslant 0$ and ν is the degrees of freedom.



A series approximation is used to evaluate the cumulative distribution.

$$P(x) = \int_0^x f(t)dt$$

$$= \left(\frac{x}{2}\right)^{\frac{\nu}{2}} \frac{e^{-\frac{x}{2}}}{\Gamma\left(\frac{\nu+2}{2}\right)} \left[1 + \sum_{k=1}^{\infty} \frac{x^k}{(\nu+2)(\nu+4)\ldots(\nu+2k)}\right]$$

The program computes successive partial sums of the above series. When two consecutive partial sums are equal, the value is used as the sum of the series.

Remarks:

- If $\nu \geqslant 141$, overflow will result.
- If both x and ν are large, $x^{\frac{\nu}{2}-1}$ may overflow the calculator.

Reference: *Handbook of Mathematical Functions,* Abramowitz and Stegun, National Bureau of Standards, 1964.

KEYSTROKES	DISPLAY	KEYSTROKES	DISPLAY
[f] CLEAR [PRGM]	000–	[f] [LBL] [B]	010–42,21,12
[f] [LBL] [A]	001–42,21,11	[STO] 2	011– 44 2
2	002– 2	[RCL] 1	012– 45 1
[÷]	003– 10	1	013– 1
[STO] 1	004– 44 1	[–]	014– 30
1	005– 1	[y^x]	015– 14
[–]	006– 30	[RCL] 2	016– 45 2
[f] [$x!$]	007– 42 0	2	017– 2
[STO] 3	008– 44 3	[÷]	018– 10
[R/S]	009– 31	[CHS]	019– 16

KEYSTROKES	DISPLAY	KEYSTROKES	DISPLAY
[e^x]	020– 12	[STO] 6	038– 44 6
[×]	021– 20	1	039– 1
2	022– 2	[STO] 4	040– 44 4
[RCL] 1	023– 45 1	[f] [LBL] 3	041–42,21, 3
[y^x]	024– 14	[RCL] 2	042– 45 2
[÷]	025– 10	[RCL] 6	043– 45 6
[RCL] 3	026– 45 3	2	044– 2
[÷]	027– 10	[+]	045– 40
[STO] 5	028– 44 5	[STO] 6	046– 44 6
[R/S]	029– 31	[÷]	047– 10
[f] [LBL] [C]	030–42,21,13	[RCL] 4	048– 45 4
[RCL] 2	031– 45 2	[×]	049– 20
[RCL] 1	032– 45 1	[STO] 4	050– 44 4
[÷]	033– 10	[+]	051– 40
[STO] [×] 5	034–44,20, 5	[f] [$x \neq y$]	052– 42 30
2	035– 2	[GTO] 3	053– 22 3
[RCL] 1	036– 45 1	[RCL] 5	054– 45 5
[×]	037– 20	[×]	055– 20

REGISTERS			R_I: Unused
R_0: Unused	R_1: $\nu/2$	R_2: x	R_3: $\Gamma(\nu/2)$
R_4: Used	R_5: Used	R_6: Used	R_7–R_9: Unused

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYSTROKES	OUTPUT DATA/UNITS
1	Clear User mode, then key in the			
	program.			
2	Set User mode.		[f] [USER]	
3	Input degrees of freedom.	ν	[A]	$\Gamma(\nu/2)$
4	Enter x and compute $f(x)$.	x	[B]	$f(x)$
5	Compute $P(x)$.		[C]	$P(x)$
6	For new x go to step 4; for new			
	case go to step 3.			

Example 1: If the degrees of freedom $(\nu) = 20$, find $f(x)$, $P(x)$ for $x = 9.6$ and $x = 15$.

Keystrokes	Display		
Set User mode.			
[f] [SCI] 2			
20 [A]	**3.63**	**05**	$\Gamma(\nu/2)$
9.6 [B]	**1.53**	**-02**	$f(x)$
[C]	**2.51**	**-02**	$P(x)$
15 [B]	**5.72**	**-02**	$f(x)$
[C]	**2.24**	**-01**	$P(x)$

Example 2: If $\nu = 3$, find $f(x)$ and $P(x)$ for $x = 7.82$.

Keystrokes	Display		
3 [A]	**8.86**	**-01**	$\Gamma(\nu/2)$
7.82 [B]	**2.24**	**-02**	$f(x)$
[C]	**9.50**	**-01**	$P(x)$

t Distribution

This program evaluates the integral for the t distribution

$$I(x, \nu) = \int_{-x}^{x} \frac{\Gamma\left(\frac{\nu+1}{2}\right)\left(1+\frac{y^2}{\nu}\right)^{-\frac{\nu+1}{2}}}{\sqrt{\pi\nu}\,\Gamma\left(\frac{\nu}{2}\right)}\,dy$$

where $x > 0$ and ν is the degrees of freedom.



Equations:

For ν even

$I(x, \nu) =$

$$\sin\theta\left[1 + \frac{1}{2}\cos^2\theta + \frac{1\cdot 3}{2\cdot 4}\cos^4\theta + \dots + \frac{1\cdot 3\cdot 5 \dots (\nu-3)}{2\cdot 4\cdot 6 \dots (\nu-2)}\cos^{\nu-2}\theta\right]$$

for ν odd

$$I(x,\nu)=\begin{cases}\dfrac{2\theta}{\pi} & \text{if } \nu=1\\[2ex] \dfrac{2\theta}{\pi}+\dfrac{2}{\pi}\cos\theta\left\{\sin\theta\left[1+\dfrac{2}{3}\cos^2\theta+\ldots\right.\right. & \\[2ex] \left.\left.\quad+\dfrac{2\cdot 4\ldots(\nu-3)}{1\cdot 3\ldots(\nu-2)}\cos^{\nu-3}\theta\right]\right\} & \text{if } \nu>1\end{cases}$$

where

$$\theta=\tan^{-1}\left(\frac{x}{\sqrt{\nu}}\right),\ \theta \text{ is in radians.}$$

Reference: *Handbook of Mathematical Functions,* Abramowitz and Stegun, National Bureau of Standards, 1964.

KEYSTROKES	DISPLAY	KEYSTROKES	DISPLAY
[f] CLEAR [PRGM]	000–	[GTO] 2	014– 22 2
[f] [LBL] [A]	001–42,21,11	0	015– 0
[STO] 1	002– 44 1	[STO] 5	016– 44 5
[g] [RAD]	003– 43 8	[f] [LBL] [B]	017–42,21,12
[$\sqrt{x}$]	004– 11	[RCL] 2	018– 45 2
[÷]	005– 10	[COS]	019– 24
[g] [TAN^{-1}]	006– 43 25	[g] [x^2]	020– 43 11
[STO] 2	007– 44 2	[STO] 3	021– 44 3
[RCL] 1	008– 45 1	[RCL] 2	022– 45 2
2	009– 2	[SIN]	023– 23
[÷]	010– 10	[STO] 4	024– 44 4
[g] [INT]	011– 43 44	[RCL] 1	025– 45 1
[g] [LSTx]	012– 43 36	2	026– 2
[f] [$x\neq y$]	013– 42 30	[f] [$x=y$]	027– 42 40

KEYSTROKES	DISPLAY	KEYSTROKES	DISPLAY
[GTO] 3	028– 22 3	2	056– 2
[÷]	029– 10	[×]	057– 20
1	030– 1	[f] [π]	058– 42 16
[−]	031– 30	[÷]	059– 10
[STO] [I]	032– 44 25	[STO] 7	060– 44 7
1	033– 1	[RCL] 1	061– 45 1
[STO] 6	034– 44 6	1	062– 1
[f] [LBL] 1	035–42,21, 1	[STO] 5	063– 44 5
[RCL] 3	036– 45 3	[STO] [−] 1	064–44,30, 1
[×]	037– 20	[f] [x=y]	065– 42 40
[RCL] 5	038– 45 5	[GTO] 4	066– 22 4
1	039– 1	[GSB] [B]	067– 32 12
[+]	040– 40	[RCL] 2	068– 45 2
[×]	041– 20	[COS]	069– 24
[g] [LSTx]	042– 43 36	[×]	070– 20
1	043– 1	2	071– 2
[+]	044– 40	[×]	072– 20
[STO] 5	045– 44 5	[f] [π]	073– 42 16
[÷]	046– 10	[÷]	074– 10
[STO] [+] 6	047–44,40, 6	[RCL] 7	075– 45 7
[f] [DSE]	048– 42 5	[+]	076– 40
[GTO] 1	049– 22 1	[g] [RTN]	077– 43 32
[RCL] 6	050– 45 6	[f] [LBL] 3	078–42,21, 3
[RCL] 4	051– 45 4	[RCL] 4	079– 45 4
[×]	052– 20	[g] [RTN]	080– 43 32
[g] [RTN]	053– 43 32	[f] [LBL] 4	081–42,21, 4
[f] [LBL] 2	054–42,21, 2	[RCL] 7	082– 45 7
[RCL] 2	055– 45 2		

REGISTERS			R_I: Used
R_0: Unused	R_1: ν or $\nu-1$	R_2: θ	R_3: $\cos^2\theta$
R_4: $\sin\theta$	R_5: Used	R_6: Used	R_7: $2\theta/\pi$
R_8–$R_{.6}$ Unused			

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYSTROKES	OUTPUT DATA/UNITS
1	Clear User mode, then key in the			
	program.			
2	Set User mode.		[f] [USER]	
3	Enter x.	x	[ENTER]	
4	Input ν to calculate $I(x, \nu)$.	ν	[A]	$I(x, \nu)$
5	For new case go to step 3.			
	Note: Calculator is left in RAD			
	mode.			

Example: Find $I(2.201,11)$ and $I(2.75,30)$.

Keystrokes	Display	
Set User mode.		
[f] [FIX] 2		
2.201 [ENTER]	**2.20**	x
11 [A]	**0.95**	$I(x, \nu)$
2.75 [ENTER]	**2.75**	x
30 [A]	**0.99**	$I(x, \nu)$

F Distribution

This program evaluates the integral of the F distribution

$$Q(x) = \int_x^\infty \frac{\Gamma\left(\frac{\nu_1+\nu_2}{2}\right) y^{\frac{\nu_1}{2}-1} \left(\frac{\nu_1}{\nu_2}\right)^{\frac{\nu_1}{2}}}{\Gamma\left(\frac{\nu_1}{2}\right) \Gamma\left(\frac{\nu_2}{2}\right) \left(1+\frac{\nu_1}{\nu_2}y\right)^{\frac{\nu_1+\nu_2}{2}}} dy$$

where $x > 0$ and ν_1 and ν_2 are the degrees of freedom, provided either ν_1 or ν_2 is even.



Equations:

The integral is evaluated by means of the following series:

If ν_1 is even

$$Q(x) = t^{\frac{\nu_2}{2}} \left[1 + \frac{\nu_2}{2}(1-t) + \frac{\nu_2(\nu_2+2)}{2 \cdot 4}(1-t)^2 + \ldots + \frac{\nu_2(\nu_2+2)\ldots(\nu_2+\nu_1-4)}{2 \cdot 4 \ldots (\nu_1-2)}(1-t)^{\frac{\nu_1-2}{2}} \right]$$

If ν_2 is even

$$Q(x) = 1-(1-t)^{\frac{\nu_1}{2}} \left[1 + \frac{\nu_1}{2}t + \frac{\nu_1(\nu_1+2)t^2}{2 \cdot 4} + \ldots + \frac{\nu_1(\nu_1+2)\ldots(\nu_2+\nu_1-4)}{2 \cdot 4 \ldots (\nu_2-2)}t^{\frac{\nu_2-2}{2}} \right]$$

where

$$t = \frac{\nu_2}{\nu_2 + \nu_1 x}$$

Note: If both ν_1 and ν_2 are even, the two formulas will generate identical answers. Using the smaller of ν_1 and ν_2 could save computation time.

KEYSTROKES	DISPLAY	KEYSTROKES	DISPLAY
[f] CLEAR [PRGM]	000-	[STO] 7	021- 44 7
[f] [LBL] [A]	001-42,21,11	[÷]	022- 10
[STO] 1	002- 44 1	[y^x]	023- 14
[g] [RTN]	003- 43 32	[STO] 4	024- 44 4
[f] [LBL] [B]	004-42,21,12	[RCL] 1	025- 45 1
[STO] 2	005- 44 2	2	026- 2
[g] [RTN]	006- 43 32	[−]	027- 30
[f] [LBL] [C]	007-42,21,13	2	028- 2
[RCL] 1	008- 45 1	[÷]	029- 10
[×]	009- 20	[STO] [I]	030- 44 25
[RCL] 2	010- 45 2	[g] [$x=0$]	031- 43 40
[+]	011- 40	[GTO] 1	032- 22 1
[RCL] 2	012- 45 2	1	033- 1
[$x \gtrless y$]	013- 34	[STO] 5	034- 44 5
[÷]	014- 10	[RCL] 3	035- 45 3
[STO] 3	015- 44 3	[−]	036- 30
[g] [RTN]	016- 43 32	[STO] 3	037- 44 3
[f] [LBL] [D]	017-42,21,14	[RCL] 2	038- 45 2
[RCL] 3	018- 45 3	2	039- 2
[RCL] 2	019- 45 2	[÷]	040- 10
2	020- 2	[×]	041- 20

KEYSTROKES	DISPLAY	KEYSTROKES	DISPLAY
[STO] [+] 5	042-44,40, 5	[RCL] 5	063- 45 5
[f] [DSE]	043- 42 5	[RCL] 4	064- 45 4
[GTO] 3	044- 22 3	[×]	065- 20
[GTO] 2	045- 22 2	[g] [RTN]	066- 43 32
[f] [LBL] 3	046-42,21, 3	[f] [LBL] [E]	067-42,21,15
[RCL] 2	047- 45 2	[RCL] 1	068- 45 1
2	048- 2	[RCL] 2	069- 45 2
[+]	049- 40	[STO] 1	070- 44 1
[STO] 2	050- 44 2	[$x \gtrless y$]	071- 34
[RCL] 7	051- 45 7	[STO] 2	072- 44 2
2	052- 2	1	073- 1
[+]	053- 40	[RCL] 3	074- 45 3
[STO] 7	054- 44 7	[−]	075- 30
[÷]	055- 10	[STO] 3	076- 44 3
[RCL] 3	056- 45 3	[GSB] [D]	077- 32 14
[×]	057- 20	1	078- 1
[×]	058- 20	[$x \gtrless y$]	079- 34
[STO] [+] 5	059-44,40, 5	[−]	080- 30
[f] [DSE]	060- 42 5	[R/S]	081- 31
[GTO] 3	061- 22 3	[f] [LBL] 1	082-42,21, 1
[f] [LBL] 2	062-42,21, 2	[RCL] 4	083- 45 4

REGISTERS			R_I: Index
R_0: Unused	R_1: ν_1 or ν_2	R_2: ν_2 or ν_1	R_3: t, $1-t$
R_4: $t^{\nu_2/2}$ or $t^{\nu_1/2}$	R_5: Used	R_6–$R_{.6}$: Unused	

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYSTROKES	OUTPUT DATA/UNITS
1	Clear User mode, then key in the			
	program.			
2	Set User mode.		[f] [USER]	
3	Input ν_1.	ν_1	[A]	ν_1
4	Input ν_2.	ν_2	[B]	ν_2
5	Input x.	x	[C]	t
6	Calculate $Q(x)$.			
	If:			
6a	ν_1 is even.		[D]	$Q(x)$
6b	ν_2 is even.		[E]	$Q(x)$
7	For a new case go to step 3.			

Example 1: Calculate $Q(4.21)$ where $\nu_1 = 7$ and $\nu_2 = 6$.

Keystrokes	Display	
Set User mode.		
[f] [FIX] 2		
7 [A]	**7.00**	
6 [B]	**6.00**	
4.21 [C]	**0.17**	
[E]	**0.05**	$Q(4.21)$

Example 2: Calculate $Q(2.25)$ where $\nu_1 = 4$ and $\nu_2 = 20$.

Keystrokes	Display	
4 [A]	**4.00**	
20 [B]	**20.00**	
2.25 [C]	**0.69**	
[D]	**0.10**	$Q(2.25)$

Moments, Skewness and Kurtosis

For grouped or ungrouped data, moments are used to describe sets of data, skewness is used to measure the lack of symmetry in a distribution, and kurtosis is the relative peakness or flatness of a distribution.

Equations: (For ungrouped data)

1st moment: $\bar{x} = \frac{1}{n} \sum_{i=1}^{n} x_i$

2nd moment: $m_2 = \frac{1}{n} \Sigma x_i^2 - \bar{x}^2$

3rd moment: $m_3 = \frac{1}{n} \Sigma x_i^3 - \frac{3}{n} \bar{x} \Sigma x_i^2 + 2\bar{x}^3$

4th moment: $m_4 = \frac{1}{n} \Sigma x_i^4 - \frac{4}{n} \bar{x} \Sigma x_i^3 + \frac{6}{n} \bar{x}^2 \Sigma x_i^2 - 3\bar{x}^4$

Moment coefficient of skewness

$$\gamma_1 = \frac{m_3}{m_2^{3/2}}$$

Moment coefficient of kurtosis

$$\gamma_2 = \frac{m_4}{m_2^{2}}$$

Grouped data uses similar formulas where

Data	x_1	x_2	...	x_m
Frequency	f_1	f_2	...	f_m

Note that for this case the first moment

$$\bar{x} = \frac{\sum_{i=1}^{m} f_i x_i}{\sum_{i=1}^{m} f_i}.$$

References: *Theory and Problems of Statistics,* M.R. Spiegel, Schaum's Outline, McGraw-Hill, 1961.

KEYSTROKES	DISPLAY	KEYSTROKES	DISPLAY
f CLEAR PRGM	000–	g F? 0	029–43, 6, 0
f LBL A	001–42,21,11	CHS	030– 16
f CLEAR REG	002– 42 34	STO + 0	031–44,40, 0
g CF 0	003–43, 5, 0	RCL 0	032– 45 0
R/S	004– 31	g CF 0	033–43, 5, 0
f LBL B	005–42,21,12	g RTN	034– 43 32
1	006– 1	f LBL D	035–42,21,14
GSB 0	007– 32 0	RCL 2	036– 45 2
R/S	008– 31	RCL 1	037– 45 1
f LBL C	009–42,21,13	÷	038– 10
GSB 0	010– 32 0	STO 6	039– 44 6
R/S	011– 31	R/S	040– 31
f LBL 0	012–42,21, 0	RCL 3	041– 45 3
g F? 0	013–43, 6, 0	RCL 1	042– 45 1
CHS	014– 16	÷	043– 10
STO + 1	015–44,40, 1	RCL 6	044– 45 6
$x \gtrless y$	016– 34	g x^2	045– 43 11
×	017– 20	STO 8	046– 44 8
STO + 2	018–44,40, 2	−	047– 30
g LSTx	019– 43 36	STO 7	048– 44 7
×	020– 20	R/S	049– 31
STO + 3	021–44,40, 3	RCL 4	050– 45 4
g LSTx	022– 43 36	RCL 3	051– 45 3
×	023– 20	RCL 6	052– 45 6
STO + 4	024–44,40, 4	×	053– 20
g LSTx	025– 43 36	3	054– 3
×	026– 20	×	055– 20
STO + 5	027–44,40, 5	−	056– 30
1	028– 1	RCL 1	057– 45 1

KEYSTROKES	DISPLAY	KEYSTROKES	DISPLAY
[÷]	058– 10	[RCL] 1	080– 45 1
[RCL] 6	059– 45 6	[÷]	081– 10
[RCL] 8	060– 45 8	[RCL] 8	082– 45 8
[×]	061– 20	[g] [x^2]	083– 43 11
2	062– 2	3	084– 3
[×]	063– 20	[×]	085– 20
[+]	064– 40	[−]	086– 30
[STO] 9	065– 44 9	[STO] 6	087– 44 6
[R/S]	066– 31	[R/S]	088– 31
[RCL] 5	067– 45 5	[f] [LBL] [E]	089–42,21,15
[RCL] 6	068– 45 6	[RCL] 9	090– 45 9
[RCL] 4	069– 45 4	[RCL] 7	091– 45 7
[×]	070– 20	1	092– 1
4	071– 4	[·]	093– 48
[×]	072– 20	5	094– 5
[−]	073– 30	[y^x]	095– 14
[RCL] 8	074– 45 8	[÷]	096– 10
[RCL] 3	075– 45 3	[R/S]	097– 31
[×]	076– 20	[RCL] 6	098– 45 6
6	077– 6	[RCL] 7	099– 45 7
[×]	078– 20	[g] [x^2]	100– 43 11
[+]	079– 40	[÷]	101– 10

REGISTERS			R_I: Unused
R_0: i	R_1: Σf_i	R_2: $\Sigma f_i x_i$	R_3: $\Sigma f_i x_i^2$
R_4: $\Sigma f_i x_i^3$	R_5: $\Sigma f_i x_i^4$	R_6: $\bar{x}$, m_4	R_7: m_2
R_8: $\bar{x}^2$	R_9: m_3	$R_{.0}$–$R_{.3}$: Unused	

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYSTROKES	OUTPUT DATA/UNITS
1	Clear User mode, then key in the			
	program.			
2	Set User mode.		[f] [USER]	
3	Initialize the program.		[A]	
	Ungrouped Data:			
	Repeat steps 4-5 for $i=1, \ldots, n$.			
4	Input x_i.	x_i	[B]	i
5	(Optional) Correct erroneous x_k.	x_k	[g] [SF] 0 [B]	$k-1$
6	Go to step 9 for moments			
	calculation.			
	Grouped Data:			
	Repeat steps 7-8 for $i=1, \ldots, m$.			
7	Input x_i and f_i.	x_i	[ENTER]	
		f_i	[C]	i
8	(Optional) Correct erroneous x_h *or*			
	f_h.	x_h	[ENTER]	
		f_h	[g] [SF] 0	
			[C]	$h-1$
9	Calculate moments, etc.:			
	$\bar{x}$		[D]	$\bar{x}$
	m_2		[R/S]	m_2
	m_3		[R/S]	m_3
	m_4		[R/S]	m_4
	γ_1		[E]	γ_1
	γ_2		[R/S]	γ_2
10	Repeat step 9 to review results.			
11	For a new case go to step 3.			

Example 1: Ungrouped Data

i	1	2	3	4	5	6	7	8	9
x_i	2.1	3.5	4.2	6.5	4.1	3.6	5.3	3.7	4.9

Keystrokes	Display	
Set User mode.		
[f] [FIX] 4		
[A]		
2.1 [B]	1.000	i
⋮	⋮	
4.9 [B]	9.000	i
[D]	4.2111	$\bar{x}$
[R/S]	1.3899	m_2
[R/S]	0.3864	m_3
[R/S]	5.4894	m_4
[E]	0.2358	γ_1
[R/S]	2.8417	γ_2

Example 2: Grouped Data

i	1	2	3	4	5
x_i	3	2	4	6	1
f_i	4	5	3	2	1

Keystrokes	Display	
[A] 3 [ENTER] 4 [C]	1.000	
⋮	⋮	
1 [ENTER] [C]	5.0000	i
[D]	3.1333	$\bar{x}$
[R/S]	1.9822	m_2
[R/S]	2.1381	m_3
[R/S]	11.0479	m_4
[E]	0.7661	γ_1
[R/S]	2.8117	γ_2

Analysis of Variance (One Way)

The one-way analysis of variance tests the differences between the population means of k treatment groups. Group i ($i = 1, 2, ..., k$) has n_i observations (treatment groups may have equal or unequal numbers of observations.

Equations:

$$\text{Sum}_i = \text{sum of observations in treatment group } i = \sum_{j=1}^{n_i} x_{ij}$$

$$\text{Total } SS = \sum_{i=1}^{k} \sum_{j=1}^{n_i} x_{ij}^2 - \frac{\left(\sum_{i=1}^{k} \sum_{j=1}^{n_i} x_{ij}\right)^2}{\sum_{i=1}^{k} n_i}$$

$$\text{Treat } SS = \sum_{i=1}^{k} \left(\frac{\left(\sum_{j=1}^{n_i} x_{ij}\right)^2}{n_i} \right) - \frac{\left(\sum_{i=1}^{k} \sum_{j=1}^{n_i} x_{ij}\right)^2}{\sum_{i=1}^{k} n_i}$$

$$\text{Error } SS = \text{Total } SS - \text{Treat } SS$$

$$df_1 = \text{Treat } df = k - 1$$

$$df_2 = \text{Error } df = \left(\sum_{i=1}^{k} n_i\right) - k$$

$$\text{Treat } MS = \frac{\text{Treat } SS}{\text{Treat } df}$$

$$\text{Error } MS = \frac{\text{Error } SS}{\text{Error } df}$$

$$F = \frac{\text{Treat } MS}{\text{Error } MS} \text{(with } k-1 \text{ and } \sum_{i=1}^{k} n_i - k \text{ degrees of freedom)}$$

Reference: *Mathematical Statistics,* J.E. Freund, Prentice Hall, 1962.

KEYSTROKES	DISPLAY	KEYSTROKES	DISPLAY
[f] CLEAR [PRGM]	000-	[RCL] 8	024- 45 8
[f] [LBL] [A]	001-42,21,11	[÷]	025- 10
[f] CLEAR [REG]	002- 42 34	[−]	026- 30
[R/S]	003- 31	[STO] 1	027- 44 1
[f] [LBL] [B]	004-42,21,12	[RCL] 6	028- 45 6
1	005- 1	[RCL] 9	029- 45 9
[STO] [+] 7	006-44,40, 7	[g] [x^2]	030- 43 11
[RCL] 1	007- 45 1	[RCL] 8	031- 45 8
[STO] [·] 0	008- 44 .0	[÷]	032- 10
[STO] [+] 9	009-44,40, 9	[−]	033- 30
[g] [x^2]	010- 43 11	[STO] 0	034- 44 0
[RCL] 0	011- 45 0	[−]	035- 30
[STO] [+] 8	012-44,40, 8	[STO] 6	036- 44 6
[÷]	013- 10	[g] [LSTx]	037- 43 36
[STO] [+] 6	014-44,40, 6	[RCL] 7	038- 45 7
0	015- 0	1	039- 1
[STO] 0	016- 44 0	[−]	040- 30
[STO] 1	017- 44 1	[STO] [·] 0	041- 44 .0
[RCL] [·] 0	018- 45 .0	[÷]	042- 10
[R/S]	019- 31	[$x \gtrless y$]	043- 34
[f] [LBL] [C]	020-42,21,13	[RCL] 8	044- 45 8
[RCL] 2	021- 45 2	[RCL] 7	045- 45 7
[RCL] 9	022- 45 9	[−]	046- 30
[g] [x^2]	023- 43 11	[STO] [·] 1	047- 44 .1

KEYSTROKES	DISPLAY	KEYSTROKES	DISPLAY
[÷]	048– 10	[RCL] [·] 0	051– 45 .0
[÷]	049– 10	[R/S]	052– 31
[R/S]	050– 31	[RCL] [·] 1	053– 45 .1

REGISTERS			R_I: Unused
R_0: Used	R_1: Used	R_2: $\Sigma\Sigma x_{ij}^2$	R_3: Σy
R_4: Σy^2	R_5: Σxy	R_6: Used	R_7: Used
R_8: Σn_i	R_9: $\Sigma\Sigma x_{ij}$	$R_{.0}$ df_1	$R_{.1}$: df_2
$R_{.2}$–$R_{.9}$: Unused			

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYSTROKES	OUTPUT DATA/UNITS
1	Clear User mode, then key in the			
	program.			
2	Set User mode.		[f] [USER]	
3	Initialize the program.		[A]	
4	Repeat steps 5 to 8 for $i=1, \ldots, k$.			
5	Repeat steps 6 to 8 for $j=1, \ldots, k$.			
6	Input x_{ij}.	x_{ij}	[Σ+]	j
7	(Optional) Delete erroneous x_{ij}.	x_{ij}	[g] [Σ-]	j-1
8	Calculate Σx_{ij}*		[B]	Σx_{ij}*
9	Calculate F.		[C]	F
10	Calculate df_1.		[R/S]	df_1
11	Calculate df_2.		[R/S]	df_2
12	For a new case go to step 3.			

* $$\sum_{j=1}^{n_i} x_{ij}$$

Example: Find Sum_1, Sum_2, Sum_3, F, df_1, and df_2 for the following:

	i \ j	1	2	3	4	5	6
	1	10	8	5	12	14	11
Treatment	*2*	6	9	8	13		
	3	14	13	10	17	16	

Keystrokes	**Display**	
Set User mode.		
[f] [FIX] 2		
[A] 10 [Σ+]	**1.00**	
8 [Σ+]	**2.00**	
⋮	⋮	
11 [Σ+]	**6.00**	
[B]	**60.00**	Sum_1
6 [Σ+]	**1.00**	
⋮	⋮	
13 [Σ+]	**4.00**	
[B]	**36.00**	Sum_2
14 [Σ+]	**1.00**	
⋮	⋮	
16 [Σ+]	**5.00**	
[B]	**70.00**	Sum_3
[C]	**3.79**	F
[R/S]	**2.00**	df_1
[R/S]	**12.00**	df_2

Spearman's Rank Correlation Coefficient

Spearman's rank correlation coefficient is a measure of rank correlation under the following circumstances: n individuals are ranked from 1 to n according to some specific characteristic by 2 observers, and we wish to know if the 2 rankings are substantially in agreement with one another.

Spearman's rank correlation coefficient is defined by

$$r_s = 1 - \frac{6 \sum_{i=1}^{n} D_i^2}{n(n^2-1)}$$

where n = number of paired observations (x_i, y_i)

$D_i = \text{rank}(x_i) - \text{rank}(y_i) = R_i - S_i$

If the x and y random variables from which these n pairs of observations are derived are independent, then r_s has zero mean and a variance equal to

$$\frac{1}{n-1}.$$

A test for the null hypothesis

$$H_0:\ x, y \text{ are independent}$$

uses

$$z = r_s \sqrt{n-1}$$

which is approximately a standardized normal variable (for large n, say $n \geqslant 10$). If the null hypothesis of independence is not rejected, we infer that the population correlation coefficient $\rho(x,y) = 0$, but dependence between the variables does not necessarily imply that $\rho(x,y) \neq 0$.

Note: $-1 \leqslant r_s \leqslant 1$

$r_s = 1$ indicates complete agreement in order of the ranks and $r_s = -1$ indicates complete agreement in the opposite order of the ranks.

Reference: *Nonparametric Statistical Inference*, J.D. Gibbons, McGraw-Hill, 1971.

KEYSTROKES	DISPLAY	KEYSTROKES	DISPLAY
[f] CLEAR [PRGM]	000-	[×]	016- 20
[f] [LBL] [A]	001-42,21,11	[RCL] 0	017- 45 0
[f] CLEAR [Σ]	002- 42 32	[g] [x^2]	018- 43 11
[g] [RTN]	003- 43 32	1	019- 1
[f] [LBL] [B]	004-42,21,12	[−]	020- 30
[−]	005- 30	[RCL] 0	021- 45 0
[Σ+]	006- 49	[×]	022- 20
[g] [RTN]	007- 43 32	[÷]	023- 10
[f] [LBL] [C]	008-42,21,13	[−]	024- 30
[−]	009- 30	[R/S]	025- 31
[g] [Σ−]	010- 43 49	[RCL] 0	026- 45 0
[g] [RTN]	011- 43 32	1	027- 1
[f] [LBL] [D]	012-42,21,14	[−]	028- 30
1	013- 1	[$\sqrt{x}$]	029- 11
[RCL] 2	014- 45 2	[×]	030- 20
6	015- 6		

REGISTERS			R_I: Unused
R_0: n	R_1: Used	R_2: Used	R_3: Used
R_4: Used	R_5: Used	R_6–R_9 Unused	

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYSTROKES	OUTPUT DATA/UNITS
1	Clear User mode, then key in the			
	program.			
2	Set User mode.		[f] [USER]	
3	Initialize.		[A]	
	Repeat steps 4 and 5 for			
	$i = 1, \ldots, n$.			

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYSTROKES	OUTPUT DATA/UNITS
4	Input:			
		R_i	[ENTER]	
		S_i	[B]	i
5	(Optional) Correct erroneous R_k	R_k	[ENTER]	
	and/or S_k	S_k	[C]	$k-1$
6	Calculate:			
	r_s		[D]	r_s
	z		[R/S]	z
7	For a new case go to step 3.			

Example: The following data set is the result of two tests in a class; find r_s and z.

Student	x_1 Math Grade	y_1 Stat Grade	R_1 Rank of x_1	S_1 Rank of y_1
1	82	81	6	7
2	67	75	14	11
3	91	85	3	4
4	98	90	1	2
5	74	80	11	8
6	52	60	15	15
7	86	94	4	1
8	95	78	2	9
9	79	83	9	6
10	78	76	10	10
11	84	84	5	5
12	80	69	8	13
13	69	72	13	12
14	81	88	7	3
15	73	61	12	14

Keystrokes	Display	
Set User mode.		
[f] [FIX] 2		
[A]		
6 [ENTER] 7 [B]	1.00	
14 [ENTER] 11 [B]	2.00	
3 [ENTER] 4 [B]	3.00	
1 [ENTER] 2 [B]	4.00	
11 [ENTER] 8 [B]	5.00	
5 [ENTER] [B]	6.00	Error.
5 [ENTER] [C]	5.00	Correction.
15 [ENTER] [B]	6.00	
4 [ENTER] 1 [B]	7.00	
2 [ENTER] 9 [B]	8.00	
9 [ENTER] 6 [B]	9.00	
10 [ENTER] [B]	10.00	
5 [ENTER] [B]	11.00	
8 [ENTER] 13 [B]	12.00	
13 [ENTER] 12 [B]	13.00	
7 [ENTER] 3 [B]	14.00	
12 [ENTER] 14 [B]	15.00	
[D]	0.76	r_s
[R/S]	2.85	z

Contingency Tables

Contingency tables can be used to test the null hypothesis that two variables are independent.

This program calculates the χ^2 statistic for testing the independence of the two variables. Also, Pearson's coefficient of contingency, c_c, which measures the degree of association between the two variables, is calculated.

2 × *k* Contingency Table

i \ j	1	2	...	k	Totals
1	x_{11}	x_{12}	...	x_{1k}	R_1
2	x_{21}	x_{22}	...	x_{2k}	R_2
Totals	C_1	C_2	...	C_k	T

3 × k Contingency Table

i \ j	1	2	...	k	Totals
1	x_{11}	x_{12}	...	x_{1k}	R_1
2	x_{21}	x_{22}	...	x_{2k}	R_2
3	x_{31}	x_{32}	...	x_{3k}	R_3
Totals	C_1	C_2	...	C_k	T

Equations:

$$\text{Row Sum } R_i = \sum_{i=1}^{k} x_{ij} \qquad \begin{array}{l} i = 1, 2 \text{ (for } 2 \times k) \\ i = 1, 2, 3 \text{ (for } 3 \times k) \end{array}$$

$$\text{Column Sum } C_j = \sum_{j=1}^{n} x_{ij} \qquad \begin{array}{ll} j = 1, 2, \ldots, k & \\ n = 2 & (\text{for } 2 \times k) \\ n = 3 & (\text{for } 3 \times k) \end{array}$$

$$\text{Total } T = \sum_{i=1}^{n} \sum_{j=1}^{k} x_{ij} \qquad \begin{array}{l} n = 2 \text{ (for } 2 \times k) \\ n = 3 \text{ (for } 3 \times k) \end{array}$$

Chi-square statistic

$$\chi^2 = \sum_{i=1}^{n} \sum_{j=1}^{k} \frac{(x_{ij} - E_{ij})^2}{E_{ij}} = T\left(\sum_{i=1}^{n} \sum_{j=1}^{k} \frac{x_{ij}^{\ 2}}{R_i C_j}\right) - T$$

where R_i = observed number in ith row,

C_j = observed number jth row,

E_{ij} = expected number in the (ij)th cell,

$df = (n-1)(k-1)$, $n = 2$ for $2 \times k$ and $n = 3$ for $3 \times k$.

Contingency coefficient

$$c_c = \frac{\chi^2}{T + \chi^2}$$

Reference: B. Ostle, *Statistics in Research*, Iowa State University Press, 1972.

KEYSTROKES	DISPLAY	KEYSTROKES	DISPLAY
[f] CLEAR [PRGM]	000-	[RCL] 6	029- 45 6
[f] [LBL] 0	001-42,21, 0	[RCL] 3	030- 45 3
[f] CLEAR [REG]	002- 42 34	[÷]	031- 10
[g] [CF] 0	003-43, 5, 0	[STO] [+] 9	032-44,40, 9
[g] [CF] 1	004-43, 5, 1	[RCL] 9	033- 45 9
0	005- 0	1	034- 1
[R/S]	006- 31	[−]	035- 30
[f] [LBL] [B]	007-42,21,12	[RCL] 0	036- 45 0
[g] [SF] 0	008-43, 4, 0	[×]	037- 20
[f] [LBL] [A]	009-42,21,11	[R/S]	038- 31
[g] [F?] 1	010-43, 6, 1	[ENTER]	039- 36
0	011- 0	[ENTER]	040- 36
[GSB] 9	012- 32 9	[RCL] 0	041- 45 0
[R/S]	013- 31	[+]	042- 40
[RCL] 7	014- 45 7	[÷]	043- 10
[R/S]	015- 31	[$\sqrt{x}$]	044- 11
[f] [LBL] [C]	016-42,21,13	[R/S]	045- 31
[g] [F?] 1	017-43, 6, 1	[f] [LBL] [D]	046-42,21,14
1	018- 1	[RCL] 1	047- 45 1
[g] [F?] 1	019-43, 6, 1	[R/S]	048- 31
[STO] 3	020- 44 3	[RCL] 2	049- 45 2
[RCL] 4	021- 45 4	[R/S]	050- 31
[RCL] 1	022- 45 1	[g] [F?] 1	051-43, 6, 1
[÷]	023- 10	[GTO] [E]	052- 22 15
[STO] 9	024- 44 9	[RCL] 3	053- 45 3
[RCL] 5	025- 45 5	[R/S]	054- 31
[RCL] 2	026- 45 2	[f] [LBL] [E]	055-42,21,15
[÷]	027- 10	[RCL] 0	056- 45 0
[STO] [+] 9	028-44,40, 9	[R/S]	057- 31

KEYSTROKES	DISPLAY	KEYSTROKES	DISPLAY
[GTO] [C]	058– 22 13	[STO] [+] 0	079–44,40, 0
[f] [LBL] 9	059–42,21, 9	[STO] [+] 7	080–44,40, 7
[g] [F?] 0	060–43, 6, 0	[g] [x^2]	081– 43 11
[CHS]	061– 16	[RCL] 7	082– 45 7
[STO] [+] 3	062–44,40, 3	[÷]	083– 10
[STO] [+] 0	063–44,40, 0	[STO] [+] 4	084–44,40, 4
[STO] 7	064– 44 7	[RCL] 9	085– 45 9
[g] [x^2]	065– 43 11	[RCL] 7	086– 45 7
[STO] 8	066– 44 8	[÷]	087– 10
[R↓]	067– 33	[STO] [+] 5	088–44,40, 5
[g] [F?] 0	068–43, 6, 0	[RCL] 8	089– 45 8
[CHS]	069– 16	[RCL] 7	090– 45 7
[STO] [+] 2	070–44,40, 2	[÷]	091– 10
[STO] [+] 0	071–44,40, 0	[STO] [+] 6	092–44,40, 6
[STO] [+] 7	072–44,40, 7	[RCL] [I]	093– 45 25
[g] [x^2]	073– 43 11	1	094– 1
[STO] 9	074– 44 9	[g] [F?] 0	095–43, 6, 0
[R↓]	075– 33	[CHS]	096– 16
[g] [F?] 0	076–43, 6, 0	[+]	097– 40
[CHS]	077– 16	[STO] [I]	098– 44 25
[STO] [+] 1	078–44,40, 1	[g] [CF] 0	099–43, 5, 0

REGISTERS			R_I: Index
R_0: T	R_1: R_1	R_2: R_2	R_3: R_3
R_4: $\Sigma x_{1j}{}^2/c_j$	R_5: $\Sigma x_{2j}{}^2/c_j$	R_6: $\Sigma x_{3j}{}^2/c_j$	R_7: c_j
R_8: $x_{3j}{}^2$	R_9: $x_{2j}{}^2$	$R_{.0}$—$R_{.3}$ Unused	

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYSTROKES	OUTPUT DATA/UNITS
1	Clear User mode, then key in the			
	program.			
2	Set User mode.		[f] [USER]	
	2 × *k*			
3	Initialize the program.		[GSB] 0	
			[g] [SF] 1	
	Repeat steps 4-7 for $j = 1, 2, ..., k$.			
4	Input:			
		x_{1j}	[ENTER]	
		x_{2j}	[A]	j
5	(Optional) Calculate column sum c_j.		[R/S]	c_j
6	(Optional) To correct erroneous x_{1h}	x_{1h}	[ENTER]	
	and/or x_{2h}:	x_{2h}	[B]	$h-1$
7	(Optional) Calculate column sum			
	correction c_h.		[R/S]	$-c_h$
8	Go to step 14 for contingency table			
	calculations.			
	3 × *k*			
9	Initialize the program.		[GSB] 0	
	Repeat steps 10-13 for			
	$j = 1, 2, ..., k$.			
10	Input:			
		x_{1j}	[ENTER]	
		x_{2j}	[ENTER]	
		x_{3j}	[A]	j
11	(Optional) Calculate column sum c_j.		[R/S]	c_j

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYSTROKES	OUTPUT DATA/UNITS
12	(Optional) To correct erroneous x_{1h},	x_{1h}	[ENTER]	
	x_{2h}, and x_{3h}.	x_{2h}	[ENTER]	
		x_{3h}	[B]	$h-1$
13	(Optional) Calculate column sum			
	correction c_h.		[R/S]	$-c_h$
14	Calculate test statistics:			
	χ^2		[C]	χ^2
	c_c		[R/S]	c_c
	Row sum 1		[D]	R_1
	Row sum 2		[R/S]	R_2
	Row sum 3 (for $3 \times k$ case only)		[R/S]	R_3
	Total T.		[E]	T
15	Repeat step 14 to review results.			
16	For another case, go to step 3 or			
	step 9.			

Example 1: Calculate χ^2, c_c and T for the following table.

i \ j	1	2	3
A	2	5	4
B	3	8	7

Keystrokes	**Display**	
Set User mode.		
[f] [FIX] 4		
[GSB] 0 [g] [SF] 1	0.0000	
2 [ENTER] 3 [A]	1.0000	
5 [ENTER] 8 [A]	2.0000	
6 [ENTER] 9 [A]	3.0000	Error
[R/S]	15.0000	C_3

Keystrokes	Display	
6 [ENTER] 9 [B]	2.0000	Correction
[R/S]	-15.0000	$-C_3$
4 [ENTER] 7 [A]	3.0000	
[C]	0.0221	χ^2
[R/S]	0.0276	c_c
[D]	11.0000	R_1
[R/S]	18.0000	R_2
[E]	29.0000	T

Example 2: Find the test statistic χ^2 and the coefficient of contingency c_c for the following set of data.

i \ j	1	2	3	4
1	36	67	49	58
2	31	60	49	54
3	58	87	80	68

Keystrokes	Display	
[GSB] 0	0.0000	
36 [ENTER] 31 [ENTER]		
58 [A]	1.0000	
67 [ENTER] 60 [ENTER]		
87 [A]	2.0000	
4 [ENTER] 49 [ENTER]		
80 [A]	3.0000	
[R/S]	133.0000	$-C_3$
4 [ENTER] 49 [ENTER]		
80 [B]	2.0000	
[R/S]	-133.0000	$-C_3$
49 [ENTER] [ENTER]		
80 [A]	3.0000	

Keystrokes	Display	
58 [ENTER] 54 [ENTER]		
68 [A]	**4.0000**	
[C]	**3.3574**	χ^2
[R/S]	**0.0692**	c_c
[D]	**210.0000**	R_1
[R/S]	**194.0000**	R_2
[R/S]	**293.0000**	R_3
[E]	**697.0000**	T

Electrical Engineering

Reactance Chart

This program calculates inductive and capacitive reactance and provides interchangeable solutions between frequency, inductance and capacitance at resonance. Since, at resonance, capacitive and inductive reactances are equal, they will both be termed simply reactance.

Equations:

$$f = \frac{1}{2\pi\sqrt{LC}}$$

$$X = \frac{1}{2\pi f C}$$

where

f = resonance frequency (hertz)

L = inductance (henrys)

C = capacitance (farads)

X = reactance (ohms)

KEYSTROKES	DISPLAY	KEYSTROKES	DISPLAY
[f] CLEAR [PRGM]	000–	[STO] 1	008– 44 1
[RCL] 1	001– 45 1	[g] [CLx]	009– 43 35
[R/S]	002– 31	[R/S]	010– 31
[RCL] 2	003– 45 2	[f] [LBL] [B]	011–42,21,12
[R/S]	004– 31	[STO] 2	012– 44 2
[RCL] 3	005– 45 3	[g] [CLx]	013– 43 35
[R/S]	006– 31	[R/S]	014– 31
[f] [LBL] [A]	007–42,21,11	[f] [LBL] [C]	015–42,21,13

KEYSTROKES	DISPLAY	KEYSTROKES	DISPLAY
[STO] 3	016– 44 3	[R↓]	036– 33
[g] [CLx]	017– 43 35	[RCL] 3	037– 45 3
[R/S]	018– 31	[÷]	038– 10
[f] [LBL] [D]	019–42,21,14	[STO] 2	039– 44 2
[RCL] 1	020– 45 1	[GTO] 2	040– 22 2
[g] [x≠0]	021– 43 30	[f] [LBL] 1	041–42,21, 1
[GTO] 0	022– 22 0	[÷]	042– 10
[RCL] 2	023– 45 2	[STO] 3	043– 44 3
[RCL] 3	024– 45 3	[f] [LBL] 2	044–42,21, 2
[×]	025– 20	[RCL] 1	045– 45 1
[√x]	026– 11	[RCL] 3	046– 45 3
[GSB] 3	027– 32 3	[×]	047– 20
[STO] 1	028– 44 1	[f] [LBL] 3	048–42,21, 3
[GTO] 2	029– 22 2	[f] [π]	049– 42 16
[f] [LBL] 0	030–44,21, 0	[×]	050– 20
[GSB] 3	031– 32 3	2	051– 2
[g] [x^2]	032– 43 11	[×]	052– 20
[RCL] 2	033– 45 2	[1/x]	053– 15
[g] [x≠0]	034– 43 30	[STO] 4	054– 44 4
[GTO] 1	035– 22 1		

REGISTERS			R_I: Unused
R_0: Unused	R_1: f	R_2: L	R_3: C
R_4: X	R_5—$R_{.9}$ Unused		

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYSTROKES	OUTPUT DATA/UNITS
1	Clear User mode, then key in the			
	program.			
2	Set User mode.		[f] [USER]	
3	Input two known values. Input 0			
	if unknown.	*f*, hertz	[A]	0
		L, henrys	[B]	0
		C, farads	[C]	0
4	Calculate *X* and the unknown.		[D]	*X*, ohms
			[R/S]	*f*, hertz
			[R/S]	*L*, henrys
			[R/S]	*C*, farads

Example 1: $L = 0.1\ mh$, $C = 0.2\ \mu f$; calculate f and X.

Keystrokes	Display	
Set User mode.		
[f] [FIX] 4		
.1 [EEX] [CHS] 3 [B]	0.0000	
.2 [EEX] [CHS] 6 [C]	0.0000	
[A]	0.0000	
[D]	22.3607	X
[R/S]	35,588.1272	f
[R/S]	0.0001	L
[R/S]	2.0000 -07	C

Example 2: $f = 100\ h$, $C = 0.1\ \mu f$; calculate X and L.

Keystrokes	Display	
100 [A]	0.0000	
[B]	0.0000	
.1 [EEX] [CHS] 6 [C]	0.0000	
[D]	15,915.4943	X
[R/S]	100.0000	f
[R/S]	25.3303	L
[R/S]	1.0000 -07	C

Ohm's Law

This program provides interchangeable solutions for the Ohm's Law relations. Specifically, one may solve for:

- Resistance and power dissipation given voltage and current.
- Current and power dissipation given voltage and resistance.
- Voltage and power dissipation given current and resistance.
- Current and resistance given voltage and power dissipation.
- Voltage and resistance given current and power dissipation.
- Voltage and current given resistance and power dissipation.

Equations:

$$V = IR$$

$$P = I^2R$$

where

V = voltage (volts)

I = current (amps)

R = resistance (ohms)

P = power dissipation (watts)

KEYSTROKES	DISPLAY	KEYSTROKES	DISPLAY
[f] CLEAR [PRGM]	000-	[g] [CLx]	011- 43 35
[f] [LBL] [A]	001-42,21,11	[R/S]	012- 31
[STO] 0	002- 44 0	[f] [LBL] [D]	013-42,21,14
[g] [CLx]	003- 43 35	[STO] 3	014- 44 3
[R/S]	004- 31	[g] [CLx]	015- 43 35
[f] [LBL] [B]	005-42,21,12	[R/S]	016- 31
[STO] 1	006- 44 1	[f] [LBL] [E]	017-42,21,15
[g] [CLx]	007- 43 35	[RCL] 0	018- 45 0
[R/S]	008- 31	[g] [x≠0]	019- 43 30
[f] [LBL] [C]	009-42,21,13	[GTO] 1	020- 22 1
[STO] 2	010- 44 2	[RCL] 1	021- 45 1

KEYSTROKES	DISPLAY	KEYSTROKES	DISPLAY
[g] [x≠0]	022– 43 30	[f] [LBL] 3	051–42,21, 3
[GTO] 2	023– 22 2	[RCL] 1	052– 45 1
[RCL] 3	024– 45 3	[RCL] 2	053– 45 2
[RCL] 2	025– 45 2	[×]	054– 20
[÷]	026– 10	[STO] 0	055– 44 0
[√x]	027– 11	[GTO] 5	056– 22 5
[STO] 1	028– 44 1	[f] [LBL] 4	057–42,21, 4
[GTO] 3	029– 22 3	[RCL] 0	058– 45 0
[f] [LBL] 1	030–42,21, 1	[RCL] 1	059– 45 1
[RCL] 1	031– 45 1	[÷]	060– 10
[g] [x≠0]	032– 43 30	[STO] 2	061– 44 2
[GTO] 4	033– 22 4	[GTO] 5	062– 22 5
[RCL] 2	034– 45 2	[f] [LBL] 0	063–42,21, 0
[g] [x≠0]	035– 43 30	[RCL] 0	064– 45 0
[GTO] 0	036– 22 0	[RCL] 2	065– 45 2
[RCL] 3	037– 45 3	[÷]	066– 10
[RCL] 0	038– 45 0	[STO] 1	067– 44 1
[÷]	039– 10	[f] [LBL] 5	068–42,21, 5
[STO] 1	040– 44 1	[RCL] 1	069– 45 1
[GTO] 4	041– 22 4	[RCL] 0	070– 45 0
[f] [LBL] 2	042–42,21, 2	[×]	071– 20
[RCL] 2	043– 45 2	[STO] 3	072– 44 3
[g] [x≠0]	044– 43 30	[RCL] 0	073– 45 0
[GTO] 3	045– 22 3	[R/S]	074– 31
[RCL] 3	046– 45 3	[RCL] 1	075– 45 1
[RCL] 1	047– 45 1	[R/S]	076– 31
[g] [x²]	048– 43 11	[RCL] 2	077– 45 2
[÷]	049– 10	[R/S]	078– 31
[STO] 2	050– 44 2	[RCL] 3	079– 45 3

REGISTERS			R_I: Unused
R_0: *V*	R_1: *I*	R_2: *R*	R_3: *P*
R_4—R_6 Unused			

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYSTROKES	OUTPUT DATA/UNITS
1	Clear User mode, then key in the			
	program.			
2	Set User mode.		[f] [USER]	
3	Input two known values. Input 0			
	if unknown.	*V*, volts	[A]	0
		I, amps	[B]	0
		R, ohms	[C]	0
		P, watts	[D]	0
4	Calculate the two unknowns.		[E]	*V*, volts
			[R/S]	*I*, amps
			[R/S]	*R*, ohms
			[R/S]	*P*, watts

Example 1: $V = 43.2$ volts, $I = 0.1$ amps; calculate R and P.

Keystrokes	**Display**	
Set User mode.		
[f] [FIX] 4		
43.2 [A]	**0.0000**	
.1 [B]	**0.0000**	
[C]	**0.0000**	
[D]	**0.0000**	
[E]	**43.2000**	V
[R/S]	**0.1000**	I
[R/S]	**432.0000**	R
[R/S]	**4.3200**	P

Example 2: $V = 43.2$ volts, $R = 430$ ohms; calculate I and P.

Note that since V is the same in the second example as the first, its value need not be re-input.

Keystrokes	**Display**	
0 [B]	0.0000	
430 [C]	0.0000	
[D]	0.0000	
[E]	43.2000	V
[R/S]	0.1005	I
[R/S]	430.0000	R
[R/S]	4.3401	P

Impedance of a Ladder Network

This program computes the input impedance of an arbitrary ladder network. Elements are added one at a time starting from the right. The first element must be parallel.

Suppose we have a network whose input admittance is Y_{in}. Adding a shunt R, L or C, the input admittance becomes

$$Y_{new} = \begin{cases} Y_{in} + \left(\dfrac{1}{R_p} + j0\right) \\ Y_{in} + \left(0 - j\dfrac{1}{\omega L_p}\right) \\ Y_{in} + (0 + j\omega C_p) \end{cases}$$

Adding a series R, L or C, we have

$$Y_{new} = \begin{cases} \left(\dfrac{1}{Y_{in}} + (R_s + j0)\right)^{-1} \\ \left(\dfrac{1}{Y_{in}} + (0 + j\omega L_s)\right)^{-1} \\ \left(\dfrac{1}{Y_{in}} + \left(0 - j\dfrac{1}{\omega C_s}\right)\right)^{-1} \end{cases}$$

where $Z = \dfrac{1}{Y}$, $\omega = 2\pi f$

The program converts this admittance to an impedance for display.

> **Note:** An erroneous entry may be corrected by entering the negative of the incorrect value.

KEYSTROKES	DISPLAY	KEYSTROKES	DISPLAY
[f] CLEAR [PRGM]	000-	[GTO] 6	021- 22 6
[f] [LBL] [A]	001-42,21,11	[f] [LBL] 5	022-42,21, 5
[f] CLEAR [REG]	002- 42 34	[RCL] 0	023- 45 0
2	003- 2	[×]	024- 20
[×]	004- 20	[1/x]	025- 15
[f] [π]	005- 42 16	[CHS]	026- 16
[×]	006- 20	0	027- 0
[STO] 0	007- 44 0	[x≷y]	028- 34
[g] [CF] 0	008-43, 5, 0	[GTO] 9	029- 22 9
[g] [RTN]	009- 43 32	[f] [LBL] [E]	030-42,21,15
[f] [LBL] [B]	010-42,21,12	[g] [F?] 0	031-43, 6, 0
[g] [SF] 0	011-43, 4, 0	[GTO] 5	032- 22 5
[g] [RTN]	012- 43 32	[f] [LBL] 6	033-42,21, 6
[f] [LBL] [C]	013-43,21,13	[RCL] 0	034- 45 0
[1/x]	014- 15	[×]	035- 20
[g] [F?] 0	015-43, 6, 0	0	036- 0
[1/x]	016- 15	[x≷y]	037- 34
0	017- 0	[f] [LBL] 9	038-42,21, 9
[GTO] 9	018- 22 9	[RCL] 2	039- 45 2
[f] [LBL] [D]	019-42,21,14	[RCL] 1	040- 45 1
[g] [F?] 0	020-43, 6, 0	[g] [F?] 0	041-43, 6, 0

KEYSTROKES	DISPLAY	KEYSTROKES	DISPLAY
[GSB] 0	042– 32 0	[GSB] 0	054– 32 0
[g] [R↑]	043– 43 33	[g] [CF] 0	055–43, 5, 0
[+]	044– 40	[g] [→P]	056– 43 26
[R↓]	045– 33	[g] [RTN]	057– 43 32
[+]	046– 40	[f] [LBL] 0	058–42,21, 0
[g] [R↑]	047– 43 33	[g] [→P]	059– 43 26
[g] [F?] 0	048–43, 6, 0	[1/x]	060– 15
[GSB] 0	049– 32 0	[x≷y]	061– 34
[STO] 1	050– 44 1	[CHS]	062– 16
[x≷y]	051– 34	[x≷y]	063– 34
[STO] 2	052– 44 2	[f] [→R]	064– 42 26
[x≷y]	053– 34		

REGISTERS			R_I: Unused
R_0: ω	R_1: $Re[Y_{in}]$	R_2: $Im[Y_{in}]$	R_3—$R_{.8}$ Unused

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYSTROKES	OUTPUT DATA/UNITS
1	Clear User mode, then key in the			
	program.			
2	Set User mode.		[f] [USER]	
3	Set desired trig. mode.			
4	Initialize.	*f*, hertz	[A]	ω, rad/sec
5	Input elements (first must be			
	parallel).			
	A parallel element:			
		R, ohms	[C]	$\|Z_{in}\|$, ohms
	or,	*L*, henrys	[D]	$\|Z_{in}\|$, ohms
	or,	*C*, farads	[E]	$\|Z_{in}\|$, ohms

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYSTROKES	OUTPUT DATA/UNITS
	A series element:			
		R, ohms	[B] [C]	$\|Z_{in}\|$, ohms
	or,	L, henrys	[B] [D]	$\|Z_{in}\|$, ohms
	or,	C, farads	[B] [E]	$\|Z_{in}\|$, ohms
6	Repeat step 5 for next element.			
7	(Optional) Whenever $\|Z_{in}\|$ is			
	displayed the angle can be found			
	in the Y-register.		[x≷y]	$\measuredangle Z_{in}$

Example:

$f = 4\text{MHz}$



Keystrokes	**Display**	
Set User mode.		
[g] [DEG]		
[f] [FIX] 4		
4 [EEX] 6 [A]	**25,132,741.23**	
50 [C]	**50.0000**	$\|Z_{in}\|$, ohms
2400 [EEX] [CHS] 12 [E]	**15.7362**	$\|Z_{in}\|$, ohms
[x≷y]	**−71.6559**	$\measuredangle Z_{in}$, deg.
2.56 [EEX] [CHS]		
6 [B] [D]	**49.6509**	$\|Z_{in}\|$, ohms
[x≷y]	**84.2754**	$\measuredangle Z_{in}$, deg.
796 [EEX] [CHS] 12 [E]	**497.6942**	$\|Z_{in}\|$, ohms
[x≷y]	**0.9840**	$\measuredangle Z_{in}$, deg.

Smith Chart Conversions

The distance between a point on a Smith Chart and its center may be measured by a number of parameters. This program allows conversion between some of the most commonly used parameters: standing wave ratio, reflection coefficient, and return loss. One may also convert between impedance and reflection coefficient.

Parameters:

σ = voltage standing wave ratio $= \dfrac{1+\rho}{1-\rho}$

SWR = standing wave ratio expressed in decibels

ρ = reflection coefficient

$R.L.$ = return loss

These parameters are related as follows:

$$SWR = 20 \log \sigma$$

$$R.L. = 20 \log \frac{1}{\rho}$$

$$\sigma = \frac{1+\rho}{1-\rho}$$



These relationships are perhaps more clearly seen in this sketch:



For a system having characteristic impedance Z_0, the impedance and reflection coefficient are related by

$$\Gamma = \rho \measuredangle \phi = \frac{\dfrac{\boldsymbol{Z}}{Z_0} - 1}{\dfrac{\boldsymbol{Z}}{Z_0} + 1}$$

and

$$\boldsymbol{Z} = Z \measuredangle \theta = Z_0 \frac{1 + \Gamma}{1 - \Gamma}$$

where

Γ = complex reflection coefficient

$\rho = |\Gamma|$

$\phi = \measuredangle \Gamma$

$\mathbf{Z}$ = impedance

$Z = |\mathbf{Z}|$

$\theta = \measuredangle \mathbf{Z}$

KEYSTROKES	DISPLAY	KEYSTROKES	DISPLAY
[f] CLEAR [PRGM]	000-	[f] [LBL] 4	029-42,21, 4
[f] [LBL] 3	001-42,21, 3	[STO] 1	030- 44 1
[1/x]	002- 15	[g] [RTN]	031- 43 32
[f] [LBL] 1	003-42,21, 1	[f] [LBL] [D]	032-42,21,14
[g] [LOG]	004- 43 13	1	033- 1
2	005- 2	[GSB] 7	034- 32 7
0	006- 0	[RCL] 1	035- 45 1
[×]	007- 20	[CHS]	036- 16
[g] [RTN]	008- 43 32	[×]	037- 20
[f] [LBL] [C]	009-42,21,13	[f] [→R]	038- 42 26
[CHS]	010- 16	[g] [→P]	039- 43 26
[f] [LBL] [A]	011-42,21,11	[GTO] 9	040- 22 9
2	012- 2	[f] [LBL] [E]	041-42,21,15
0	013- 0	[RCL] 1	042- 45 1
[÷]	014- 10	[CHS]	043- 16
[10^x]	015- 13	[GSB] 7	044- 32 7
[g] [RTN]	016- 43 32	[f] [LBL] 9	045-42,21, 9
[f] [LBL] 2	017-42,21, 2	[$x \gtrless y$]	046- 34
[1/x]	018- 15	[R/S]	047- 31
[CHS]	019- 16	[$x \gtrless y$]	048- 34
[f] [LBL] [B]	020-42,21,12	[R/S]	049- 31
1	021- 1	[g] [RTN]	050- 43 32
[$x \gtrless y$]	022- 34	[f] [LBL] 7	051-42,21, 7
[+]	023- 40	[ENTER]	052- 36
1	024- 1	[R↓]	053- 33
[g] [LSTx]	025- 43 36	[R↓]	054- 33
[−]	026- 30	[f] [→R]	055- 42 26
[÷]	027- 10	[g] [R↑]	056- 43 33
[g] [RTN]	028- 43 32	[−]	057- 30

KEYSTROKES	DISPLAY	KEYSTROKES	DISPLAY
[STO] 0	058– 44 0	[g] [→P]	066– 43 26
[R↓]	059– 33	[R↓]	067– 33
[R↓]	060– 33	[x≷y]	068– 34
[+]	061– 40	[g] [R↑]	069– 43 33
[+]	062– 40	[÷]	070– 10
[g] [→P]	063– 43 26	[R↓]	071– 33
[g] [R↑]	064– 43 33	[−]	072– 30
[RCL] 0	065– 45 0	[g] [R↑]	073– 43 33

REGISTERS			R_I: Unused
R_0: Used	R_1: Z_0	R_2—R_7 Unused	

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYSTROKES	OUTPUT DATA/UNITS
1	Clear User mode, then key in the			
	program.			
2	Set User mode.		[f] [USER]	
3	To convert among σ, *SWR*, ρ			
	and *R.L.*:			
	● $\sigma \rightarrow$ *SWR*	σ	[GSB] 1	*SWR*
	● *SWR* $\rightarrow \sigma$	*SWR*	[A]	σ
	● $\sigma \rightarrow \rho$	σ	[GSB] 2	ρ
	● $\rho \rightarrow \sigma$	ρ	[B]	σ
	● $\rho \rightarrow$ *R.L.*	ρ	[GSB] 3	*R.L.*
	● *R.L.* $\rightarrow \rho$	*R.L.*	[C]	ρ
4	Store characteristic impedance.	Z_0	[GSB] 4	Z_0

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYSTROKES	OUTPUT DATA/UNITS
5	Convert between **Z** and Γ.			
	Z → Γ	θ	[ENTER]	
		Z	[E]	φ
			[R/S]	ρ
	Γ → **Z**	φ	[ENTER]	
		ρ	[D]	θ
			[R/S]	Z

Example 1: Convert a 6 dB *SWR* to σ.

Keystrokes	**Display**	
Set User mode.		
[f] [FIX] 2		
6 [A]	**2.00**	σ

Example 2: Convert a 7 dB return loss to *SWR*.

Keystrokes	**Display**	
7 [C]	**0.45**	ρ
[B]	**2.61**	σ
[GSB] 1	**8.35**	*SWR*

Example 3: A 50 Ω system is terminated with an impedance of 62 ∡ 37°. What is the reflection coefficient?

Keystrokes	**Display**	
50 [GSB] 4	**50.00**	
37 [ENTER] 62 [E]	**70.19**	φ
[R/S]	**0.35**	ρ

Example 4: A reflection coefficient of .5 ∡ 7° is observed in a 72 Ω system. What is the impedance?

Keystrokes	**Display**	
72 [GSB] 4	**72.00**	
7 [ENTER] .5 [D]	**9.23**	θ
[R/S]	**212.50**	Z

Mechanical Engineering

Ideal Gas Equation of State

Many gases obey the ideal gas laws quite closely at reasonable temperatures and pressures. This program calculates any one of the four variables when the other three are known.

Equation:

$$PV = nRT$$

where:

P is the absolute pressure.

V is the volume.

n is the number of moles present.

R is the Universal Gas Constant.

T is the absolute temperature.

Values of the Universal Gas Constant

Value of R	Units of R	Units of P	Units of V	Units of T
8.314	N-m/g mole-K	N/M^2	m^3/g mole	K
83.14	cm^3-bar/g mole-K	bar	cm^3/g mole	K
82.05	cm^3-atm/g mole-K	atm	cm^3/g mole	K
0.08205	-atm/g mole-K	atm	/g mole	K
0.7302	atm-ft^3/lbm mole-°R	atm	ft^3/lbm mole	°R
10.73	psi-ft^3/lbm mole-°R	psi	ft^3/lbm mole	°R
1545	psi-ft^3/lbm mole-°R	psf	ft^3/lbm mole	°R

Remarks:

- At low temperatures or high pressures, the ideal gas law does not represent the behavior of real gases.
- The value of R used must be compatible with the units of P, V and T.

KEYSTROKES	DISPLAY	KEYSTROKES	DISPLAY
f CLEAR PRGM	000–	f LBL 1	029–42,21, 1
f LBL A	001–42,21,11	GSB 8	030– 32 8
STO 0	002– 44 0	RCL 2	031– 45 2
R/S	003– 31	÷	032– 10
f LBL B	004–42,21,12	g RTN	033– 43 32
STO 1	005– 44 1	f LBL 2	034–42,21, 2
R/S	006– 31	GSB 8	035– 32 8
f LBL C	007–42,21,13	RCL 1	036– 45 1
STO 2	008– 44 2	÷	037– 10
R/S	009– 31	g RTN	038– 43 32
f LBL D	010–42,21,14	f LBL 3	039–42,21, 3
STO 3	011– 44 3	GSB 9	040– 32 9
R/S	012– 31	RCL 5	041– 45 5
f LBL E	013–42,21,15	÷	042– 10
STO 4	014– 44 4	g RTN	043– 43 32
R/S	015– 31	f LBL 4	044–42,21, 4
f LBL 5	016–42,21, 5	GSB 9	045– 32 9
·	017– 48	RCL 3	046– 45 3
0	018– 0	÷	047– 10
0	019– 0	STO 4	048– 44 4
4	020– 4	g RTN	049– 43 32
STO I	021– 44 25	f LBL 8	050–42,21, 8
f LBL 6	022–42,21, 6	RCL 3	051– 45 3
RCL (i)	023– 45 24	RCL 0	052– 45 0
g x=0	024– 43 40	×	053– 20
GTO I	025– 22 25	RCL 4	054– 45 4
f ISG	026– 42 6	×	055– 20
GTO 6	027– 22 6	g RTN	056– 43 32
g RTN	028– 43 32	f LBL 9	057–42,21, 9

KEYSTROKES	DISPLAY	KEYSTROKES	DISPLAY
[RCL] 1	058– 45 1	[RCL] 0	061– 45 0
[RCL] 2	059– 45 2	[÷]	062– 10
[×]	060– 20		

REGISTERS			R_I: Unused
R_0: *R*	R_1: *P*	R_2: *V*	R_3: *n*
R_4: *T*	R_5—$R_{.9}$ Unused		

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYSTROKES	OUTPUT DATA/UNITS
1	Clear User mode, then key in the			
	program.			
2	Set User mode.		[f] [USER]	
3	Initialize.		[f] CLEAR [REG]	
4	Input *R* in appropriate units.	*R*	[A]	*R*
5	Input any three knowns:			
	● pressure	*P*	[B]	*P*
	● volume	*V*	[C]	*V*
	● moles	*n*	[D]	*n*
	● temperature	*T*	[E]	*T*
6	Calculate the unknown.			
6a	If *T* was the last input.		[R/S]	Unknown
6b	If *T* was not the last input.		[GSB] 5	Unknown
7	For the same unknown and different			
	known values, go to step 4.			
8	For new unknowns, go to step 3.			

Example 1: 0.63 moles of air are enclosed in 25,000 cm^3 of space at 1,200 K. What is the pressure in bars? In atmospheres?

Keystrokes	Display	
Set User mode.		
[f] [FIX] 4		
[f] CLEAR [REG]		
83.14 [A]	**83.1400**	
25000 [C]	**25,000.0000**	
.63 [D]	**0.6300**	
1200 [E]	**1200.0000**	
[R/S]	**2.5142**	*P*, bars
82.05 [A]	**82.0500**	
[GSB] 5	**2.4812**	*P*, atm

Example 2: What is the specific volume (ft^3/lbm) of a gas at atmospheric pressure and a temperature of 513°R? The molecular weight is 29 lbm/lbm-mole.

Keystrokes	Display	
[f] CLEAR [REG]		
.7302 [A]	**0.7302**	
1 [B]	**1.0000**	
29 [f] [1/x] [D]	**0.0345**	
513 [E]	**513.0000**	
[R/S]	**12.9170**	*V*, ft^3/lbm

What is its density?

[f] [1/x]	**0.0774**	ρ, lbm/ft^3

What is the density at 1.32 atm and 555°R?

1.32 [B]	**1.3200**	
555 [E]	**555.0000**	
[R/S]	**10.5868**	*V*, ft^3/lbm
[f] [1/x]	**0.0945**	ρ, lbm/ft^3

Conduit Flow

This program solves for the average velocity or the pressure drop for viscous, incompressible flow in conduits.

Equations:

$$v^2 = \frac{\Delta P/\rho}{2\left(f\frac{L}{D} + \frac{K_T}{4}\right)}$$

For laminar flow ($Re < 2300$)

$$f = 16/Re$$

For turbulent flow ($Re > 2300$)

$$\frac{1}{\sqrt{f}} = 1.737 \ln \frac{D}{\epsilon} + 2.28 - 1.737 \ln \left(4.67 \frac{D}{\epsilon Re \sqrt{f}} + 1\right)$$

is solved by Newton's method.

$$\frac{1}{\sqrt{f_0}} = 1.737 \ln \frac{D}{\epsilon} + 2.28$$

is used as an initial guess in the iteration.

where:

Re is the Reynolds number, defined as $\rho D v/\mu$;

D is the pipe diameter;

ϵ is the dimension of irregularities in the conduit surface (see Table 2);

f is the Fanning friction factor for conduit flow;

ΔP is the pressure drop along the conduit;

ρ is the density of the fluid;

μ is the dynamic viscosity of the fluid;

υ is the kinematic viscosity of the fluid (μ/ρ);

L is the conduit length;

v is the average fluid velocity;

K_T is the total of the applicable fitting coefficients in Table 1.

Table 1
Fitting Coefficients

Fitting	K
Globe valve, wide open	7.5 − 10
Angle valve, wide open	3.8
Gate valve, wide open	0.15 − 0.19
Gate valve, ¾ open	0.85
Gate valve, ½ open	4.4
Gate valve, ¼ open	20
90° elbow	0.4 − 0.9
Standard 45° elbow	0.35 − 0.42
Tee, through side outlet	1.5
Tee, straight through	0.4
180° bend	1.6
Entrance to circular pipe	0.25 − 0.50
Sudden expansion	$(1 - A_{up}/A_{dn})^{2}$*
Acceleration from $v = 0$ to $v = v_{entrance}$	1.0

Table 2
Surface Irregularities

Material	ϵ (Feet)	ϵ (Meters)
Drawn or Smooth Tubing	5.0×10^{-6}	1.5×10^{-6}
Commercial Steel or Wrought Iron	1.5×10^{-4}	4.6×10^{-5}
Asphalted Cast Iron	4.0×10^{-4}	1.2×10^{-4}
Galvanized Iron	5.0×10^{-4}	1.5×10^{-4}
Cast Iron	8.3×10^{-4}	2.5×10^{-4}
Wood Stave	6.0×10^{-4} to 3.0×10^{-3}	1.8×10^{-4} to 9.1×10^{-4}
Concrete	1.0×10^{-3} to 1.0×10^{-2}	3.0×10^{-4} to 3.0×10^{-3}
Riveted Steel	3.0×10^{-3} to 3.0×10^{-2}	9.1×10^{-4} to 9.1×10^{-3}

* A_{up} is the upstream area and A_{dn} is the downstream area.

Remarks:

- The correlation gives meaningless results in the region $2300 < Re < 4000$.
- Any consistent set of units may be used.
- If the conduit is not circular, an equivalent diameter may be calculated using the following formula:

$$D_{eq} = 4 \frac{\text{cross sectional area}}{\text{wetted parameter}}$$

Reference: Welty, Wicks, Wilson, *Fundamentals of Momentum, Heat and Mass Transfer*, John Wiley and Sons, Inc., 1969.

KEYSTROKES	DISPLAY	KEYSTROKES	DISPLAY
[f] CLEAR [PRGM]	000–	[RCL] 8	019– 45 8
[f] [LBL] [B]	001–42,21,12	[×]	020– 20
[g] [SF] 0	002–43, 4, 0	2	021– 2
[GSB] 9	003– 32 9	[·]	022– 48
[f] [LBL] 3	004–42,21, 3	2	023– 2
[g] [RND]	005– 43 34	8	024– 8
[STO] [·] 1	006– 44 .1	[+]	025– 40
[GSB] 8	007– 32 8	[STO] [·] 0	026– 44 .0
[g] [RND]	008– 43 34	[STO] 9	027– 44 9
[RCL] [·] 1	009– 45 .1	[g] [F?] 0	028–43, 6, 0
[$x \gtrless y$]	010– 34	[GTO] 7	029– 22 7
[f] [$x \neq y$]	011– 42 30	[f] [LBL] 8	030–42,21, 8
[GTO] 3	012– 22 3	[RCL] 6	031– 45 6
[g] [RTN]	013– 43 32	[RCL] 4	032– 45 4
[f] [LBL] 9	014–42,21, 9	[×]	033– 20
[RCL] 4	015– 45 4	[RCL] 0	034– 45 0
[RCL] 2	016– 45 2	[÷]	035– 10
[÷]	017– 10	[STO] [I]	036– 44 25
[g] [LN]	018– 43 12	2	037– 2

KEYSTROKES	DISPLAY	KEYSTROKES	DISPLAY
3	038– 3	[×]	067– 20
0	039– 0	[RCL] [·] 0	068– 45 .0
0	040– 0	[−]	069– 30
[f] [x≤y]	041– 42 10	[RCL] 9	070– 45 9
[GTO] 2	042– 22 2	[+]	071– 40
[R↓]	043– 33	[x≷y]	072– 34
[√x]	044– 11	[1/x]	073– 15
4	045– 4	1	074– 1
[÷]	046– 10	[−]	075– 30
[STO] 9	047– 44 9	[RCL] 8	076– 45 8
[GTO] 7	048– 22 7	[×]	077– 20
[f] [LBL] 2	049–42,21, 2	[RCL] 9	078– 45 9
4	050– 4	[÷]	079– 10
[·]	051– 48	1	080– 1
6	052– 6	[−]	081– 30
7	053– 7	[÷]	082– 10
[RCL] 4	054– 45 4	[STO] [+] 9	083–44,40, 9
[×]	055– 20	[RCL] 9	084– 45 9
[RCL] 2	056– 45 2	[÷]	085– 10
[÷]	057– 10	[g] [ABS]	086– 43 16
[RCL] [I]	058– 45 25	[EEX]	087– 26
[÷]	059– 10	3	088– 3
[RCL] 9	060– 45 9	[CHS]	089– 16
[×]	061– 20	[f] [x≤y]	090– 42 10
1	062– 1	[GTO] 2	091– 22 2
[+]	063– 40	[f] [LBL] 7	092–42,21, 7
[ENTER]	064– 36	[RCL] 7	093– 45 7
[g] [LN]	065– 43 12	[RCL] 1	094– 45 1
[RCL] 8	066– 45 8	[÷]	095– 10

KEYSTROKES	DISPLAY	KEYSTROKES	DISPLAY
[RCL] 3	096– 45 3	[STO] 6	108– 44 6
[RCL] 4	097– 45 4	[g] [RTN]	109– 43 32
[RCL] 9	098– 45 9	[f] [LBL] [A]	110–42,21,11
[g] [x^2]	099– 43 11	[g] [CF] 0	111–43, 5, 0
[×]	100– 20	[GSB] 9	112– 32 9
[÷]	101– 10	[g] [LSTx]	113– 43 36
[RCL] 5	102– 45 5	[RCL] 6	114– 45 6
[+]	103– 40	[g] [x^2]	115– 43 11
[÷]	104– 10	[×]	116– 20
[g] [F?] 0	105–43, 6, 0	[RCL] 1	117– 45 1
[$\sqrt{x}$]	106– 11	[×]	118– 20
[g] [F?] 0	107–43, 6, 0	[STO] 7	119– 44 7

REGISTERS			R_I: *Re*
R_0: ν	R_1: ρ	R_2: ϵ	R_3: $2L$
R_4: D	R_5: $K_T/2$	R_6: v_{input}	R_7: ΔP_{input}
R_8: 1.737	R_9: $1/\sqrt{f}$	$R_{.0}$: $1/\sqrt{f_0}$	$R_{.1}$: v_{calc}

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYSTROKES	OUTPUT DATA/UNITS
1	Clear User mode, then key in the			
	program.			
2	Set User mode.		[f] [USER]	
3	Store the variables.*			
	Kinematic viscosity	ν	[STO] 0	
	or			
	Dynamic viscosity	μ	[ENTER]	
		ρ	[÷] [STO] 0	
	Density	ρ	[STO] 1	
	Surface irregularity	ϵ	[STO] 2	
	Twice conduit length	$2L$	[STO] 3	
	Equivalent passage diameter	D	[STO] 4	
	Half the total fitting coefficient	$K_T/2$	[STO] 5	
	Fluid velocity	v	[STO] 6	
	or pressure drop	ΔP	[STO] 7	
	And a constant.	1.737	[STO] 8	
4	Calculate either ΔP		[A]	ΔP
	or v.		[B]	v
5	Calculate *Re*.		[RCL] [I]	*Re*
6	Calculate *f*.		[RCL] 9	
			[f] [1/x]	
			[g] [x^2]	*f*
7	For a new case, go to step 3.			

*These values are not altered by the program and don't have to be reentered after each run.

Example: A heat exchanger has twenty 3-meter tube passes (60 m of pipe) with 180 degree bends connecting each pair of tubes (from Table 1, $K_T = 10 \times 1.6$). The fluid is water ($\nu = 9.3 \times 10^{-7}$ m^2/s, $\rho = 10^3$ kg/m^3). The surface roughness is 3×10^{-4} m and the diameter is 2.54×10^{-2} m. If the fluid velocity is 3.05 m/s, what is the pressure loss? What is the Reynolds number? What is the Fanning friction factor?

Keystrokes	Display		
Set User mode.			
[f] [ENG] 2			
9.3 [EEX] [CHS]			
7 [STO] 0	**930.**	**-09**	
[EEX] 3 [STO] 1	**1.00**	**03**	
3 [EEX] [CHS] 4 [STO] 2	**300.**	**-06**	
60 [ENTER] 2 [×] [STO] 3	**120.**	**00**	
2.54 [EEX] [CHS]			
2 [STO] 4	**25.4**	**-03**	
16 [ENTER] 2 [÷] [STO] 5	**8.00**	**00**	
3.05 [STO] 6	**3.05**	**00**	
1.737 [STO] 8	**1.74**	**00**	
[A]	**522.**	**03**	ΔP, N/m^2
[RCL] [I]	**83.3**	**03**	*Re*
[RCL] 9 [f] [1/x] [g] [x^2]	**10.2**	**-03**	*f*

Simply Supported Beams

This program calculates deflection, slope, moment and shear at any specified point along a simply supported beam of uniform cross section. Distributed loads, point loads, applied moments or combinations of all three may be modeled. By using the principle of superposition, complicated beams with multiple point loads, multiple distributed loads, and multiple applied moments can be analyzed.



Equations:

$y = y_1 + y_2 + y_3$ (total deflection)

$$y_1 = \frac{P(\ell - a)x}{6EI\ell}[x^2 + (\ell - a)^2 - \ell^2]^*$$ (deflection due to point load)

$$y_2 = \frac{-Wx}{24EI}[\ell^3 + x^2(x - 2\ell)]$$ (deflection due to distributed load)

$$y_3 = \frac{-Mx}{EI\ell}\left[c\ell - \frac{x^2}{6} - \frac{\ell^2}{3} - \frac{c^2}{2}\right]^\dagger$$ (deflection due to applied moment)

$\theta = \theta_1 + \theta_2 + \theta_3$ (total slope)

$$\theta_1 = \frac{P(\ell - a)}{6EI\ell}[3x^2 + (\ell - a)^2 - \ell^2]^*$$ (slope due to point load)

$$\theta_2 = -\frac{W}{24EI}[\ell^3 + x^2(4x - 6\ell)]$$ (slope due to distributed load)

$$\theta_3 = \frac{-M}{EI}\left[c - \frac{x^2}{2\ell} - \frac{\ell}{3} - \frac{c^2}{2\ell}\right]^\dagger$$ (slope due to applied moment)

---

* If x is greater than a, $(\ell - a)$ is replaced by $-a$ and x is replaced by $(x - \ell)$.

† If x is greater than c, x is replaced by $(x - \ell)$ and c is replaced by $(\ell - c)$.

$M_x = M_{x1} + M_{x2} + M_{x3}$ (total moment)

$$M_{x1} = \frac{P(\ell - a)x}{\ell}^{*} \quad \text{(moment due to point load)}$$

$$M_{x2} = -\frac{Wx}{2}[x - \ell] \quad \text{(moment due to distributed load)}$$

$$M_{x3} = \frac{Mx}{\ell}^{\dagger} \quad \text{(moment due to applied moment)}$$

$V = V_1 + V_2 + V_3$ (total shear)

$$V_1 = \frac{P(\ell - a)}{\ell}^{*} \quad \text{(shear due to point load)}$$

$$V_2 = W\left(\frac{\ell}{2} - x\right) \quad \text{(shear due to distributed load)}$$

$$V_3 = \frac{M}{\ell} \quad \text{(shear due to applied moment)}$$

where:

y is the deflection at a distance x from the left support;

θ is the slope (change in y per change in x) at x;

M_x is the moment at x;

V is the shear at x;

I is the moment of inertia of the beam;

E is the modulus of elasticity of the beam;

ℓ is the length of the beam;

P is a concentrated load;

W is the uniformly distributed load with dimensions of force per unit length;

---

* If x is greater than a, $(\ell - a)$ is replaced by $-a$ and x is replaced by $(x - \ell)$.

† If x is greater than c, x is replaced by $(x - \ell)$ and c is replaced by $(\ell - c)$.

M is the applied moment;

a is the distance from the left support to the point load;

c is the distance from the left support to the applied moment.

Remarks:

- Any consistent set of units may be used.
- Deflections must not significantly alter the geometry of the problem. Beams must be of constant cross section for deflection and slope equations to be valid. Stresses must be in the elastic region.

SIGN CONVENTIONS FOR BEAMS

NAME	VARIABLE	SENSE	SIGN
DEFLECTION	y	↑	+
SLOPE	θ	↑	+
INTERNAL MOMENT	M_x	↶□↷	+
SHEAR	V	↑□↓	+
EXTERNAL FORCE OR LOAD	P or W	↓	+
EXTERNAL MOMENT	M	↻	+

KEYSTROKES	DISPLAY	KEYSTROKES	DISPLAY
[f] CLEAR [PRGM]	000–	[g] [F?] 0	009–43, 6, 0
[f] [LBL] [A]	001–42,21,11	[×]	010– 20
[STO] 4	002– 44 4	[RCL] 1	011– 45 1
[RCL] 1	003– 45 1	2	012– 2
3	004– 3	[×]	013– 20
[y^x]	005– 14	[GSB] 3	014– 32 3
[RCL] 4	006– 45 4	[–]	015– 30
[g] [F?] 0	007–43, 6, 0	[RCL] 4	016– 45 4
4	008– 4	[g] [x^2]	017– 43 11

KEYSTROKES	DISPLAY	KEYSTROKES	DISPLAY
[×]	018– 20	[RCL] 1	047– 45 1
[+]	019– 40	[÷]	048– 10
[R/S]	020– 31	6	049– 6
[×]	021– 20	[÷]	050– 10
2	022– 2	[GSB] 3	051– 32 3
4	023– 4	[RCL] 1	052– 45 1
[÷]	024– 10	3	053– 3
[RCL] 4	025– 45 4	[÷]	054– 10
[$x \gtrless y$]	026– 34	[+]	055– 40
[×]	027– 20	[RCL] [I]	056– 45 25
[g] [F?] 0	028–43, 6, 0	[g] [x^2]	057– 43 11
[g] [LST x]	029– 43 36	2	058– 2
[CHS]	030– 16	[÷]	059– 10
[GSB] 1	031– 32 1	[RCL] 1	060– 45 1
[RCL] 5	032– 45 5	[÷]	061– 10
[g] [x^2]	033– 43 11	[+]	062– 40
[GSB] 3	034– 32 3	[RCL] [I]	063– 45 25
[RCL] [I]	035– 45 25	[−]	064– 30
[g] [x^2]	036– 43 11	[×]	065– 20
[+]	037– 40	[RCL] 1	066– 45 1
[RCL] 1	038– 45 1	[×]	067– 20
[g] [x^2]	039– 43 11	[RCL] 6	068– 45 6
[−]	040– 30	[+]	069– 40
[×]	041– 20	[RCL] 0	070– 45 0
6	042– 6	[÷]	071– 10
[÷]	043– 10	[g] [RTN]	072– 43 32
[GSB] 2	044– 32 2	[f] [LBL] [B]	073–42,21,12
[RCL] 5	045– 45 5	[STO] 4	074– 44 4
[g] [x^2]	046– 43 11	2	075– 2

KEYSTROKES	DISPLAY	KEYSTROKES	DISPLAY
[÷]	076– 10	[RCL] 2	105– 45 2
[RCL] 1	077– 45 1	[CHS]	106– 16
[g] [F?] 0	078–43, 6, 0	[STO] [I]	107– 44 25
2	079– 2	[RCL] 1	108– 45 1
[g] [F?] 0	080–43, 6, 0	[STO] [−] 5	109–44,30, 5
[÷]	081– 10	[f] [LBL] 0	110–42,21, 0
[RCL] 4	082– 45 4	[R/S]	111– 31
[−]	083– 30	[RCL] [I]	112– 45 25
[R/S]	084– 31	[×]	113– 20
[×]	085– 20	[RCL] 1	114– 45 1
[×]	086– 20	[÷]	115– 10
[g] [F?] 0	087–43, 6, 0	[g] [F?] 0	116–43, 6, 0
[g] [LSTx]	088– 43 36	[g] [RTN]	117– 43 32
[GSB] 1	089– 32 1	[RCL] 5	118– 45 5
[GSB] 2	090– 32 2	[×]	119– 20
[RCL] 6	091– 45 6	[g] [RTN]	120– 43 32
[+]	092– 40	[f] [LBL] 2	121–42,21, 2
[g] [RTN]	093– 43 32	[STO] [+] 6	122–44,40, 6
[f] [LBL] 1	094–42,21, 1	[RCL] 4	123– 45 4
[STO] 6	095– 44 6	[STO] 5	124– 44 5
[RCL] 1	096– 45 1	[RCL] 3	125– 45 3
[RCL] 2	097– 45 2	[STO] [I]	126– 44 25
[−]	098– 30	[f] [x>y]	127– 42 20
[STO] [I]	099– 44 25	[GTO] 0	128– 22 0
[RCL] 4	100– 45 4	[RCL] 4	129– 45 4
[STO] 5	101– 44 5	[RCL] 1	130– 45 1
[RCL] 2	102– 45 2	[−]	131– 30
[f] [x>y]	103– 42 20	[STO] 5	132– 44 5
[GTO] 0	104– 22 0	[RCL] 1	133– 45 1

KEYSTROKES	DISPLAY	KEYSTROKES	DISPLAY
[RCL] 3	134– 45 3	[RCL] 5	143– 45 5
[−]	135– 30	[×]	144– 20
[STO] [I]	136– 44 25	[g] [RTN]	145– 43 32
[f] [LBL] 0	137–43,21, 0	[f] [LBL] 3	146–42,21, 3
[R/S]	138– 31	[g] [F?] 0	147–43, 6, 0
[RCL] 1	139– 45 1	3	148– 3
[÷]	140– 10	[g] [F?] 0	149–43, 6, 0
[g] [F?] 0	141–43, 6, 0	[×]	150– 20
[g] [RTN]	142– 43 32		

REGISTERS			R_I: Used
R_0: *EI*	R_1: ℓ	R_2: *a*	R_3: *c*
R_4: *x*	R_5: *x*, $x-\ell$	R_6: Sums	

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYSTROKES	OUTPUT DATA/UNITS
1	Clear User mode, then key in the			
	program.			
2	Set User mode.		[f] [USER]	
3	Initialize.		[f] CLEAR [REG]	
4	Store beam constants:			
	Moment of inertia	*I*	[ENTER]	
	Modulus of elasticity	*E*	[×] [STO] 0	
	Length		[STO] 1	
5	If applicable, store load location(s):			
	Location of point load	*a*	[STO] 2	
	Location of applied moment	*c*	[STO] 3	

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYSTROKES	OUTPUT DATA/UNITS
6	Calculate any or all of the following at the point of interest, x (0 must be input for the distributed load, W, point load, P, and applied moment, M, if any of those are not applicable to the beam in question).			
	● Deflection:	x	[g] [CF] 0 [A]	
		W	[R/S]	
		P	[R/S]	
		M	[R/S]	y
	● Slope:	x	[g] [SF] 0 [A]	
		W	[R/S]	
		P	[R/S]	
		M	[R/S]	θ
	● Moment:	x	[g] [CF] 0 [B]	
		W	[R/S]	
		P	[R/S]	
		M	[R/S]	M_x
	● Shear:	x	[g] [SF] 0 [B]	
		W	[R/S]	
		P	[R/S]	
		M	[R/S]	V
7	For a new beam go to step 3.			
8	For different load locations go to step 5.			
9	For different position, x, go to step 6.			

Example 1: Find the deflection, slope, internal moment and shear at a distance of 24 inches for the beam below. Neglect the weight of the beam.



Keystrokes	Display		
Set User mode.			
[f] CLEAR [REG]			
[f] [ENG] 3			
.92 [ENTER]	920.0	-03	
30 [EEX] 6 [×] [STO] 0	27.60	06	
72 [STO] 1	72.00	00	
40 [STO] 3	40.00	00	
24 [g] [CF] 0 [A]	304.1	03	
0 [R/S]	72.00	00	
0 [R/S]	40.00	00	
10000 [R/S]	-30.92	-03	y_{24}, in
24 [g] [SF] 0 [A]	179.7	03	
0 [R/S]	72.00	00	
0 [R/S]	40.00	00	
10000 [R/S]	-322.1	-06	θ_{24}
24 [g] [CF] 0 [B]	48.00	00	
0 [R/S]	72.00	00	
0 [R/S] 10000 [R/S]	3.333	03	M_{24}, in-lbf
24 [g] [SF] 0 [B]	12.00	00	
0 [R/S]	72.00	00	
0 [R/S]	40.00	00	
10000 [R/S]	138.9	00	V_{24}, lbf

Example 2: What is the slope of the beam below at $x = 38$ inches?



$E = 30 \times 10^6$ psi

$I = 1.30$ in^4

Keystrokes	Display		
[f] CLEAR [REG]			
1.3 [ENTER]	1.300	00	
30 [EEX] 6 [×] [STO] 0	39.00	06	
50 [STO] 1	50.00	00	
44 [STO] 2	44.00	00	
38 [g] [SF] 0 [A]	-88.71	03	
25 [R/S]	44.00	00	
1000 [R/S]	50.00	00	
0 [R/S]	3.327	-03	θ_{38}, in/in

Example 3: What is the total moment at the center of the beam below? (It is not necessary to know E or I, since these do not show up in the moment equations.)



First solve for the effect of P_2 by itself.

Keystrokes	Display		
[f] CLEAR [REG]			
70 [STO] 1	70.00	00	
50 [STO] 2	50.00	00	
35 [g] [CF] 0 [B]	35.00	00	
0 [R/S]	50.00	00	
1000 [R/S]	70.00	00	
0 [R/S]	10.00	03	in-lbf

Now solve for the effect of the distributed load, P_1, and M and add to it the effect of P_2. This is the final answer assuming superposition is valid.

Keystrokes	Display		
20 [STO] 2	20.00	00	
70 [STO] 3	70.00	00	
35 [B]	35.00	00	
37 [R/S]	70.00	00	
400 [R/S]	70.00	00	
10000 [CHS] [R/S]	21.66	03	in-lbf
10 [EEX] 3 [+]	31.66	03	M_{35}, in-lbf

Equations of Motion

This program provides an interchangeable solution between displacement, final velocity, acceleration, time and initial velocity for an object that undergoes constant acceleration. Given any three known parameters the two unknowns will be calculated. The motion must be linear.

Equations:

$$x = \frac{t(v + v_0)}{2}$$

$$x = v_0 t + \frac{1}{2} a t^2$$

$$x = vt - \frac{1}{2} a t^2$$

$$v = v_0 + at$$

$$x = \frac{v^2 - {v_0}^2}{2a}$$

where:

x = displacement;

v = final velocity;

a = acceleration;

t = time;

v_0 = initial velocity.

Remarks:

- Any consistent set of units may be used.
- Displacement, acceleration, and velocity should be considered signed (vector) quantities. For example: if initial velocity and acceleration are in opposite directions, one should be positive and the other negative.
- All equations assume that initial displacement, x_0, and initial time, t_0, equal zero.
- When there are two possible solutions for t, the program will calculate t as follows: If at least one positive real solution exists, it will be found. If two real solutions of the same sign exist, the smallest solution will be found. If no real solutions exist, **Error 0** will be displayed.

KEYSTROKES	DISPLAY	KEYSTROKES	DISPLAY
[f] CLEAR [PRGM]	000–	[STO] 1	011– 44 1
[f] [LBL] [A]	001–42,21,11	[f] [LBL] 2	012–42,21, 2
[RCL] 1	002– 45 1	[RCL] 1	013– 45 1
[g] [x≠0]	003– 43 30	[RCL] 0	014– 45 0
[GTO] 1	004– 22 1	[÷]	015– 10
[RCL] 2	005– 45 2	[RCL] 3	016– 45 3
[g] [x≠0]	006– 43 30	[RCL] 0	017– 45 0
[GTO] 0	007– 22 0	[×]	018– 20
[RCL] [I]	008– 45 25	2	019– 2
[GSB] 4	009– 32 4	[÷]	020– 10
[+]	010– 40	[+]	021– 40

KEYSTROKES	DISPLAY	KEYSTROKES	DISPLAY
[STO] 2	022– 44 2	[f] [LBL] 0	051–42,21, 0
[GTO] 8	023– 22 8	[RCL] 0	052– 45 0
[f] [LBL] 0	024–42,21, 0	[g] [x≠0]	053– 43 30
[RCL] 3	025– 45 3	[GTO] 0	054– 22 0
[g] [x≠0]	026– 43 30	[GSB] [E]	055– 32 15
[GTO] 0	027– 22 0	[RCL] 3	056– 45 3
[GSB] [E]	028– 32 15	2	057– 2
[RCL] [I]	029– 45 25	[×]	058– 20
[RCL] 2	030– 45 2	[÷]	059– 10
[+]	031– 40	[STO] 1	060– 44 1
2	032– 2	[GTO] 5	061– 22 5
[÷]	033– 10	[f] [LBL] 0	062–42,21, 0
[RCL] 0	034– 45 0	[RCL] 2	063– 45 2
[×]	035– 20	[GSB] 4	064– 32 4
[STO] 1	036– 44 1	[−]	065– 30
[÷]	037– 10	[STO] 1	066– 44 1
2	038– 2	[GTO] 8	067– 22 8
[÷]	039– 10	[f] [LBL] 1	068–42,21, 1
[STO] 3	040– 44 3	[RCL] 2	069– 45 2
[f] [LBL] 5	041–42,21, 5	[g] [x≠0]	070– 43 30
[RCL] 1	042– 45 1	[GTO] 3	071– 22 3
[RCL] 2	043– 45 2	[RCL] 3	072– 45 3
[RCL] [I]	044– 45 25	[g] [x≠0]	073– 43 30
[+]	045– 40	[GTO] 0	074– 22 0
[÷]	046– 10	[RCL] 1	075– 45 1
2	047– 2	[RCL] 0	076– 45 0
[×]	048– 20	[÷]	077– 10
[STO] 0	049– 44 0	2	078– 2
[g] [RTN]	050– 43 32	[×]	079– 20

KEYSTROKES	DISPLAY	KEYSTROKES	DISPLAY
RCL (i)	080– 45 25	RCL 0	109– 45 0
−	081– 30	g x≠0	110– 43 30
STO 2	082– 44 2	GTO 7	111– 22 7
GTO 7	083– 22 7	GSB E	112– 32 15
f LBL 0	084–42,21, 0	RCL 1	113– 45 1
RCL 0	085– 45 0	÷	114– 10
g x≠0	086– 43 30	2	115– 2
GTO 2	087– 22 2	÷	116– 10
RCL 1	088– 45 1	STO 3	117– 44 3
RCL 3	089– 45 3	GTO 5	118– 22 5
×	090– 20	f LBL 6	119–42,21, 6
2	091– 2	RCL 2	120– 45 2
×	092– 20	g x^2	121– 43 11
RCL (i)	093– 45 25	RCL 1	122– 45 1
g x^2	094– 43 11	RCL 3	123– 45 3
+	095– 40	×	124– 20
$\sqrt{x}$	096– 11	2	125– 2
RCL 1	097– 45 1	×	126– 20
ENTER	098– 36	−	127– 30
g ABS	099– 43 16	$\sqrt{x}$	128– 11
÷	100– 10	RCL 1	129– 45 1
×	101– 20	ENTER	130– 36
STO 2	102– 44 2	g ABS	131– 43 16
GTO 5	103– 22 5	÷	132– 10
f LBL 3	104–42,21, 3	CHS	133– 16
RCL 3	105– 45 3	×	134– 20
g x≠0	106– 43 30	RCL 2	135– 45 2
GTO 6	107– 22 6	+	136– 40
f LBL 0	108–42,21, 0	RCL 3	137– 45 3

KEYSTROKES	DISPLAY	KEYSTROKES	DISPLAY
[÷]	138– 10	[−]	157– 30
[STO] 0	139– 44 0	[STO] [I]	158– 44 25
[GTO] 8	140– 22 8	[g] [RTN]	159– 43 32
[f] [LBL] 7	141–42,21, 7	[f] [LBL] 4	160–42,21, 4
[RCL] 2	142– 45 2	[RCL] 0	161– 45 0
[RCL] 1	143– 45 1	[×]	162– 20
[RCL] 0	144– 45 0	[RCL] 3	163– 45 3
[÷]	145– 10	[RCL] 0	164– 45 0
[−]	146– 30	[g] [x^2]	165– 43 11
[RCL] 0	147– 45 0	[×]	166– 20
[÷]	148– 10	2	167– 2
2	149– 2	[÷]	168– 10
[×]	150– 20	[g] [RTN]	169– 43 32
[STO] 3	151– 44 3	[f] [LBL] [E]	170–42,21,15
[f] [LBL] 8	152–42,21, 8	[RCL] 2	171– 45 2
[RCL] 2	153– 45 2	[g] [x^2]	172– 43 11
[RCL] 3	154– 45 3	[RCL] [I]	173– 45 25
[RCL] 0	155– 45 0	[g] [x^2]	174– 43 11
[×]	156– 20	[−]	175– 30

REGISTERS			R_I: v_0
R_0: t	R_1: x	R_2: v	R_3: a

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYSTROKES	OUTPUT DATA/UNITS
1	Clear Usr mode, then key in the			
	program.			
2	Set User mode.		[f] [USER]	
3	Store any 3 of the following:			
	Time	t	[STO] 0	
	Displacement	x	[STO] 1	
	Final velocity	v	[STO] 2	
	Acceleration	a	[STO] 3	
	Initial velocity	v_0	[STO] [I]	
4	Clear the 2 unknowns:			
	If t is unknown	0	[STO] 0	
	If x is unknown	0	[STO] 1	
	If v is unknown	0	[STO] 2	
	If a is unknown	0	[STO] 3	
	If v_0 is unknown	0	[STO] [I]	
5	Calculate the unknowns.		[A]	
6	Recall the desired values:			
	Time		[RCL] 0	t
	Displacement		[RCL] 1	x
	Final velocity		[RCL] 2	v
	Acceleration		[RCL] 3	a
	Initial velocity		[RCL] [I]	v_0

Example 1: An automobile accelerates for 4 seconds from a speed of 35 mph and covers a distance of 264 feet. What is the acceleration in ft/sec^2? If the acceleration continues to be constant, what distance is covered in the next second?

Keystrokes	Display	
Set User mode.		
[f] [FIX] 4		
264 [STO] 1		
35 [ENTER] 5280 [×]		
3600 [÷] [STO] [I]	**51.3333**	v_0, ft/sec
4 [STO] 0		
0 [STO] 2 [STO] 3		
[A] [RCL] 3	**7.3333**	a, ft/sec^2
5 [STO] 0		
0 [STO] 1 [STO] 2		
[A] [RCL] 1	**348.3333**	$x_{(t+1)}$, ft
264 [−]	**84.3333**	$x_{(t+1)} - x_{(t)}$, ft

Example 2: An airplane's take-off velocity is 125 mph. Assume a constant acceleration of 15 ft/sec^2. How much runway length, in feet, will be used from start to take-off? How long will it take for the plane to reach take-off velocity?

Keystrokes	Display	
125 [ENTER] 5280 [×]		
3600 [÷] [STO] 2	**183.3333**	v, ft/sec
15 [STO] 3		
0 [STO] [I] [STO] 1		$v_0 = 0$
[STO] 0 [A]	**12.2222**	
[RCL] 1	**1,120.3704**	x, ft
[RCL] 0	**12.2222**	t, sec

Soderberg's Equation for Fatigue

This program may be used to estimate maximum safe cyclic loads for a given size part or the minimum cross-sectional area needed to sustain a cyclic loading. The program uses Soderberg's equation which is graphically represented in Figure 1.



Working Stress Diagram

Equations:

$$\frac{s_{yp}}{FS} = \frac{s_{max} + s_{min}}{2} + K\left(\frac{s_{yp}}{s_e}\right)\left(\frac{s_{max} - s_{min}}{2}\right)$$

$$\frac{s_{max} + s_{min}}{2} = \frac{P_{max} + P_{min}}{2A}$$

$$\frac{s_{max} - s_{min}}{2} = \frac{P_{max} - P_{min}}{2A}$$

where:

s_{yp} is the yield point stress of the material;

s_e is the material endurance stress from reversed bending tests;

K is the stress concentration factor of the part;

FS is the factor of safety ($FS \geqslant 1.00$);

s_{max} is the maximum stress;

s_{min} is the minimum stress;

P_{max} is the maximum load;

P_{min} is the minimum load;

A is the cross-sectional area of the part.

References:

1. Spotts, M. F., *Design of Machine Elements,* Prentice-Hall, Inc., 1971.
2. Baumeister, T., *Marks Standard Handbook for Mechanical Engineers,* McGraw-Hill Book Company, 1967.

Remarks: This implementation of Soderberg's equation is for ductile materials only. Values of stress concentration factors and material endurance limits may be found in the reference sources. In the presence of corrosive media, or for rough surfaces, fatigue effects may be much more significant than predicted by this program.

KEYSTROKES	DISPLAY	KEYSTROKES	DISPLAY
[f] CLEAR [PRGM]	000–	[RCL] 3	014– 45 3
[f] [LBL] [A]	001–42,21,11	[×]	015– 20
[STO] 7	002– 44 7	[RCL] 7	016– 45 7
[R↓]	003– 33	[÷]	017– 10
[STO] 1	004– 44 1	[RCL] 5	018– 45 5
[R↓]	005– 33	[−]	019– 30
[STO] 2	006– 44 2	[RCL] 5	020– 45 5
[R↓]	007– 33	[RCL] 3	021– 45 3
[STO] 3	008– 44 3	[RCL] 1	022– 45 1
[R/S]	009– 31	[×]	023– 20
[f] [LBL] [B]	010–42,21,12	[RCL] 2	024– 45 2
[STO] 4	011– 44 4	[÷]	025– 10
[R/S]	012– 31	[×]	026– 20
[RCL] 6	013– 45 6	[g] [LSTx]	027– 43 36

KEYSTROKES	DISPLAY	KEYSTROKES	DISPLAY
1	028– 1	[RCL] 2	057– 45 2
[+]	029– 40	[÷]	058– 10
[R↓]	030– 33	1	059– 1
[+]	031– 40	[−]	060– 30
[g] [R↑]	032– 43 33	[÷]	061– 10
[÷]	033– 10	[STO] 5	062– 44 5
[STO] 4	034– 44 4	[g] [RTN]	063– 43 32
[g] [RTN]	035– 43 32	[f] [LBL] [D]	064–42,21,14
[f] [LBL] [C]	036–42,21 13	2	065– 2
[STO] 5	037– 44 5	[×]	066– 20
[R/S]	038– 31	[STO] 6	067– 44 6
[RCL] 3	039– 45 3	2	068– 2
[RCL] 1	040– 45 1	[÷]	069– 10
[×]	041– 20	[R/S]	070– 31
[RCL] 2	042– 45 2	[RCL] 4	071– 45 4
[÷]	043– 10	[RCL] 5	072– 45 5
1	044– 1	[+]	073– 40
[+]	045– 40	[RCL] 4	074– 45 4
[RCL] 4	046– 45 4	[RCL] 5	075– 45 5
[×]	047– 20	[−]	076– 30
[RCL] 6	048– 45 6	[RCL] 3	077– 45 3
[RCL] 3	049– 45 3	[×]	078– 20
[×]	050– 20	[RCL] 1	079– 45 1
[RCL] 7	051– 45 7	[×]	080– 20
[÷]	052– 10	[RCL] 2	081– 45 2
[−]	053– 30	[÷]	082– 10
[RCL] 3	054– 45 3	[+]	083– 40
[RCL] 1	055– 45 1	[RCL] 3	084– 45 3
[×]	056– 20	[RCL] 7	085– 45 7

KEYSTROKES	DISPLAY	KEYSTROKES	DISPLAY
[÷]	086– 10	2	089– 2
[÷]	087– 10	[÷]	090– 10
[STO] 6	088– 44 6		

REGISTERS			R_I: Unused
R_0: Unused	R_1: K	R_2: s_e	R_3: s_{yp}
R_4: P_{max}	R_5: P_{min}	R_6: $2A$	R_7: FS
R_8—$R_{.5}$ Unused			

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYSTROKES	OUTPUT DATA/UNITS
1	Clear User mode, then key in the			
	program.			
2	Set User mode.		[f] [USER]	
3	Input:			
	Yield point stress	s_{yp}	[ENTER]	s_{yp}
	Endurance stress	s_e	[ENTER]	s_e
	Stress concentration factor	K	[ENTER]	K
	Factor of safety	FS	[A]	s_{yp}
4	Input two of the following:			
	Maximum load	P_{max}	[B]	P_{max}
	Minimum load	P_{min}	[C]	P_{min}
	Area of cross section	A	[D]	A
5	Calculate the unknown value:			
	Maximum load		[B] [R/S]	P_{max}
	Minimum load		[C] [R/S]	P_{min}
	Area of cross section		[D] [R/S]	A
6	For new loading or area go to			
	step 4.			
7	For new case go to step 3.			

Example: What is the maximum permissible cyclic load for a part if the minimum load is 2,000 pounds and the area is 0.5 square inches?

$$s_{yp} = 70{,}000 \text{ psi}$$

$$s_e = 25{,}000 \text{ psi}$$

$$K = 1.25$$

$$FS = 2.0$$

Keystrokes	**Display**	
Set User mode.		
[f] [FIX] 2		
70000 [ENTER]	**70,000.00**	
25000 [ENTER]	**25,000.00**	
1.25 [ENTER]	**1.25**	
2 [A]	**70,000.00**	
2000 [C]	**2,000.00**	
.5 [D]	**0.50**	
[B] [R/S]	**8,888.89**	P_{max}, lbf

Composite Section Properties

The properties of arbitrarily shaped sections can be evaluated using this program. Exact solutions are obtained when the section is broken into a finite number of rectangles. Approximate solutions can be achieved by assuming that finite areas are concentrated at their centers.

The program calculates the area of the section, the centroid of the area, the moments of inertia about any specified set of axes, the polar moment of inertia about the specified axis, the moments of inertia about an axis translated to the centroid, the moments of inertia of the principal axis, the rotation angle between the translated axis and the principal axis, and the polar moments of inertia about the principal axis.

Equations:

$$A_{si} = \Delta x_i \, \Delta y_i$$

$$A = A_{s1} + A_{s2} + A_{s3} + \ldots + A_{sn}$$

$$\bar{x} = \frac{\sum_{i=1}^{n} x_{0i} A_{si}}{A}$$

$$\bar{y} = \frac{\sum_{i=1}^{n} y_{0i} A_{si}}{A} \qquad I_{xy} = \sum_{i=1}^{n} x_{0i} y_{0i} A_{si}$$

$$I_x = \sum_{i=1}^{n} \left({y_{0i}}^2 + \frac{{\Delta y_i}^2}{12} \right) A_{si} \qquad I_{\bar{x}} = I_x - A\bar{y}^2$$

$$I_{\bar{y}} = I_y - A\bar{x}^2$$

$$I_y = \sum_{i=1}^{n} \left({x_{0i}}^2 + \frac{{\Delta x_i}^2}{12} \right) A_{si} \qquad I_{\bar{x}\bar{y}} = I_{xy} - A\bar{x}\bar{y}$$

$$J = I_x + I_y \qquad \phi = \frac{1}{2} \tan^{-1} \left(\frac{2 I_{\bar{x}\bar{y}}}{I_{\bar{x}} - I_{\bar{y}}} \right)$$

$$I_{\bar{x}\phi} = I_{\bar{x}} \cos^2 \phi + I_{\bar{y}} \sin^2 \phi - I_{\bar{x}\bar{y}} \sin 2\phi$$

$$I_{\bar{y}\phi} = I_{\bar{y}} \cos^2 \phi + I_{\bar{x}} \sin^2 \phi + I_{\bar{x}\bar{y}} \sin 2\phi$$

$$J_\phi = I_{\bar{x}\phi} + I_{\bar{y}\phi}$$

where:

Δx_i is the width of a rectangular element;

Δy_i is the height of a rectangular element;

A_{si} is the area of an element;

A is the total area of the section;

$\bar{x}$ is the x coordinate of the centroid;

$\bar{y}$ is the y coordinate of the centroid;

x_{0i} is the x coordinate of the centroid of an element;

y_{0i} is the y coordinate of the centroid of an element;

I_x is the moment of inertia about the x-axis;

I_y is the moment of inertia about the y-axis;

J is the polar moment of inertia about the origin;

I_{xy} is the product of inertia;

$I_{\bar{x}}$ is the moment of inertia about the x-axis translated to the centroid;

$I_{\bar{y}}$ is the moment of inertia about the y-axis translated to the centroid;

$I_{\bar{x}\bar{y}}$ is the product of inertia about the translated axis;

ϕ is the angle between the translated axis and the principal axis;

$I_{\bar{x}\phi}$ is the moment of inertia about the translated, rotated, principal x-axis;

$I_{\bar{y}\phi}$ is the moment of inertia about the translated, rotated, principal y-axis;

J_{ϕ} is the polar moment of inertia about the origin of the translated, rotated, principal axes.

Remarks:

- Values of the polar moment of inertia J should not be used in torsional stress or strain analysis.
- The angle θ will be output in the current angular mode of the calculator.

References:

1. Crandall, S. H., Dahl, N.C., *An Introduction to the Mechanics of Solids*, McGraw-Hill, 1959.
2. Rhodes, G. F., *Section Properties*, HP-65 Users' Library, Number 262.

KEYSTROKES	DISPLAY	KEYSTROKES	DISPLAY
[f] CLEAR [PRGM]	000-	[STO] 2	011- 44 2
[f] [LBL] [A]	001-42,21,11	[x≷y]	012- 34
[STO] [I]	002- 44 25	[×]	013- 20
[R↓]	003- 33	[g] [LSTx]	014- 43 36
[STO] 1	004- 44 1	[f] [LBL] 1	015-42,21, 1
[R/S]	005- 31	[g] [x^2]	016- 43 11
[f] [LBL] [C]	006-42,21,13	1	017- 1
0	007- 0	2	018- 2
[STO] 2	008- 44 2	[÷]	019- 10
[GTO] 1	009- 22 1	[RCL] 1	020- 45 1
[f] [LBL] [B]	010-42,21,12	[ENTER]	021- 36

KEYSTROKES	DISPLAY	KEYSTROKES	DISPLAY
[×]	022– 20	[R/S]	051– 31
[+]	023– 40	[RCL] 5	052– 45 5
[×]	024– 20	[R/S]	053– 31
[STO] [+] 6	025–44,40, 6	[RCL] 4	054– 45 4
[g] [CLx]	026– 43 35	[RCL] 5	055– 45 5
[RCL] 2	027– 45 2	[÷]	056– 10
[g] [x^2]	028– 43 11	[R/S]	057– 31
1	029– 1	[RCL] 3	058– 45 3
2	030– 2	[RCL] 5	059– 45 5
[÷]	031– 10	[÷]	060– 10
[RCL] [I]	032– 45 25	[R/S]	061– 31
[g] [x^2]	033– 43 11	[RCL] 7	062– 45 7
[+]	034– 40	[R/S]	063– 31
[×]	035– 20	[RCL] 6	064– 45 6
[STO] [+] 7	036–44,40, 7	[R/S]	065– 31
[g] [CLx]	037– 43 35	[RCL] 6	066– 45 6
[RCL] [I]	038– 45 25	[RCL] 7	067– 45 7
[×]	039– 20	[+]	068– 40
[STO] [+] 3	040–44,40, 3	[R/S]	069– 31
[RCL] 1	041– 45 1	[RCL] 7	070– 45 7
[g] [R↑]	042– 43 33	[RCL] 3	071– 45 3
[×]	043– 20	[g] [x^2]	072– 43 11
[STO] [+] 4	044–44,40, 4	[RCL] 5	073– 45 5
[×]	045– 20	[÷]	074– 10
[g] [R↑]	046– 43 33	[−]	075– 30
[÷]	047– 10	[STO] 1	076– 44 1
[STO] [+] 0	048–44,40, 0	[R/S]	077– 31
[g] [R↑]	049– 43 33	[RCL] 0	078– 45 0
[STO] [+] 5	050–44,40, 5	[RCL] 3	079– 45 3

KEYSTROKES	DISPLAY	KEYSTROKES	DISPLAY
[RCL] 4	080– 45 4	[RCL] 3	109– 45 3
[×]	081– 20	1	110– 1
[RCL] 5	082– 45 5	[f] [→R]	111– 42 26
[÷]	083– 10	[g] [x^2]	112– 43 11
[−]	084– 30	[STO] 5	113– 44 5
[STO] 2	085– 44 2	[RCL] 1	114– 45 1
[RCL] 6	086– 45 6	[×]	115– 20
[RCL] 4	087– 45 4	[$x \gtrless y$]	116– 34
[g] [x^2]	088– 43 11	[g] [x^2]	117– 43 11
[RCL] 5	089– 45 5	[STO] 6	118– 44 6
[÷]	090– 10	[RCL] [I]	119– 45 25
[−]	091– 30	[×]	120– 20
[STO] [I]	092– 44 25	[+]	121– 40
[R/S]	093– 31	[RCL] 3	122– 45 3
[RCL] 2	094– 45 2	2	123– 2
2	095– 2	[×]	124– 20
[×]	096– 20	[SIN]	125– 23
[RCL] 1	097– 45 1	[RCL] 2	126– 45 2
[RCL] [I]	098– 45 25	[×]	127– 20
[−]	099– 30	[STO] 7	128– 44 7
[g] [$x=0$]	100– 43 40	[+]	129– 40
[GTO] 0	101– 22 0	[STO] 4	130– 44 4
[÷]	102– 10	[R/S]	131– 31
[g] [TAN^{-1}]	103– 43 25	[RCL] 5	132– 45 5
2	104– 2	[RCL] [I]	133– 45 25
[÷]	105– 10	[×]	134– 20
[f] [LBL] 0	106–42,21, 0	[RCL] 6	135– 45 6
[STO] 3	107– 44 3	[RCL] 1	136– 45 1
[R/S]	108– 31	[×]	137– 20

KEYSTROKES	DISPLAY	KEYSTROKES	DISPLAY
[+]	138– 40	[R/S]	142– 31
[RCL] 7	139– 45 7	[RCL] 4	143– 45 4
[–]	140– 30	[RCL] 5	144– 45 5
[STO] 5	141– 44 5	[+]	145– 40

REGISTERS			R_I: X_{0i}, $I_{\bar{x}}$
R_0: ΣI_{xyi}	R_1: Y_{0i}, $I_{\bar{y}}$	R_2: Δx_i, $I_{\bar{x}\bar{y}}$	R_3: $\Sigma x_{0i} A_{si}$, θ
R_4: $\Sigma Y_{0i} A_{si}$, $I_{\bar{y}\phi}$	R_5: Used	R_6: ΣI_{xi}, $\sin^2\theta$	R_7: Used

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYSTROKES	OUTPUT DATA/UNITS
1	Clear User mode, then key in the			
	program.			
2	Set User mode.		[f] [USER]	
3	Clear registers.		[f] CLEAR [REG]	
4	Set the appropriate angular mode.			
5	Input y-coordinate of centroid of			
	element.	y_{0i}	[ENTER]	y_{0i}
6	Input x-coordinate of centroid of			
	element.	x_{0i}	[A]	x_{0i}
7	● Input height of rectangular			
	element	Δy_i	[ENTER]	Δy_i
	and width of rectangular			
	element.	Δx_i	[B]	A_{si}
	or,			
	● input area of element for			
	approximate solution.	A_{si}	[C]	A_{si}
8	Go to step 5 for each element.			

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYSTROKES	OUTPUT DATA/UNITS
9	Calculate:			
	Area of section		[R/S]	A
	y-coordinate of centroid		[R/S]	$\bar{y}$
	x-coordinate of centroid		[R/S]	$\bar{x}$
	Moment of inertia about the			
	y-axis		[R/S]	I_y
	Moment of inertia about the			
	x-axis		[R/S]	I_x
	Polar moment of inertia about			
	the origin.		[R/S]	J
10	(Optional) Display product of			
	inertia.		[RCL] 0	I_{xy}
11	Calculate:			
	Moment of inertia about			
	translated y-axis		[R/S]	$I_{\bar{y}}$
	Moment of inertia about the			
	translated x-axis		[R/S]	$I_{\bar{x}}$
12	(Optional) Display product of			
	inertia for translated axes.		[RCL] 2	$I_{\bar{x}\bar{y}}$
13	Calculate:			
	Axis rotation angle		[R/S]	ϕ
	Moment of inertia about the			
	principal y-axis		[R/S]	$I_{\bar{y}\phi}$
	Moment of inertia about the			
	principal x-axis		[R/S]	$I_{\bar{x}\phi}$
	Polar moment of inertia about			
	the origin of the principal axis.		[R/S]	J_ϕ
14	For a new case go to step 4.			

Example 1: What is the moment of inertia about the x-axis (I_x) for the rectangular section shown? What is the moment of inertia about the neutral axis through the centroid of the section ($I_{\bar{x}\phi}$)?



Table of Inputs

Section	y_0	x_0	Δy	Δx
1	2.5	1.5	5	3

Keystrokes	**Display**	
Set User mode.		
[f] [FIX] 2		
[f] CLEAR [REG]		
2.5 [ENTER] 1.5 [A]	**2.50**	
5 [ENTER] 3 [B]	**15.00**	A_{si}
[R/S]	**15.00**	A
[R/S]	**2.50**	$\bar{y}$
[R/S]	**1.50**	$\bar{x}$
[R/S]	**45.00**	I_y
[R/S]	**125.00**	I_x
[R/S]	**170.00**	J
[R/S]	**11.25**	$I_{\bar{y}}$
[R/S]	**31.25**	$I_{\bar{x}}$

Keystrokes	Display	
[R/S]	0.00	ϕ
[R/S]	11.25	$I_{\bar{y}\phi}$
[R/S]	31.25	$I_{\bar{x}\phi}$

Note that $I_{\bar{x}\phi} = I_{\bar{x}}$ and $I_{\bar{y}\phi} = I_{\bar{y}}$ since $\phi = 0$.

Example 2: Calculate the section properties for the beam shown below.



Table of Inputs

Section	y_{0i}	x_{0i}	Δy	Δx
1	1	6	2	10
2	7	0.5	14	1
3	13.5	8.5	1	15

Keystrokes	Display	
[f] CLEAR [REG]		
[g] [DEG]		
1 [ENTER] 6 [A]	1.00	x_{01}
2 [ENTER] 10 [B]	20.00	A_{s1}
7 [ENTER] .5 [A]	7.00	x_{02}
14 [ENTER] 1 [B]	14.00	A_{s2}
13.5 [ENTER] 8.5 [A]	13.50	x_{03}
1 [ENTER] 15 [B]	15.00	A_{s3}
[R/S]	49.00	A
[R/S]	6.54	$\bar{y}$
[R/S]	5.19	$\bar{x}$
[R/S]	2,256.33	I_y
[R/S]	3,676.33	I_x
[R/S]	5,932.67	J
[RCL] 0	1,890.25	I_{xy}
[R/S]	934.49	$I_{\bar{y}}$
[R/S]	1,580.00	$I_{\bar{x}}$
[RCL] 2	225.61	$I_{\bar{x}\bar{y}}$
[R/S]	–17.48	ϕ
[R/S]	863.46	$I_{\bar{y}\phi}$
[R/S]	1,651.04	$I_{\bar{x}\phi}$
[R/S]	2,514.49	J_{ϕ}

Below is a figure showing the translated axis and the translated, rotated, principal axis of example 2.



Chemistry

pH of Weak Acid/Base Solutions

This program calculates the pH/pOH of a weak acid or base solution by Newton-Raphson iteration of

$$f(x) = x^3 + Kx^2 - (KC + K_w)x - KK_w = 0$$

where for weak acids, x, K and C are $[H^+]$, K_a and C_a, respectively, and $[OH^-]$, K_b and C_b for weak bases. The first estimate of x used in the iteration is

$$x_0 = (KC + K_w)^{1/2}.$$

Necessary inputs are C and K, but K may be in any of the following forms when solving for either acidic or basic solutions: K_a, K_b, pK_a, pK_b. C must be expressed in terms of molarity.

When solving for an acid solution, output is in the form of pH, $[H^+]$, K_a and an error term while for basic solutions the form is pOH, $[OH^-]$, K_b and the error term. Either output may be freely converted to its alternate form.

Note: To determine the pH of a salt of a weak acid, it must be remembered that the salt will hydrate to form the undissociated acid and a strong base (OH^-)

$$RA + H_2O \rightleftharpoons HA + R^+ + OH^-$$

Therefore, since such solutions are basic, the procedure **FOR BASES** should be followed. Similarly, salts of weak bases produce an acidic solution, so the procedure **FOR ACIDS** should be used.

Reference: J.N. Butler, *Ionic Equilibrium: A Mathematical Approach*, Addison-Wesley, Reading, Mass., 1964, pp. 70–71 and pp. 80–81.

This program was derived from a program for the HP-67, contributed to the HP User's Library by Alan J. Rubin.

KEYSTROKES	DISPLAY	KEYSTROKES	DISPLAY
f CLEAR PRGM	000–	x≷y	029– 34
f LBL A	001–42,21,11	g R↑	030– 43 33
EEX	002– 26	GSB A	031– 32 11
1	003– 1	g CF 0	032–43, 5, 0
4	004– 4	g R↑	033– 43 33
×	005– 20	g CLx	034– 43 35
1/x	006– 15	x≷y	035– 34
g F? 0	007–43, 6, 0	R↓	036– 33
g RTN	008– 43 32	R↓	037– 33
GTO D	009– 22 14	g RTN	038– 43 32
f LBL B	010–42,21,12	f LBL D	039–42,21,14
CHS	011– 16	g CF 0	040–43, 5, 0
10^x	012– 13	STO 1	041– 44 1
g F? 0	013–43, 6, 0	×	042– 20
g RTN	014– 43 32	EEX	043– 26
GTO D	015– 22 14	1	044– 1
f LBL C	016–42,21,13	4	045– 4
g SF 0	017–43, 4, 0	CHS	046– 16
GSB B	018– 32 12	+	047– 40
GSB A	019– 32 11	STO 3	048– 44 3
GTO D	020– 22 14	g LSTx	049– 43 36
f LBL E	021–42,21,15	RCL 1	050– 45 1
1	022– 1	×	051– 20
4	023– 4	STO 4	052– 44 4
x≷y	024– 34	RCL 3	053– 45 3
−	025– 30	$\sqrt{x}$	054– 11
x≷y	026– 34	STO 2	055– 44 2
g SF 0	027–43, 4, 0	f LBL 1	056–42,21, 1
GSB A	028– 32 11	RCL 2	057– 45 2

KEYSTROKES	DISPLAY	KEYSTROKES	DISPLAY
RCL 1	058– 45 1	−	078– 30
+	059– 40	÷	079– 10
RCL 2	060– 45 2	STO 0	080– 44 0
×	061– 20	g ABS	081– 43 16
RCL 3	062– 45 3	RCL 2	082– 45 2
−	063– 30	9	083– 9
RCL 2	064– 45 2	9	084– 9
×	065– 20	÷	085– 10
RCL 4	066– 45 4	f $x > y$	086– 42 20
−	067– 30	GTO 0	087– 22 0
RCL 2	068– 45 2	RCL 0	088– 45 0
3	069– 3	STO − 2	089–44,30, 2
×	070– 20	GTO 1	090– 22 1
RCL 1	071– 45 1	f LBL 0	091–42,21, 0
2	072– 2	RCL 0	092– 45 0
×	073– 20	RCL 1	093– 45 1
+	074– 40	RCL 2	094– 45 2
RCL 2	075– 45 2	ENTER	095– 36
×	076– 20	g LOG	096– 43 13
RCL 3	077– 45 3	CHS	097– 16

REGISTERS			R_I: Unused
R_0: $f(x)/f'(x)$	R_1: K	R_2: $[H^+]$ or $[OH^-]$	R_3: $KC + K_w$
R_4: KK_w	R_5—$R_{.4}$ Unused		

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYSTROKES	OUTPUT DATA/UNITS
1	Clear User mode, then key in the			
	program.			
2	Set User mode.		[f] [USER]	
3	Initialize.		[g] [CF] 0	
4	Enter molar concentration.	*C*	[ENTER]	*C*
	FOR ACIDS			
5a	Enter K_a	K_a	[D]	*pH*
	or enter K_b	K_b	[A]	*pH*
	or enter pK_a	pK_a	[B]	*pH*
	or enter pK_b.	pK_b	[C]	*pH*
5b	Find $[H^+]$.		[R↓]	$[H^+]$
5c	Find K_a.		[R↓]	K_a
5d	Find the error in $[H^+]$.		[R↓]	Error in $[H^+]$
6	Find *pOH*, $[OH^-]$ and K_b (*pH*,			
	$[H^+]$ and K_a must be in the X, Y,			
	and Z registers, respectively).		[R↓]	*pH*
			[E]	*pOH*
			[R↓]	$[OH^-]$
			[R↓]	K_b
	FOR BASES			
5a	Enter K_b		[D]	*pOH*
	or enter K_a		[A]	*pOH*
	or enter pK_b		[B]	*pOH*
	or enter pK_a.		[C]	*pOH*
5b	Find $[OH^-]$.		[R↓]	$[OH^-]$
5c	Find K_b.		[R↓]	K_b
5d	Find the error in $[OH^-]$.		[R↓]	Error in $[OH^-]$

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYSTROKES	OUTPUT DATA/UNITS
6	Find *pH*, $[H^+]$ and K_a (*pOH*,			
	$[OH^-]$ and K_b must occupy the			
	X, Y and Z registers, respectively).		[R↓]	*pOH*
			[E]	*pH*
			[R↓]	$[H^+]$
			[R↓]	K_a
7	For a new case go to step 4.			

Example 1: Find the *pH* of 10^{-4} M acetic acid if its K_a is 1.8×10^{-5}.

Keystrokes	Display		
Set User mode.			
[f] [FIX] 2			
[EEX] 4 [CHS] [ENTER]			
1.8 [EEX] 5 [CHS] [D]	4.46		*pH*
[R↓]	3.45	−05	$[H^+]$
[R↓]	1.80	−05	K_a
[R↓]	1.10	−07	Error in $[H^+]$

Example 2: Find the *pH* of 3.0×10^{-6} M NH_4C1 if the pK_b for ammonia is 4.75. Note that we are solving for the *pH* of the ammonium chloride solution not the *pOH* of an ammonia solution.

Keystrokes	Display		
3 [EEX] 6 [CHS] [ENTER]			
4.75 [C]	6.97		*pH*
[R↓]	1.08	−07	$[H^+]$
[R↓]	5.62	−10	K_a
[R↓]	4.04	−11	Error in $[H^+]$

Example 3: Calculate the pH of 0.002 M KCN whose pK_a is 9.32.

Keystrokes	Display		
.002 [ENTER]			
9.32 [C]	**3.71**		pOH
[R↓]	**1.95**	**-04**	$[OH^-]$
[R↓]	**2.09**	**-05**	K_b
[R↓]	**6.98**	**-07**	Error in $[OH^-]$
[R↓]	**3.71**		pOH
[E]	**10.29**		pH
[R↓]	**5.13**	**-11**	$[H^+]$
[R↓]	**4.79**	**-10**	K_a

Beer's Law

This is a flexible program for the calculation of the parameters of the Beer-Lambert law used in colorimetry, $A = \epsilon bC + I$, where I is the intercept (an error term). Given the light path, b, and a set of concentrations, C, and percent transmittance, $\%T$, or absorbance, A, the program computes the molar absorption coefficient, ϵ, by the least squares method:

$$\epsilon = \frac{1}{b} \frac{n\Sigma AC - \Sigma A \Sigma C}{n\Sigma C^2 - (\Sigma C)^2}$$

Initialization ([A]) clears all registers and sets b equal to 1 cm. Either concentration, absorbance or concentration, $\%T$ data may be entered. In the latter case $\%T$ is automatically converted to absorbance:

$$A = 2 - \log \%T$$

Note: Unless b is 1 cm a new value must be entered each time the program is initialized. Initialization also removes molecular weight from memory. The least squares program requires at least two C, A data pairs to calculate ϵ. Enter 0, 0 via [Σ+] if only one pair is available.

KEYSTROKES	DISPLAY	KEYSTROKES	DISPLAY
[f] CLEAR [PRGM]	000-	[x≷y]	028- 34
[f] [LBL] [A]	001-42,21,11	[GSB] [E]	029- 32 15
[f] CLEAR [REG]	002- 42 34	[x≷y]	030- 34
1	003- 1	[g] [RTN]	031- 43 32
[R/S]	004- 31	[f] [LBL] [E]	032-42,21,15
[STO] 6	005- 44 6	[g] [LOG]	033- 43 13
[R/S]	006- 31	2	034- 2
[STO] 7	007- 44 7	[−]	035- 30
[g] [RTN]	008- 43 32	[CHS]	036- 16
[f] [LBL] 2	009-42,21, 2	[g] [RTN]	037- 43 32
2	010- 2	[f] [LBL] [C]	038-42,21,13
[−]	011- 30	[GSB] 5	039- 32 5
[CHS]	012- 16	[GTO] 6	040- 22 6
[10^x]	013- 13	[f] [LBL] [D]	041-42,21,14
[g] [RTN]	014- 43 32	[GSB] 5	042- 32 5
[f] [LBL] 3	015-42,21, 3	[f] [LBL] [B]	043-42,21,12
[GSB] 5	016- 32 5	[GSB] 4	044- 32 4
[g] [Σ−]	017- 43 49	[f] [LBL] 6	045-42,21, 6
[R/S]	018- 31	[Σ+]	046- 49
[GTO] 1	019- 22 1	[R/S]	047- 31
[f] [LBL] 4	020-42,21, 4	[f] [LBL] 1	048-42,21, 1
[RCL] 7	021- 45 7	[f] [L.R.]	049- 42 49
[÷]	022- 10	[x≷y]	050- 34
[EEX]	023- 26	[RCL] 6	051- 45 6
3	024- 3	[÷]	052- 10
[÷]	025- 10	[R/S]	053- 31
[g] [RTN]	026- 43 32	0	054- 0
[f] [LBL] 5	027-42,21, 5	[f] [ŷ,r]	055- 42 48

KEYSTROKES	DISPLAY	KEYSTROKES	DISPLAY
[R/S]	056– 31	[R/S]	058– 31
[$x \gtrless y$]	057– 34	[GTO] 1	059– 22 1

REGISTERS			R_I: Unused
R_0: n	R_1: ΣA	R_2: ΣA^2	R_3: ΣC
R_4: ΣC^2	R_5: ΣAC	R_6: b	R_7: MW
R_8–$R_{.9}$: Unused			

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYSTROKES	OUTPUT DATA/UNITS
1	Clear User mode, then key in the			
	program.			
2	Set User mode.		[f] [USER]	
3	Initialize:		[A]	1
	Input light path	b	[R/S]	b
	Enter molecular weight if using			
	mg/ℓ as units	MW	[R/S]	MW
4	Inputs:			
	Either			
	● absorbance and molarity	A_i	[ENTER]	A_i
		C_i	[Σ+]	i
	or			
	● absorbance and mg/ℓ	A_i	[ENTER]	A_i
		mg/ℓ	[B]	i
	or			
	● percent transmittance	$\%T_i$	[ENTER]	$\%T_i$
	and molarity	C_i	[C]	i
	or			
	● percent transmittance	$\%T_i$	[ENTER]	$\%T_i$
	and mg/ℓ	mg/ℓ	[D]	i

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYSTROKES	OUTPUT DATA/UNITS
5	Repeat step 4 for each *i* (being			
	consistent with units).			
6	To correct			
	● absorbance entry	A_k	[ENTER]	A_k
		C_k	[g] [Σ-]	$k-1$
	● transmittance entry	$\%T_k$	[ENTER]	$\%T_k$
		C_k	[GSB] 3	$k-1$
7	Calculation:			
7a	Absorbance coefficient. If			
	absorbance and molarity was			
	input in step 4		[GSB] 1	ϵ
	Otherwise		[R/S]	ϵ
7b	Intercept		[R/S]	ℓ
7c	Correlation coefficient		[R/S]	r
8	To repeat ϵ, ℓ, r		[R/S]	ϵ
			[R/S]	ℓ
			[R/S]	r
9	To add further data, go to step 4.			
10	For a new case, go to step 3.			
11	To calculate molarity from mg/ℓ			
	values	MW	[STO] 7	
		mg/ℓ	[GSB] 4	M
12	To calculate A from $\%T$	$\%T$	[E]	A
13	To calculate $\%T$ from A	A	[GSB] 2	$\%T$

Example: Calculate the molar absorption coefficient for phosphorus as determined by the "ascorbic acid" method at 880 nm. The molecular weight of phosphorus is 31. The light path is 1.2 cm.

Data

%*T*	mg/l
97.9	0.0
58.0	0.25
37.2	0.50
23.1	0.75
14.5	1.00
9.0	1.25

Keystrokes	**Display**	
Set User mode.		
[f] [FIX] 4		
[A]	**1.0000**	
1.2 [R/S]	**1.2000**	
31 [R/S]	**31.000**	
97.9 [ENTER] 0 [D]	**1.0000**	
58 [ENTER] 0.25 [D]	**2.0000**	
37.2 [ENTER] 0.5 [D]	**3.0000**	
23.1 [ENTER] 0.75 [D]	**4.0000**	
14.5 [ENTER] 1 [D]	**5.0000**	
9 [ENTER] 1.25 [D]	**6.0000**	
[R/S]	**21,244.7779**	ϵ
[R/S]	**0.0187**	ℓ
[R/S]	**0.9998**	r

Economic Analysis

Mortgage Loan Interest Rate



This program will calculate the interest rate on a loan with equal periodic payments. The user must specify the number of periods, the present value or initial loan amount, and the payment amount.

The program performs an iterative solution for i using Newton's method:

$$i_{k+1} = i_k - \frac{f(i_k)}{f'(i_k)}$$

where

$$f(i) = \frac{1-(1+i)^{-n}}{i} - \left|\frac{PV}{PMT}\right|$$

The initial guess for i is given by

$$i_0 = \left|\frac{PMT}{PV}\right| - \frac{1}{n^2}\left|\frac{PV}{PMT}\right|$$

where

PV is the initial loan amount.

PMT is the periodic payment.

i is the periodic interest rate expressed as a decimal.

n is the number of periods.

Note: Cash received is represented by a positive value (+). Cash paid out is represented by a negative value (−).

KEYSTROKES	DISPLAY	KEYSTROKES	DISPLAY
[f] CLEAR [PRGM]	000–	[−]	029– 30
[f] [LBL] [A]	001–42,21,11	[−]	030– 30
[÷]	002– 10	[RCL] 1	031– 45 1
[STO] 3	003– 44 3	[RCL] 2	032– 45 2
[R↓]	004– 33	[1/x]	033– 15
[STO] 1	005– 44 1	1	034– 1
[RCL] 3	006– 45 3	[+]	035– 40
[g] [ABS]	007– 43 16	[÷]	036– 10
[ENTER]	008– 36	1	037– 1
[1/x]	009– 15	[+]	038– 40
[x≷y]	010– 34	[RCL] 0	039– 45 0
[RCL] 1	011– 45 1	[×]	040– 20
[g] [x^2]	012– 43 11	1	041– 1
[÷]	013– 10	[−]	042– 30
[−]	014– 30	[RCL] 2	043– 45 2
[STO] 2	015– 44 2	[÷]	044– 10
[f] [LBL] 0	016–42,21, 0	[÷]	045– 10
[RCL] 3	017– 45 3	[STO] [+] 2	046–44,40, 2
[g] [ABS]	018– 43 16	[g] [ABS]	047– 43 16
[RCL] 2	019– 45 2	[EEX]	048– 26
[×]	020– 20	6	049– 6
1	021– 1	[CHS]	050– 16
[RCL] 2	022– 45 2	[f] [x≤y]	051– 42 10
1	023– 1	[GTO] 0	052– 22 0
[+]	024– 40	[RCL] 2	053– 45 2
[RCL] 1	025– 45 1	[EEX]	054– 26
[CHS]	026– 16	2	055– 2
[y^x]	027– 14	[×]	056– 20
[STO] 0	028– 44 0		

REGISTERS			R_I: Unused
R_0: $(1+i)^{-n}$	R_1: n	R_2: i	R_3: PV/PMT
R_4–R_9: Unused			

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYSTROKES	OUTPUT DATA/UNITS
1	Clear User mode, then key in the			
	program.			
2	Set User mode.		[f] [USER]	
3	Enter the number of payments,	n	[ENTER]	n
	the present value,	PV*	[ENTER]	PV
	and payment amount.	PMT*	[A]	i (%)
				periodic rate
4	Calculate the annual			
	percentage rate.	# periods/yr.	[×]	i (%)
				annual rate
5	For a new case go to step 3.			

Example: You recently obtained a $2500 car loan for 36 months. If your monthly payment is $86.67, what is the annual percentage rate?

Keystrokes	Display	
Set User mode.		
[f] [FIX] 2		
36 [ENTER]		
2500 [ENTER]		
86.67 [CHS] [A]	**1.25**	% Monthly rate.
12 [×]	**15.01**	% Annual rate.

*Note: Cash received is represented by a positive value (+) while cash paid out is represented by a negative value (−).

Discounted Cash Flow Analysis

Assuming a minimum desired yield (cost of capital, discount rate), this program finds the present value of the future cash flows generated by the investment and subtracts the initial investment from this amount. If the final net present value is a positive value, the investment exceeds the profit objectives assumed. If the final net present value is a negative value, then the investment is not profitable to the extent of the desired yield. If the net present value is zero, the investment meets the profit objectives.

The function associated with [C] (#) is designed to accommodate those situations where a series of cash flows are equal. You enter the number of times these equal periodic cash flows occur with [C], and then the amount only once with [D]. If the cash flow occurs only once, there is no need to enter anything for #.

Zero must be entered for all periods with no cash flow. When a cash flow is an outlay (initial or additional investment, loss, etc.) the value must be entered as a negative number.

Cash flows are assumed to occur at the end of the cash flow periods.

KEYSTROKES	DISPLAY	KEYSTROKES	DISPLAY
[f] CLEAR [PRGM]	000-	[STO] 1	014- 44 1
[f] [LBL] [A]	001-42,21,11	[g] [LSTx]	015- 43 36
[STO] 0	002- 44 0	[×]	016- 20
0	003- 0	[g] [RTN]	017- 43 32
[STO] 5	004- 44 5	[f] [LBL] [C]	018-42,21,13
1	005- 1	[STO] 2	019- 44 2
[STO] 2	006- 44 2	[g] [RTN]	020- 43 32
[RCL] 0	007- 45 0	[f] [LBL] [D]	021-42,21,14
[CHS]	008- 16	[STO] 3	022- 44 3
[g] [RTN]	009- 43 32	1	023- 1
[f] [LBL] [B]	010-42,21,12	[RCL] 1	024- 45 1
[EEX]	011- 26	[+]	025- 40
2	012- 2	[RCL] 2	026- 45 2
[÷]	013- 10	[STO] [+] 5	027-44,40, 5

KEYSTROKES	DISPLAY	KEYSTROKES	DISPLAY
[y^x]	028– 14	1	041– 1
[STO] 4	029– 44 4	[RCL] 1	042– 45 1
[RCL] 0	030– 45 0	[+]	043– 40
[×]	031– 20	[RCL] 5	044– 45 5
[RCL] 4	032– 45 4	[y^x]	045– 14
1	033– 1	[÷]	046– 10
[−]	034– 30	1	047– 1
[RCL] 1	035– 45 1	[STO] 2	048– 44 2
[÷]	036– 10	[R↓]	049– 33
[RCL] 3	037– 45 3	[g] [RTN]	050– 43 32
[×]	038– 20	[f] [LBL] [E]	051–42,21,15
[+]	039– 40	[RCL] 5	052– 45 5
[STO] 0	040– 44 0		

REGISTERS			R_I: Unused
R_0: *NPV*	R_1: $i/100$	R_2: #	R_3: *CF*
R_4: $(1+i)^n$	R_5: Σn	R_6—$R_{.9}$: Unused	

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYSTROKES	OUTPUT DATA/UNITS
1	Clear User mode, then key in the			
	program.			
2	Set User mode.		[f] [USER]	
3	Input			
	Initial investment	*INV*	[A]	*INV*
	Periodic interest (discount) rate.	i(%)	[B]	i%
4	Key in the number of equal cash			
	flows if greater than 1.	#	[C]	#

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYSTROKES	OUTPUT DATA/UNITS
5	Key in cash flow amount(s) then			
	calculate net present value.	*CF*	[D]	*NPV*
6	(Optional) Display total number of			
	cash flows entered so far.		[E]	*n*
7	For next cash flow go to step 4.			
8	For next case go to step 3.			

Example 1: An investor has an opportunity to purchase a piece of property for $70,000. If the going rate of return on this type of investment is 13.75%, and the after-tax cash flows are forecast as follows, should the investor purchase the property?

Year	Cash Flow ($)
1	$14,000
2	11,000
3	10,000
4	10,000
5	10,000
6	9,100
7	9,000
8	9,000
9	4,500
10	71,000
	(property sold in 10th year)

Keystrokes	Display	
Set User mode.		
[f] [FIX] 2		
70000 [CHS] [A]		
13.75 [B] 14000 [D]	**-57,692.31**	*NPV* after 1 cash flow.
11000 [D]	**-49,190.92**	*NPV* after 2 cash flows.
3 [C] 10000 [D]	**-31,172.57**	*NPV* after 5 cash flows.
9100 [D]	**-26,971.76**	*NPV* after 6 cash flows.
2 [C] 9000 [D]	**-20,108.39**	*NPV* after 8 cash flows.

Keystrokes	Display	
[E]	8.00	Check that we've entered 8 periods cash flows so far.
4500 [D]	-18,696.99	*NPV* after 9 cash flows.
71000 [D]	879.93	*NPV* after 10 cash flows.

Since the final *NPV* is positive, the investment meets the profit objectives.

Example 2: The Cooper Company needs a new photocopier and is considering leasing the equipment as an alternative to buying. The end-of-the-year net cash cost of each option is:

Purchase

Year	Net Cash Cost
1	$ 533
2	948
3	1,375
4	1,815
5	2,270
Total Net Cash Cost	$6,941

Lease

Year	Net Cash Cost
1	$1,310
2	1,310
3	1,310
4	1,310
5	1,310
Total Net Cash Cost	$6,550

Looking at total cost, leasing appears to be less. But purchasing costs less the first two years. Mr. Cooper knows that he can make a 15% return on every dollar he puts in the business; the sooner he can reinvest money, the sooner he earns 15%. Therefore, he decides to consider the *timing of the costs,* discounting the cash flows at 15% to find the present value of the alternatives. Which option should he choose?

Keystrokes	Display
Purchase:	
0 [A] 15 [B] 533 [D]	
948 [D]	**1,180.30**
1375 [D] 1815 [D]	**3,122.12**
2270 [D]	**4,250.71**
Lease:	
0 [A] 5 [C] 1310 [D]	**4,391.32**

Leasing has a present value of $4391.32, while purchasing has a present value of $4250.71. Since these are both expense items, the lowest present value is the most desirable. So, in this case, purchasing is the least costly alternative.

Amortization Schedules



This program finds both the total interest paid over a specified number of payment periods and the remaining balance at the end of the last specified period given the periodic interest rate, periodic payment amount (*PMT*), loan amount (*PV*), and the beginning and ending payment numbers for the time span being considered. The payments associated with both the beginning (J) and the ending (K) payment periods are included in the calculation.

The program can be used for loans with a balloon payment as well as loans arranged to be fully amortized provided two cautions are observed. First, the balloon payment of the loan must be at the same time as, and in addition to, the last payment. Second, care should be taken not to enter a value for K that is after the last payment since the program has no way of knowing the term of the loan.

The data generated is valid for loans that have a balloon payment, as well as those that are arranged to be fully amortized. For loans with a balloon payment, the remaining balance of the last payment period is the balloon payment due in addition to the last periodic payment.

For loans scheduled to be fully amortized, the remaining balance after the last payment period may be slightly more or less than zero. This is because the program assumes that *all* payments are equal to the value entered for *PMT*. In fact for most loans, the last payment is slightly more or less than the rest.

An option is available to output the amortization schedule between payments J and K ([C]).

The calculator performs all internal calculations to ten digits. If the user wishes to round the schedule to dollars and cents, the following sequence may be used:

1. Press [GTO] [·] 108
2. Set Program mode: [g] [P/R]
3. Press [g] [RND]
4. Return to Run mode: [g] [P/R]

Note: Cash received is represented by a positive value (+). Cash paid out is represented by a negative value (−). The loan amount (*PV*) and the payment (*PMT*) must have opposite signs.

Equations:

$$BAL_K = \frac{1}{(1+i)^{-K}}\left[PV - PMT\frac{(1+i)^{-K}-1}{i}\right]$$

$$INT_{J-K} = BAL_{J-1} - BAL_K + (K - J + 1)\cdot PMT$$

where:

kth payment to principal $= BAL_K - BAL_{K-1}$
kth payment to interest $= PMT + (BAL_{K-1} - BAL_K)$
Total payment to interest $= (K)\times(PMT) + (PV - BAL_K)$

KEYSTROKES	DISPLAY	KEYSTROKES	DISPLAY
[f] CLEAR [PRGM]	000–	[STO] 3	016– 44 3
[f] [LBL] [A]	001–42,21,11	[R/S]	017– 31
[RCL] 0	002– 45 0	[f] [LBL] [B]	018–42,21,12
[STO] 7	003– 44 7	[RCL] 0	019– 45 0
[x≷y]	004– 34	[RCL] 7	020– 45 7
[STO] 0	005– 44 0	[f] [x≤y]	021– 42 10
[R/S]	006– 31	[GTO] 0	022– 22 0
[EEX]	007– 26	[STO] 0	023– 44 0
2	008– 2	[R↓]	024– 33
[÷]	009– 10	[STO] 7	025– 44 7
[STO] 1	010– 44 1	[f] [LBL] 0	026–42,21, 0
[g] [LSTx]	011– 43 36	1	027– 1
[×]	012– 20	[RCL] 1	028– 45 1
[R/S]	013– 31	[+]	029– 40
[STO] 2	014– 44 2	[STO] 8	030– 44 8
[R/S]	015– 31	[RCL] 0	031– 45 0

KEYSTROKES	DISPLAY	KEYSTROKES	DISPLAY
[GSB] 1	032– 32 1	[GSB] 1	061– 32 1
[STO] 4	033– 44 4	[STO] 4	062– 44 4
[RCL] 8	034– 45 8	[RCL] 8	063– 45 8
[RCL] 7	035– 45 7	[RCL] 7	064– 45 7
1	036– 1	1	065– 1
[−]	037– 30	[−]	066– 30
[GSB] 1	038– 32 1	[GSB] 1	067– 32 1
[RCL] 4	039– 45 4	[RCL] 4	068– 45 4
[−]	040– 30	[−]	069– 30
[STO] 6	041– 44 6	[STO] 6	070– 44 6
[RCL] 0	042– 45 0	[RCL] 2	071– 45 2
[RCL] 7	043– 45 7	[+]	072– 40
[−]	044– 30	[R/S]	073– 31
1	045– 1	[RCL] 6	074– 45 6
[+]	046– 40	[CHS]	075– 16
[RCL] 2	047– 45 2	[R/S]	076– 31
[×]	048– 20	[RCL] 4	077– 45 4
[+]	049– 40	[R/S]	078– 31
[R/S]	050– 31	[RCL] 7	079– 45 7
[RCL] 4	051– 45 4	[RCL] 2	080– 45 2
[R/S]	052– 31	[×]	081– 20
[f] [LBL] [C]	053–42,21,13	[RCL] 3	082– 45 3
[RCL] 7	054– 45 7	[RCL] 4	083– 45 4
[R/S]	055– 31	[−]	084– 30
1	056– 1	[+]	085– 40
[RCL] 1	057– 45 1	[R/S]	086– 31
[+]	058– 40	1	087– 1
[STO] 8	059– 44 8	[STO] [+] 7	088–44,40, 7
[RCL] 7	060– 45 7	[RCL] 0	089– 45 0

KEYSTROKES	DISPLAY	KEYSTROKES	DISPLAY
[RCL] 7	090– 45 7	[RCL] 1	100– 45 1
[f] [$x \le y$]	091– 42 10	[÷]	101– 10
[GTO] [C]	092– 22 13	[RCL] 2	102– 45 2
[g] [RTN]	093– 43 32	[×]	103– 20
[f] [LBL] 1	094–42,21, 1	[RCL] 3	104– 45 3
[CHS]	095– 16	[$x \ge y$]	105– 34
[y^x]	096– 14	[−]	106– 30
[STO] 5	097– 44 5	[RCL] 5	107– 45 5
1	098– 1	[÷]	108– 10
[−]	099– 30		

REGISTERS			R_I: Unused
R_0: *K*	R_1: $i/100$	R_2: *PMT*	R_3: *PV*
R_4: Used	R_5: Used	R_6: Used	R_7: *J*
R_8: $1 + i/100$	R_9–$R_{.2}$: Unused		

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYSTROKES	OUTPUT DATA/UNITS
1	Clear User mode, then key in the			
	program.			
2	Set User mode.		[f] [USER]	
3	Input			
	Starting period number,	*J*	[A]	*J*
	Ending period number,	*K*	[A]	*K*
	If *i*, *PMT*, and *PV* have been			
	previously entered, go to step 4.			
	Otherwise input:			
	Periodic interest rate	*i* (%)	[R/S]	*i* (%)
	Periodic payment amount	*PMT*	[R/S]	*PMT*
	Initial loan amount	*PV*	[R/S]	*PV*

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYSTROKES	OUTPUT DATA/UNITS
4	Computation			
4a	Compute the total interest paid			
	between periods *J* and *K* inclusive			
	and the remaining balance at end			
	of period *K*.		[B]	*INT*
			[R/S]	*BAL*
	OR,			
4b	Generate the amortization			
	schedule between payments *J*			
	and *K* inclusive.		[C]	*J*
4c	Output			
	Amount paid toward interest for			
	period		[R/S]	*PMT* to *INT*
	Amount paid toward principal			
	for period		[R/S]	*PMT* to *PRIN*
	Remaining balance at end of			
	period		[R/S]	*BAL*
	Total interest paid since J		[R/S]	*TOT INT*
	Index of next period		[R/S]	*J* + 1
	Go to 4c for next period.			
5	For new case go to step 3.			

Example 1: A mortgage is arranged such that the first payment is made at the end of October 1981 (i.e., October is payment period 1). It is an $80,000 loan at 16% interest, with monthly payments of $1,075.81. What is the accumulated interest for 1981 (periods 1-3) and 1982 (periods 4-15), and what would the remaining balance be at the end of each year?

Keystrokes **Display**

Set User mode.
[f] [FIX] 2

Keystrokes	Display	
1 [A] 3 [A] 16 [ENTER]		
12 [÷] [R/S]	1.33	Periodic interest rate.
1075.81 [CHS] [R/S]		
80000 [R/S] [B]	-3,199.63	Interest paid in 1981.
[R/S]	79,972.20	Remaining balance at the end of 1981.
4 [A] 15 [A] [B]	-12,786.80	Interest paid in 1982.
[R/S]	79,849.28	Remaining balance at the end of 1982.

Example 2: Generate an amortization schedule for the first two payments of an old $30,000 mortgage having monthly payments of $200 at 7% interest. Then jump ahead and generate the data for the 36th payment.

Keystrokes	Display	
1 [A] 2 [A] 7 [ENTER]		
12 [÷] [R/S]		
200 [CHS] [R/S]		
30000 [R/S] [C]	1.00	Starting 1st period.
[R/S]	-175.00	Payment to interest.
[R/S]	-25.00	Payment to principal.
[R/S]	29,975.00	Remaining balance.
[R/S]	-175.00	Total interest to date.
[R/S]	2.00	Starting 2nd period.
[R/S]	-174.85	Payment to interest.
[R/S]	-25.15	Payment to principal.
[R/S]	29,949.85	Remaining balance.
[R/S]	-349.85	Total interest to date.

Now let's skip ahead to the 36th payment period.

36 [A] [A] [C]	36.00	Starting 36th period.
[R/S]	-169.36	Payment to interest.
[R/S]	-30.64	Payment to principal.
[R/S]	29,001.75	Remaining balance.
[R/S]	-6,201.75	Total interest to date.

Depreciation

This program will calculate the depreciation schedule of an asset using the straight-line, sum-of-the-years digits or declining-balance methods. Input are the starting book value (SB), salvage value (SAL), useful life expectancy ($LIFE$), the declining rate factor (not applicable to $SOYD$) and the first year of the desired schedule (YR).

In the business community, the "variable rate" is indicated as either a factor or a percentage with equal frequency. Thus,a "1.5 declining-balance factor" and a "150% declining-balance" have the same meaning. The number to be keyed in for $FACT$ (the variable rate) in this program, should be in factor form, that is 1.25, 1.5, 2, and not 125, 150 or 200. An input of 1 for $FACT$ gives straight-line values for output.

This program does *not* calculate partial-year depreciation. If the life is input as a non-integer (i.e., it has a fractional part), the fractional year would be considered to follow the last whole year. For example, if 2.5 were input for the life, the half-year's depreciation would be calculated for year 3.

Equations:

Sum-of-the-Years Digits Schedule

$$SOYD = \left(\frac{(W+1)(W+2F)}{2} \right)$$

$$DEP_K = \left(\frac{LIFE + 1 - K}{SOYD} \right) \times (SBV - SAL)$$

$$DEP_K \text{ (last year)} = RDV_{K-1}$$

$$RDV_K = \left[\frac{(W-K+1) \times (W-K+2F)}{2 \times (SOYD)} \right] \times (SBV - SAL)$$

$$RBV_K = RDV_K + SAL$$

Variable-Rate Declining-Balance Schedule

$$DEP_K = SBV \times \left(1 - \frac{FACT}{LIFE}\right)^{K-1} \times \left(\frac{FACT}{LIFE}\right)$$

$$RDV_K = (SBV - SAL) - SBV \times \left[1 - \left(1 - \frac{FACT}{LIFE}\right)^K\right]$$

$$\mathrm{RBV_K} = RDV_K + SAL$$

$$DEP_K\,(\text{last year}) = RDV_{K-1}$$

Straight-Line Schedule

$$DEP_K = \frac{SBV - SAL}{LIFE}$$

$$DEP_K\,(\text{last year}) = \left(\frac{SBV - SAL}{LIFE}\right) \times F = RDV_{K-1}$$

$$RDV_K = (LIFE - K) \times \left(\frac{SBV - SAL}{LIFE}\right)$$

$$RBV_K = RDV_K + SAL$$

where:

K = value for YR

ΣDEP_K = total depreciation for years 1 through K

W = integer portion of $LIFE$

F = decimal portion of $LIFE$

(i.e., for a $LIFE$ of 12.25 years $W = 12$ and $F = .25$).

KEYSTROKES	DISPLAY	KEYSTROKES	DISPLAY
[f] CLEAR [PRGM]	000–	[f] [LBL] 7	003–42,21, 7
[f] [LBL] [A]	001–42,21,11	[g] [CF] 0	004–43, 5, 0
[g] [CF] 1	002–43, 5, 1	[f] [FIX] 2	005–42, 7, 2

KEYSTROKES	DISPLAY	KEYSTROKES	DISPLAY
[STO] 0	006– 44 0	1	035– 1
[R↓]	007– 33	[–]	036– 30
[STO] 1	008– 44 1	[y^x]	037– 14
[R↓]	009– 33	[×]	038– 20
[STO] 2	010– 44 2	[RCL] 1	039– 45 1
[R↓]	011– 33	[×]	040– 20
[STO] 3	012– 44 3	1	041– 1
[GSB] 2	013– 32 2	[RCL] 5	042– 45 5
[g] [F?] 1	014–43, 6, 1	[RCL] 0	043– 45 0
[GTO] 4	015– 22 4	[y^x]	044– 14
[R/S]	016– 31	[–]	045– 30
[STO] 4	017– 44 4	[RCL] 1	046– 45 1
[f] [LBL] 0	018–42,21, 0	[×]	047– 20
[g] [CF] 0	019–43, 5, 0	[RCL] 1	048– 45 1
[GSB] 2	020– 32 2	[RCL] 2	049– 45 2
[g] [F?] 1	021–43, 6, 1	[–]	050– 30
[GTO] 4	022– 22 4	[$x \gtrless y$]	051– 34
[RCL] 4	023– 45 4	[–]	052– 30
1	024– 1	[$x \gtrless y$]	053– 34
[f] [$x = y$]	025– 42 40	[GTO] 3	054– 22 3
[GTO] 1	026– 22 1	[f] [LBL] 1	055–42,21, 1
[RCL] 4	027– 45 4	[RCL] 1	056– 45 1
[RCL] 3	028– 45 3	[RCL] 2	057– 45 2
[÷]	029– 10	[–]	058– 30
[–]	030– 30	[RCL] 3	059– 45 3
[STO] 5	031– 44 5	[÷]	060– 10
[g] [LST x]	032– 43 36	[STO] 5	061– 44 5
[$x \gtrless y$]	033– 34	[RCL] 3	062– 45 3
[RCL] 0	034– 45 0	[×]	063– 20

KEYSTROKES	DISPLAY	KEYSTROKES	DISPLAY
[RCL] 5	064– 45 5	[g] [CL x]	093– 43 35
[RCL] 0	065– 45 0	[g] [RTN]	094– 43 32
[×]	066– 20	[f] [LBL] [C]	095–42,21,13
[−]	067– 30	[STO] 0	096– 44 0
[RCL] 5	068– 45 5	[GTO] 0	097– 22 0
[f] [LBL] 3	069–42,21, 3	[f] [LBL] [B]	098–42,21,12
[g] [F?] 0	070–43 6 0	[g] [SF] 1	099–43, 4, 1
[GTO] 9	071– 22 9	[GTO] 7	100– 22 7
[R/S]	072– 31	[f] [LBL] 4	101–42,21, 4
[x≷y]	073– 34	[RCL] 3	102– 45 3
[R/S]	074– 31	[g] [INT]	103– 43 44
[RCL] 2	075– 45 2	[STO] 4	104– 44 4
[+]	076– 40	[g] [LST x]	105– 43 36
[R/S]	077– 31	[ENTER]	106– 36
1	078– 1	[f] [FRAC]	107– 42 44
[STO] [+] 0	079–44,40, 0	[+]	108– 40
[RCL] 0	080– 45 0	[RCL] 4	109– 45 4
[RCL] 3	081– 45 3	1	110– 1
[f] [x>y]	082– 42 20	[+]	111– 40
[GTO] 0	083– 22 0	[×]	112– 20
[g] [CL x]	084– 43 35	2	113– 2
[g] [RTN]	085– 43 32	[÷]	114– 10
[f] [LBL] 9	086–42,21, 9	[STO] 5	115– 44 5
[+]	087– 40	1	116– 1
[R/S]	088– 31	[RCL] 4	117– 45 4
[g] [CL x]	089– 43 35	[RCL] 0	118– 45 0
[R/S]	090– 31	[−]	119– 30
[RCL] 2	091– 45 2	[+]	120– 40
[R/S]	092– 31	[g] [LST x]	121– 43 36

KEYSTROKES	DISPLAY	KEYSTROKES	DISPLAY
[RCL] 3	122– 45 3	[RCL] 3	137– 45 3
[f] [FRAC]	123– 42 44	1	138– 1
2	124– 2	[+]	139– 40
[×]	125– 20	[RCL] 0	140– 45 0
[+]	126– 40	[−]	141– 30
[×]	127– 20	[RCL] 5	142– 45 5
2	128– 2	[÷]	143– 10
[÷]	129– 10	[×]	144– 20
[RCL] 5	130– 45 5	[GTO] 3	145– 22 3
[÷]	131– 10	[f] [LBL] 2	146–42,21, 2
[RCL] 1	132– 45 1	[RCL] 0	147– 45 0
[RCL] 2	133– 45 2	[RCL] 3	148– 45 3
[−]	134– 30	[f] [$x \leq y$]	149– 42 10
[×]	135– 20	[g] [SF] 0	150–43, 4, 0
[g] [LSTx]	136– 43 36		

REGISTERS			R_I: Unused
R_0: *YR*	R_1: *SBV*	R_2: *SAL*	R_3: *LIFE*
R_4: *FACT, W*	R_5: *DEP, SOYD*	R_6: Unused	

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYSTROKES	OUTPUT DATA/UNITS
1	Clear User mode, then key in the			
	program.			
2	Set User mode.		[f] [USER]	
3	• Declining-Balance and			

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYSTROKES	OUTPUT DATA/UNITS
	Straight-Line			
	Enter			
	Life	*LIFE*	[ENTER]	*LIFE*
	Salvage value	*SAL*	[ENTER]	*SAL*
	Starting book value	*SBV*	[ENTER]	*SBV*
	Starting year	*YR*	[A]	*LIFE*
	Factor ($1 \leq FACT \leq 2$),			
	(1 for straight-line)	*FACT*	[R/S]	DEP_i
	OR			
	● Sum-of-the-Years-Digits			
	Enter			
	Life	*LIFE*	[ENTER]	*LIFE*
	Salvage value	*SAL*	[ENTER]	*SAL*
	Starting book value	*SBV*	[ENTER]	*SBV*
	Year	*YR*	[B]	DEP_i
4	To find DEP_i, RDV_i,			DEP_i
	and RBV_i.		[R/S]	RDV_i
			[R/S]	RBV_i
			[R/S]	DEP_{i+1}
5	Press [R/S], then repeat step 4			
	for the next year.			
6	To skip to a desired year			
	(forward or back).	*YR*	[C]	DEP_k
			[R/S]	RDV_K
			[R/S]	RBV_K
			[R/S]	DEP_{K+1}
7	Press [R/S], then go to step 4			
	for next year.			

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYSTROKES	OUTPUT DATA/UNITS
8	For a new case go to step 3.			
	Note: The factor may be changed at			
	any point in steps 4 or 5 by storing			
	the new *FACT* in R_4.			

Example:

For a starting book value of $375,000, a salvage value of $30,000 and an expected life of 40 years, generate the 1st-year's depreciation schedule using each of the common methods. Assume a declining-balance factor of 1.5. Then jump ahead to the 15th-year and generate the data for that year.

Keystrokes	Display	
Set User mode.		
40 [ENTER]		
30000 [ENTER]		
375000 [ENTER] 1 [A]	**40.00**	Life.
1 [R/S] (straight line)	**8,625.00**	1st-year's depreciation.
[R/S]	**336,375.00**	Remaining depreciable value.
[R/S]	**366,375.00**	Remaining book value.
15 [C]	**8,625.00**	15th-year's depreciation.
[R/S]	**215,625.00**	Remaining depreciable value.
[R/S]	**245,625.00**	Remaining book value.
1.5 [STO] 4		Change to declining-balance by storing a factor other than 1.
1 [C]	**14,062.50**	1st-year's depreciation.
[R/S]	**330,937.50**	Remaining depreciable value.
[R/S]	**360,937.50**	Remaining book value.
15 [C]	**8,235.18**	15th-year's depreciation.
[R/S]	**181,369.51**	Remaining depreciable value.

Keystrokes	Display	
[R/S]	**211,369.51**	Remaining book value.
40 [ENTER]		SOYD
30000 [ENTER]		
375000 [ENTER] 1 [B]	**16,829.27**	1st-year's depreciation.
[R/S]	**328,170.73**	Remaining depreciable value.
[R/S]	**358,170.73**	Remaining book value.
15 [C]	**10,939.02**	15th-year's depreciation.
[R/S]	**136,737.80**	Remaining depreciable value.
[R/S]	**166,737.80**	Remaining book value.

Moving Average

In a moving average, a specified number of data points is averaged. When there is a new piece of input data, the oldest piece of data is discarded to make room for the most recent input. This replacement scheme makes the moving average a valuable tool in following trends. The fewer the number of data points, the more trend-sensitive the average becomes. With a large number of data points, the average behaves more like a regular average, responding slowly to new input data.

This program allows for a moving average span of 1 to 17 units. The number of units, n, must be specified before any data input begins by keying it in and pressing [A]. Then the data is input by keying in each value, x_k, and pressing [B] in turn. The calculator will display the current input number, k, until at least n values have been entered. After the nth value (and for all succeeding values), the calculator will flash the current input number before halting with the moving average, AVG, in the display.

The value of the average may be displayed at any time by pressing [C]. This feature allows the average to be calculated before n data points have been input. The average is based on the number of inputs or n, whichever is smaller.

Equations:

$\bar{x}$ = Moving Average

m = Number of elements in the moving average

$$\bar{x}_1 = \frac{x_1 + x_2 + x_3 + \ldots + x_m}{m}$$

$$\bar{x}_2 = \frac{x_2 + x_3 + x_4 + \ldots + x_{m+1}}{m}$$

The program is adapted from the HP-65 User's Library program number 01275A by Louis Martinez.

KEYSTROKES	DISPLAY	KEYSTROKES	DISPLAY
[f] CLEAR [PRGM]	000–	[STO] [+] 0	016–44,40, 0
[f] [LBL] [A]	001–42,21,11	[R↓]	017– 33
[f] [FIX] 2	002–42, 7, 2	[$x \gtrless y$]	018– 34
[f] CLEAR [REG]	003– 42 34	[STO] [.] 9	019– 44 .9
[STO] [.] 8	004– 44 .8	[RCL] [.] 8	020– 45 .8
[STO] [I]	005– 44 25	[f] [$x \leq y$]	021– 42 10
[R/S]	006– 31	[GSB] 0	022– 32 0
[f] [LBL] [B]	007–42,21,12	[f] [DSE]	023– 42 5
[RCL] [.] 9	008– 45 .9	[GTO] 5	024– 22 5
1	009– 1	[RCL] [.] 8	025– 45 .8
[+]	010– 40	[STO] [I]	026– 44 25
[$x \gtrless y$]	011– 34	[f] [LBL] 5	027–42,21, 5
[RCL] [(i)]	012– 45 24	[R↓]	028– 33
[STO] [−] 0	013–44,30, 0	[g] [RTN]	029– 43 32
[$x \gtrless y$]	014– 34	[f] [LBL] 0	030–42,21, 0
[STO] [(i)]	015– 44 24	[$x \gtrless y$]	031– 34

KEYSTROKES	DISPLAY	KEYSTROKES	DISPLAY
[f] [PSE]	032– 42 31	[RCL] 0	039– 45 0
[RCL] 0	033– 45 0	[RCL] [.] 9	040– 45 .9
[RCL] [.] 8	034– 45 .8	[RCL] [.] 8	041– 45 .8
[÷]	035– 10	[f] [$x \leq y$]	042– 42 10
[ENTER]	036– 36	[$x \geq y$]	043– 34
[g] [RTN]	037– 43 32	[R↓]	044– 33
[f] [LBL] [C]	038–42,21,13	[÷]	045– 10

REGISTERS			R_I: Control
R_0: Σ	R_1: Used	R_2: Used	R_3: Used
R_4: Used	R_5: Used	R_6: Used	R_7: Used
R_8: Used	R_9: Used	$R_{.0}$: Used	$R_{.1}$: Used
$R_{.2}$: Used	$R_{.3}$: Used	$R_{.4}$: Used	$R_{.5}$: Used
$R_{.6}$: Used	$R_{.7}$: Used	$R_{.8}$: n	$R_{.9}$: k

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYSTROKES	OUTPUT DATA/UNITS
1	Clear User mode, then key in the			
	program.			
2	Set User mode.		[f] [USER]	
3	Input the number of points in the			
	average ($1 \leq n \leq 17$).	n	[A]	n
4	(Optional) Display average			
	at any time.		[C]	*AVG*
5	Input data point and compute			
	the moving average.*	x_k	[B]	k, *AVG*
6	For new case go to step 3.			

*If an error is made in data input, you must start over. The average is not displayed until after the nth point is input.

Example 1:

A six-period moving average is used to project monthly sales. The first 6 months of sales are as follows:

Month	1	2	3	4	5	6
Sales	125	183	207	222	198	240

Compute the moving average. Also compute the average after month three.

Keystrokes	**Display**	
Set User mode.		
6 [A]	**6.00**	
125 [B]	**1.00**	
183 [B]	**2.00**	
207 [B]	**3.00**	
[C]	**171.67**	Average after month three.
222 [B]	**4.00**	
198 [B]	**5.00**	
240 [B]	**6.00**	
	195.83	

The actual sales for the seventh month totalled 225 units. Calculate a new moving average.

225 [B]	**7.00**	
	212.50	

Break-Even Analysis

This program solves the following equations for Break-Even Point in units (BEP_U), Break-Even Point in dollars (BEP_D), Margin of Safety Ratio (M), and Profit or Loss.

Computation based on units:

$$BEP_U = \frac{F}{S - V}$$

$$M_U = \frac{U - BEP_U}{U}$$

$$(\text{Profit or Loss})_U = U(S - V) - F$$

Computation based on Dollars:

$$BEP_D = \frac{F}{R}$$

$$M_D = \frac{D - BEP_D}{D}$$

$$(\text{Profit or Loss})_D = DR - F$$

where

F = Total fixed costs

V = Variable costs per unit

S = Sales price per unit

U = Expected sales in units

D = Expected sales in dollars

R = Marginal income ratio = $(S - V)/S$

Note: The margin of safety ratio will generally have no meaning if expected sales are less than sales at the break-even point.

KEYSTROKES	DISPLAY	KEYSTROKES	DISPLAY
[f] CLEAR [PRGM]	000–	[STO] 5	011– 44 5
[f] [LBL] [A]	001–42,21,11	[f] [LBL] 4	012–42,21, 4
[STO] 1	002– 44 1	[RCL] 1	013– 45 1
[R↓]	003– 33	[RCL] 5	014– 45 5
[STO] 2	004– 44 2	[÷]	015– 10
[x≷y]	005– 34	[STO] 7	016– 44 7
[STO] 3	006– 44 3	[RCL] 1	017– 45 1
[-]	007– 30	[RCL] 4	018– 45 4
[STO] 4	008– 44 4	[÷]	019– 10
[RCL] 2	009– 45 2	[STO] 6	020– 44 6
[÷]	010– 10	[R/S]	021– 31

KEYSTROKES	DISPLAY	KEYSTROKES	DISPLAY
[x≷y]	022– 34	[STO] 8	038– 44 8
[R/S]	023– 31	[RCL] 5	039– 45 5
[f] [LBL] [B]	024–42,21,12	[×]	040– 20
[STO] 8	025– 44 8	[RCL] 1	041– 45 1
[RCL] 4	026– 45 4	[−]	042– 30
[×]	027– 20	[R/S]	043– 31
[RCL] 1	028– 45 1	[RCL] 8	044– 45 8
[−]	029– 30	[RCL] 7	045– 45 7
[R/S]	030– 31	[−]	046– 30
[RCL] 8	031– 45 8	[RCL] 8	047– 45 8
[RCL] 6	032– 45 6	[÷]	048– 10
[−]	033– 30	[R/S]	049– 31
[RCL] 8	034– 45 8	[f] [LBL] [D]	050–42,21,14
[÷]	035– 10	[STO] [+] 1	051–44,40, 1
[R/S]	036– 31	[GTO] 4	052– 22 4
[f] [LBL] [C]	037–42,21,13		

REGISTERS			R_I: Unused
R_0: Unused	R_1: F	R_2: S	R_3: V
R_4: $S-V$	R_5: R	R_6: BEP_U	R_7: BEP_D
R_8: U or D	R_9—$R_{.9}$ Unused		

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYSTROKES	OUTPUT DATA/UNITS
1	Clear User mode, then key in the			
	program.			
2	Set User mode.		[f] [USER]	
3	Enter data:			
	Variable cost per unit	*V*	[ENTER]	
	Sales price per unit	*S*	[ENTER]	
	Fixed costs	*F*		
4	Compute break-even points			
	in units.		[A]	BEP_U
5	Compute break-even point			
	in dollars.		[R/S]	BEP_D
6	Enter expected sales in units.	*U*		
7	Compute $profit_U$ (or loss if			
	negative)		[B]	$(\text{Profit or Loss})_U$
	and compute margin of safety ratio.		[R/S]	M_U
8	Enter expected sales in dollars.	*D*		
9	Compute $profit_D$ (or loss if			
	negative)		[C]	$(\text{Profit or Loss})_D$
	and compute margin of safety ratio.		[R/S]	M_D
10	(Optional) To compute sales			
	volume necessary to provide a			
	desired profit:			
	Compute break-even point			
	in units	*Desired profit*		
		(in $)	[D]	BEP_U
	Compute break-even point			
	in dollars.		[R/S]	BEP_D
11	For a new case go to step 3.			

Example:

The Delux Publishing Company publishes a magazine with variable costs of $0.40 and a sales price of $0.50. The company has annual fixed cost of $1,000,000.

Compute the following:

1. Break-even point in (a) units and (b) dollars.
2. (a) Profit or loss and (b) Margin of safety ratio for expected sales of 12,500,000 magazines.
3. (a) Profit or loss, and (b) Margin of safety ratio for expected sales of $20,000,000.
4. Sales volume in (a) units, and (b) dollars needed to generate a profit of $5,000,000.

Keystrokes	Display	
Set User mode.		
[f] [FIX] 2		
.4 [ENTER] .5 [ENTER]		
[EEX] 6 [A]	10,000,000.00	BEP_U
[R/S]	5,000,000.00	BEP_D
125 [EEX] 5 [B]	250,000.00	Profit
[R/S]	0.20	M_U
2 [EEX] 7 [C]	3,000,000.00	Profit
[R/S]	0.75	M_D
5 [EEX] 6 [D]	60,000,000.00	BEP_U
[R/S]	30,000,000.00	BEP_D

Games

Moon Rocket Lander

Imagine for a moment the difficulties involved in landing a rocket on the moon with a strictly limited fuel supply. You're coming down tail-first, free-falling toward a hard rock surface. You'll have to ignite your rockets to slow your descent; but if you burn too much too soon, you'll run out of fuel 100 feet up, and then you'll have nothing to look forward to but cold eternal moon rocks coming faster every second. The object, clearly, is to space your burns just right so that you will alight on the moon's surface with no downward velocity.

The game starts off with the rocket descending at a velocity of 50 feet/second from a height of 500 feet. The velocity and altitude are shown in a combined display as −50.0500, the altitude appearing to the right of the decimal point and the velocity to the left with a negative sign on the velocity to indicate downward motion. Then the remaining fuel is displayed, and the rocket fire countdown begins: "3", "2", "1", "0". Exactly at zero you may key in a fuel burn. A zero burn, which is very common, is accomplished by doing nothing. After a burn the sequence is repeated unless:

1. You have successfully landed—flashing zeros.
2. You have smashed into the lunar surface—flashing crash velocity.

You must take care, however, not to burn more fuel than you have; for if you do, you will free-fall to your doom! The final velocity shown will be your impact velocity (generally rather high). You have 60 units of fuel initially.

Equations:

We don't want to get too specific, because that would spoil the fun of the game, but rest assured that the program is solidly based on some old friends from Newtonian physics:

$$x = x_0 + v_0 t + \frac{1}{2} a t^2$$

$$v = v_0 + at$$

$$v^2 = v_0^2 + 2a(x - x_0)$$

where x, v, a and t are distance, velocity, acceleration and time respectively.

Remarks:

Only integer values for fuel burn are allowed. [R/S] can be used to stop Moon Rocket Lander at any time.

KEYSTROKES	DISPLAY	KEYSTROKES	DISPLAY
[f] CLEAR [PRGM]	000–	4	017– 4
[f] [LBL] [A]	001–42,21,11	[÷]	018– 10
5	002– 5	[RCL] 1	019– 45 1
0	003– 0	[g] [ABS]	020– 43 16
0	004– 0	[+]	021– 40
[STO] 0	005– 44 0	[RCL] 1	022– 45 1
5	006– 5	[g] [x>0]	023– 43 20
0	007– 0	[GSB] 4	024– 32 4
[CHS]	008– 16	[x≷y]	025– 34
[STO] 1	009– 44 1	[CHS]	026– 16
6	010– 6	[f] [PSE]	027– 42 31
0	011– 0	[f] [PSE]	028– 42 31
[STO] 2	012– 44 2	[f] [FIX] 0	029–42, 7, 0
[f] [LBL] 0	013–42,21, 0	[RCL] 2	030– 45 2
[RCL] 0	014– 45 0	[f] [PSE]	031– 42 31
[f] [FIX] 4	015–42, 7, 4	3	032– 3
[EEX]	016– 26	[f] [PSE]	033– 42 31

KEYSTROKES	DISPLAY	KEYSTROKES	DISPLAY
2	034– 2	[GTO] 0	063– 22 0
[f] [PSE]	035– 42 31	[RCL] 1	064– 45 1
1	036– 1	[f] [LBL] 7	065–42,21, 7
[f] [PSE]	037– 42 31	[f] [PSE]	066– 42 31
0	038– 0	[GTO] 7	067– 22 7
[f] [PSE]	039– 42 31	[f] [LBL] 6	068–42,21, 6
[f] [LBL] 9	040–42,21, 9	[RCL] 2	069– 45 2
[RCL] 2	041– 45 2	2	070– 2
[$x \gtrless y$]	042– 34	[.]	071– 48
[f] [$x > y$]	043– 42 20	5	072– 5
[GTO] 6	044– 22 6	[−]	073– 30
[STO] [−] 2	045–44,30, 2	[STO] [+] 0	074–44,40, 0
2	046– 2	2	075– 2
[×]	047– 20	[×]	076– 20
5	048– 5	[STO] [+] 1	077–44,40, 1
[−]	049– 30	[RCL] 0	078– 45 0
[STO] 3	050– 44 3	1	079– 1
2	051– 2	0	080– 0
[÷]	052– 10	[×]	081– 20
[RCL] 0	053– 45 0	[RCL] 1	082– 45 1
[+]	054– 40	[g] [x^2]	083– 43 11
[RCL] 1	055– 45 1	[+]	084– 40
[+]	056– 40	[$\sqrt{x}$]	085– 11
[RCL] 3	057– 45 3	[CHS]	086– 16
[STO] [+] 1	058–44,40, 1	[GTO] 7	087– 22 7
[R↓]	059– 33	[f] [LBL] 4	088–42,21, 4
[STO] 0	060– 44 0	[$x \gtrless y$]	089– 34
[g] [INT]	061– 43 44	[CHS]	090– 16
[g] [$x > 0$]	062– 43 20	[$x \gtrless y$]	091– 34

REGISTERS			R_I: Unused
R_0: *x*	R_1: *v*	R_2: Fuel	R_3: Accel.
R_4—$R_{.5}$ Unused			

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYSTROKES	OUTPUT DATA/UNITS
1	Clear User mode, then key in the			
	program.			
2	Set User mode.		[f] [USER]	
3	Assume manual control.		[A]	V.ALT
				FUEL
				3.
				2.
				1.
				0.
4	Key in burn:			
	Upon "0" display, press			
	[R/S] then		[R/S]	
	enter burn.	BURN	[R/S]	V.ALT
				FUEL
				3.
				2.
				1.
				0.
5	Go to step 4 until you land			
	(flashing zeros) or crash			
	(flashing impact velocity).			
6	If you survived last landing			
	attempt, go to step 3 for			
	another try.			

Arithmetic Teacher

The hand-held calculator, far from threatening the traditional tenets of a sound mathematics education, may be used creatively to reinforce learning in such areas as arithmetic, algebra, geometry, trigonometry, calculus, and numerical analysis. This program, which is designed to be used in teaching children the four operations of elementary arithmetic (+, −, ×, ÷) demonstrates some of the (largely unexplored) potential of the HP-11C as an educational tool.

The basic flow of the program is to pose a problem in arithmetic, check the answer that the user enters against the correct answer, and then do one of two things: if the user's answer was correct, the program will go on to pose a new problem; if the entered answer was wrong, the program restates the original problem to give the learner a second chance.

To run the program, the user must input a value called *MAX*. This tells the program not to use any numbers as large as *MAX* in its problems. If you specify a *MAX* of 12, for example, then all the problems will deal with numbers between 0 and 11. The user must then input a "seed" *s*, a number between 0 and 1, which will determine the sequence of problems that will appear. Different seeds generate different problems, thus ensuring that the learning game doesn't get boring. The execution of the program will cause the first problem to be displayed as follows: the display will show one number to the left of the decimal place, and one number to the right. For example, the numbers 8 and 2 would be displayed as 8.02. The user may then choose what operation to perform on the two numbers: he may add (8 + 2), subtract (8 − 2), multiply (8 × 2), or divide (8 ÷ 2). After he keys in his answer and re-initiates program execution, the program will either display a new problem, if his answer was right, or display the same two numbers again, but this time with a negative sign in front (−8.02). The negative sign is an indication that the answer was incorrect, and does not denote a negative number. (All numbers in the problem are positive, though of course the results of some subtractions may be negative.) If the problem reappears with a negative sign, the user should key in a different answer and try again. As soon as the correct answer is given, the program will display a new problem.

KEYSTROKES	DISPLAY	KEYSTROKES	DISPLAY
[f] CLEAR [PRGM]	000-	[RCL] 2	025- 45 2
[f] [LBL] [A]	001-42,21,11	[RCL] 3	026- 45 3
[f] [FIX] 2	002-42, 7, 2	[+]	027- 40
[STO] 0	003- 44 0	[GTO] 1	028- 22 1
[R/S]	004- 31	[f] [LBL] [B]	029-42,21,12
[STO] [RAN#]	005- 44 36	[RCL] 2	030- 45 2
[f] [LBL] 3	006-42,21, 3	[RCL] 3	031- 45 3
[f] [RAN#]	007- 42 36	[−]	032- 30
[RCL] 0	008- 45 0	[GTO] 1	033- 22 1
[×]	009- 20	[f] [LBL] [C]	034-42,21,13
[g] [INT]	010- 43 44	[RCL] 2	035- 45 2
[STO] 3	011- 44 3	[RCL] 3	036- 45 3
[f] [RAN#]	012- 42 36	[×]	037- 20
[RCL] 0	013- 45 0	[GTO] 1	038- 22 1
[×]	014- 20	[f] [LBL] [D]	039-42,21,14
[g] [INT]	015- 43 44	[RCL] 2	040- 45 2
[STO] 2	016- 44 2	[RCL] 3	041- 45 3
[RCL] 3	017- 45 3	[÷]	042- 10
[EEX]	018- 26	[f] [LBL] 1	043-42,21, 1
2	019- 2	[f] [$x=y$]	044- 42 40
[÷]	020- 10	[GTO] 3	045- 22 3
[+]	021- 40	[RCL] 4	046- 45 4
[STO] 4	022- 44 4	[CHS]	047- 16
[f] [LBL] 2	023-42,21, 2	[GTO] 2	048- 22 2
[R/S]	024- 31		

REGISTERS			R_I: Unused
R_0: *MAX*	R_1: Unused	R_2: Left #	R_3: Right #
R_4: Problem	R_5—$R_{.9}$ Unused		

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYSTROKES	OUTPUT DATA/UNITS
1	Clear User mode, then key in the			
	program			
2	Set User mode.		[f] [USER]	
3	Input upper limit.	*MAX*	[A]	*MAX*
4	Input random seed. ($0 < s < 1$)	s	[R/S]	$r_1 . n_2$
5	Key in your answer then			
	choose an operation.			
	• Addition (+)	$r_1 + n_2$	[R/S]	
	• Subtraction (−)	$r_1 - n_2$	[B]	
	• Multiplication (×)	$r_1 \times n_2$	[C]	
	• Division (÷)	$r_1 \div n_2$	[D]	
6	If you were right, a new			
	problem will be generated;			
	go to step 5.			$r_3 . n_4$
7	If you were wrong, the same			
	problem will be displayed			
	with a negative sign;			
	go to step 5.			$-r_1 . n_2$
8	Repeat steps 5-7 as many			
	times as desired.			
9	To reset *MAX* go to step 3.			

Example:

$$MAX = 10, \quad s = 0.2$$

Keystrokes	Display
10 [A]	**10.00**
.2 [R/S]	**9.03**
12 [R/S] (9 + 3)	**5.05**
0 [B] (5 − 5)	**4.00**

Keystrokes	Display	
0 [C] (4 × 0)	**3.05**	
7 [R/S] (3 + 5)	**-3.05**	$3+5 \neq 7$
8 [R/S] (3 + 5)	**1.04**	
etc.		

Nimb

The game of Nimb begins with a collection of *N* objects, or as the calculator plays it, with the positive number *N*. Each player alternately subtracts one, two, or more from the total until only one is left. The player forced to take the last one loses.

To begin the game, you specify the maximum number that can be taken in a single move. Then you tell the calculator how many objects you wish to start with (i.e., the value of *N*).

After each move the machine will display the remaining total. A negative sign indicates that it is the player's move next, while a positive display indicates that it is the HP-11C's move.

As the challenger, you are allowed to make the first move. It is possible for you to win, but of course the HP-11C is a master player; it will not let you make an error and win. If you cheat by taking more than the specified limit, the calculator will catch you and force you to repeat the move.

This program is based on an HP-25 program by James L. Horn.

KEYSTROKES	DISPLAY	KEYSTROKES	DISPLAY
[f] CLEAR [PRGM]	000-	[.]	010- 48
[f] [LBL] [A]	001-42,21,11	1	011- 1
[STO] 0	002- 44 0	[STO] 2	012- 44 2
1	003- 1	5	013- 5
[+]	004- 40	5	014- 5
[STO] 1	005- 44 1	1	015- 1
3	006- 3	7	016- 7
5	007- 5	8	017- 8
0	008- 0	[STO] 3	018- 44 3
7	009- 7	[RCL] 0	019- 45 0

KEYSTROKES	DISPLAY	KEYSTROKES	DISPLAY
[f] [LBL] 1	020–42,21, 1	[x≥y]	042– 34
[R/S]	021– 31	[STO] [−] 0	043–44,30, 0
[f] [LBL] [B]	022–42,21,12	[RCL] 0	044– 45 0
[f] [FIX] 0	023–42, 7, 0	[R/S]	045– 31
[STO] 0	024– 44 0	1	046– 1
[f] [LBL] 4	025–42,21, 4	[−]	047– 30
[CHS]	026– 16	[RCL] 1	048– 45 1
[R/S]	027– 31	[÷]	049– 10
[+]	028– 40	[f] [FRAC]	050– 42 44
[g] [x<0]	029– 43 10	[RCL] 1	051– 45 1
[GTO] 0	030– 22 0	[×]	052– 20
[RCL] 3	031– 45 3	[g] [x=0]	053– 43 40
[GTO] 1	032– 22 1	1	054– 1
[f] [LBL] 0	033–42,21, 0	[STO] [−] 0	055–44,30, 0
[g] [LSTx]	034– 43 36	[f] [LBL] 2	056–42,21, 2
1	035– 1	[RCL] 0	057– 45 0
[f] [x>y]	036– 42 20	[g] [x≠0]	058– 43 30
[GTO] 2	037– 22 2	[GTO] 4	059– 22 4
[R↓]	038– 33	[RCL] 2	060– 45 2
[RCL] 1	039– 45 1	[f] [FIX] 1	061–42, 7, 1
[f] [x≤y]	040– 42 10	[GTO] 1	062– 22 1
[GTO] 2	041– 22 2		

REGISTERS			R_I: Unused
R_0: Total	R_1: *MAX* + 1	R_2: 3507.1	R_3: 55178
R_4—R_9 Unused			

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYSTROKES	OUTPUT DATA/UNITS
1	Clear User mode, then key in the			
	program.			
2	Set User mode.		[f] [USER]	
3	Indicate the maximum number			
	of objects which can			
	be removed in one move.	*MAX*	[A]	*MAX*
4	Indicate the total number of			
	objects with which you wish to			
	start the game, (usually 15).	*N*	[B]	−N
5	If the number in the display is			
	negative, key in your move and			
	see the number remaining.	MOVE	[R/S]	+ REM
6	If the number in the display is			
	positive, let the HP-11C move.		[R/S]	− REM
7	Repeat steps 4 and 5 until the			
	game is over.			
8	At the end of the game turn			
	the calculator upside down to			
	read the message.			
	If calculator loses:			I LOSE
	If calculator wins:			BLISS
9	For another game:			
	If max. move remains same,			
	go to step 4.			
	For different max. move			
	go to step 3.			

Example:

Starting with 15 objects, with a maximum allowable move of 3, play Nimb with the HP-11C.

Keystrokes	Display	
Set User mode.		
3 [A]	3.	
15 [B]	-15.	Ready
3 [R/S]	12.	Player takes 3.
[R/S]	-9.	HP-11C takes 3.
5 [R/S]	-9.	Player tries to cheat.
2 [R/S]	7.	Player takes 2.
[R/S]	-5.	HP-11C takes 2.
3 [R/S]	2.	Player takes 3.
[R/S]	-1.	HP-11C takes 1.
1 [R/S]	55178.	Player takes last one and loses.

Turn calculator upside down for message: BLISS.

Surveying

Field Angle Traverse

This program calculates coordinates of a traverse from field angles and horizontal or slope distances. The total horizontal distance traversed and the enclosed area (for a closed traverse) are also calculated.

In running this program, the user inputs the northing and easting of his starting point, the reference azimuth, and then the direction and distance from each point in the traverse to the next point. The direction may be input either as a deflection right or left, or as an angle right or left. The distance may be input either as horizontal distance, or as slope distance with zenith angle.

Equations:

$$HD = SD \sin(ZA)$$

$$N_{k+1} = N_k + HD \cos AZ$$

$$E_{k+1} = E_k + HD \sin AZ$$

$$LAT_k = N_{k+1} - N_k$$

$$DEP_k = E_{k+1} - E_k$$

$$\text{Area} = \sum_{k=1}^{n} LAT_k \left(\tfrac{1}{2} DEP_k + \sum_{j=1}^{k-1} DEP_j \right)$$

where:

N, E = Northing, easting of a point

k refers to current point

n equals number of points in the survey

AZ = Azimuth of a course

HD = Horizontal distance

SD = Slope distance

ZA = Zenith angle

Remarks:

- All angular inputs and outputs are in the form degrees, minutes and seconds (D.MS).
- This program uses zenith angles to calculate the horizontal distance from slope distance. If your instrument measures vertical angles rather than zenith angles, convert the vertical angle to a zenith angle by the following formula:

  zenith angle = 90° − vertical angle

  (Remember to convert D.MS input to decimal degrees before subtracting from 90°).

KEYSTROKES	DISPLAY	KEYSTROKES	DISPLAY
[f] CLEAR [PRGM]	000–	1	017– 1
[f] [LBL] [A]	001–42,21,11	8	018– 8
[f] CLEAR [REG]	002– 42 34	0	019– 0
[f] [FIX] 4	003–42, 7, 4	[+]	020– 40
[STO] 1	004– 44 1	[f] [→H.MS]	021– 42 2
[R/S]	005– 31	[f] [LBL] [B]	022–42,21,12
[STO] 2	006– 44 2	[g] [→H]	023– 43 2
[R/S]	007– 31	[STO] [+] 0	024–44,40, 0
[g] [→H]	008– 43 2	[f] [→H.MS]	025– 42 2
1	009– 1	[R/S]	026– 31
8	010– 8	[GTO] 1	027– 22 1
0	011– 0	[f] [LBL] [C]	028–42,21,13
[+]	012– 40	[x≷y]	029– 34
[STO] 0	013– 44 0	[g] [→H]	030– 43 2
[R/S]	014– 31	[SIN]	031– 23
[f] [LBL] 2	015–42,21, 2	[×]	032– 20
[g] [→H]	016– 43 2	[f] [LBL] 1	033–42,21, 1

KEYSTROKES	DISPLAY	KEYSTROKES	DISPLAY
STO + 3	034–44,40, 3	×	046– 20
RCL 0	035– 45 0	STO + 4	047–44,40, 4
x≷y	036– 34	RCL 1	048– 45 1
f →R	037– 42 26	R/S	049– 31
STO + 1	038–44,40, 1	RCL 2	050– 45 2
x≷y	039– 34	R/S	051– 31
STO + 5	040–44,40, 5	GTO 2	052– 22 2
STO + 2	041–44,40, 2	f LBL D	053–42,21,14
2	042– 2	RCL 3	054– 45 3
÷	043– 10	R/S	055– 31
RCL 5	044– 45 5	RCL 4	056– 45 4
−	045– 30	g ABS	057– 43 16

REGISTERS			R_I: Unused
R_0: *AZ*	R_1: Current *N*	R_2: Current *E*	R_3: Σ*HD*
R_4: *Area*	R_5: Σ*DEP*	R_6—R_9 Unused	

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYSTROKES	OUTPUT DATA/UNITS
1	Clear User mode, then key in the			
	program.			
2	Set User mode.		f USER	
3	Set degree mode.		g DEG	
4	Input the starting			
	point coordinates.	N_1	A	N_1
		E_1	R/S	E_1
5	Input the reference azimuth			
	away from beginning point.	Ref. *AZ*		
		(D.MS)	R/S	

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYSTROKES	OUTPUT DATA/UNITS
6a	If angle right:	*AR* (D.MS)	[R/S]	
6b	If angle left:	*AL* (D.MS)	[CHS] [R/S]	
6c	If deflection right:	*DR* (D.MS)	[B]	
6d	If deflection left:	*DL* (D.MS)	[CHS] [B]	
7	Input horizontal distance,	*HD*	[R/S]	N_i
			[R/S]	E_i
	OR,			
	input zenith angle	*ZA* (D.MS)	[ENTER]	
	and slope distance.	*SD*	[C]	N_i
			[R/S]	E_i
8	Repeat steps 6 and 7			
	for successive courses.			
9	Display total			
	horizontal distance			
	traversed and area for		[D]	ΣHD
	closed traverse.		[R/S]	Area

Example:



Keystrokes	**Display**	
Set User mode.		
[g] [DEG]		
150 [A]		
400 [R/S]		
311.3955 [R/S]		
113.3455 [R/S]		
177.966 [R/S]	**224.5150**	N_2
[R/S]	**561.6150**	E_2
100.2455 [CHS] [B]		
86.0139 [ENTER]		
161.88 [C]	**356.5285**	N_3
[R/S]	**468.5999**	E_3
87.3559 [R/S]		
203.69 [R/S]	**232.3373**	N_4
[R/S]	**307.1498**	E_4
100.4559 [CHS] [B]		
124 [R/S]	**149.9048**	N_1
[R/S]	**399.7829**	E_1
[D]	**667.1471**	ΣHD
[R/S]	**26,558.8326**	Area

Mathematics

Statistics

Electrical Engineering

Mechanical Engineering

Chemistry

Economic Analysis

Games

Surveying

HEWLETT PACKARD

1000 N.E. Circle Blvd., Corvallis, OR 97330, U.S.A.

00011-90009

Printed in U.S.A.